
v

머리말

이책을쓴이유

이책을쓰게된동기는바로독자들에게파스칼과같은언어들보다낮은수준에서

컴퓨터가실제로어떻게작동하는지에대한자세한이해를위해서이다. 컴퓨터가실

제로어떻게작동하는지에대해자세히알게된다면독자여러분은 C 나 C++ 과같은

고급언어를통해프로그램을개발하는것을훨씬효율적으로할수있을것이다.이목

적을이루기위해선어셈블리언어를배우는것이가장효과적이다.다른 PC어셈블리

언어책들은아직도1981년 PC에사용되었던8086프로세서에서어떻게프로그램을

만드는지에대해다루고있다. 이 8086 프로세서는오직실제모드만을지원한다. 이

모드에선어떠한프로그램이라도컴퓨터의메모리나다른장치들에게자유롭게접근

이 가능하다. 이 모드는 안정적인 멀티 태스킹 운영체제에는 적합하지 않다. 이 책은

그대신보호모드에서작동하는 (현재의윈도우즈나리눅스가실행되는) 80386 이나

그이후에나온프로세서들에서어떻게프로그램을만드는지에대해다루고있다. 이

모드는현대의운영체제들이바라는기능들을가지고있다.예를들자면가상메모리나

메모리보호등이다.보호모드를이용하는데에는여러이유들이있다.

1. 다른책이다루는 8086 실제모드에서보다프로그래밍하기가편하다

2. 현대모든 PC 운영체제는보호모드에서작동된다.

3. 보호모드에서작동되는많은수의무료프로그램들이있다.

따라서제가이책을쓰게된가장큰동기는바로보호모드 PC어셈블리프로그래밍에

대한책의부족때문이다.

앞에서살짝말했듯이이책은무료/공개소스소프트웨어만을사용한다. 그예로,

NASM 어셈블러와 DJGPP C/C++ 컴파일러가있다. 이둘은모두인터넷을통해서다

운로드가가능하다. 또한이책은앞으로리눅스에서 NASM 어셈블리코드를어떻게

사용할것인지와 Windows 에서볼랜드사와마이크로소프트사의 C/C++ 컴파일러를

어떻게 사용할 것인지에 대해 다룰 것 이다. 앞서 말한 모든 프로그램들은 나의 웹사

이트 http://www.drpaulcarter.com/pcasm 에서 다운로드가 가능하다. 이 책에서 다룰

예제들을어셈블하고실행시킬것이라면나의사이트에서프로그램들을꼭다운로드

해야한다.

이 책이 어셈블리 프로그래밍의 모든 부분을 다루지 않았으리라 걱정 할 필요는

없다.나는모든프로그래머가보아야할어셈블리언어의중요한주제들을모두다루

려고노력했다.

제1장

시작하기앞서서

1.1 기수법

컴퓨터내부의메모리는수들로구성되어있다.컴퓨터메모리는그수들을10개의

숫자를이용하는십진법(decimal)을 이용하여저장하지않는다. 그 대신하드웨어를

매우단순화시키는이진법(binary)을이용하여컴퓨터는모든정보를저장한다.먼저,

십진체계에대해알아보도록하자.

1.1.1 십진법

십진체계에서는10개의숫자(0-9)들을이용하여수를나타낸다.십진법에서수의

각자리숫자는그수에서의위치에대한 10의멱수로나타나게된다.예를들면

234 = 2 × 102 + 3 × 101 + 4 × 100

1.1.2 이진법

이진체계에서는오직2개의숫자(0,1)만을이용하며십진법과는달리2를기준으

로나타난다.이진수의각자리숫자는그수에서의위치에대한 2의멱수로나타나게

된다.(참고로,이진수에서각자리숫자를비트(bit)라고부른다)예를들자면

110012 = 1 × 24 + 1 × 23 + 0 × 22 + 0 × 21 + 1 × 20

= 16 + 8 + 1

= 25

위예를통해어떻게이진수가십진수로바뀌어지는지알수있다.아래표 1.1에서

일부십진수들이이진수로어떻게나타나는지보여주고있다.

그림 1.1는어떻게두이진숫자가더해지는지(i.e.,비트(bits))보여주고있다.여기,

그예가있다.

110112

+100012

1011002

십진수 이진수 십진수 이진수
0 0000 8 1000
1 0001 9 1001
2 0010 10 1010
3 0011 11 1011
4 0100 12 1100
5 0101 13 1101
6 0110 14 1110
7 0111 15 1111

표 1.1: 0 부터 15까지의십진수들을이진수로나타냄

이전에올림(carry) 없음 이전에올림있음
0 0 1 1 0 0 1 1

+0 +1 +0 +1 +0 +1 +0 +1
0 1 1 0 1 0 0 1

c c c c

그림 1.1: 이진수의덧셈 (c 는올림(carry)을나타낸다 )

누군가가아래와같은십진나눗셈을하였다고하자.

1234 ÷ 10 = 123 r 4

그는이나눗셈이피제수의가장오른쪽숫자를띄어버리고나머지숫자를오른쪽으로

한 자리 쉬프트 했다는 사실을 알 수 있다. 이진체계에서는 2 로 나누는 연산을 통해

위와동일한작업을발생시킬수있다.예를들어아래와같은이진나눗셈을생각하자.
1):

11012 ÷ 102 = 1102 r 1

이사실을통해서어떤 십진수를 값이 같은이진수로바꿀수있다는것을알수 있다.

이는 그림 1.2 에서 나타난다. 이 방법은 가장 오른쪽의 비트 부터 찾아 나가는데 그

비트를흔히최하위비트(least signiĕcant bit, lsb) 라 부른다. 또한가장왼쪽의비트는

최상위비트(most signiĕcant bit, msb) 라고부른다.메모리의가장기본적인단위는바

이트(byte) 라부르는 8개의비트들로이루어져있다.

1.1.3 16진법

16진수들은 16 을 밑으로 하여 나타내어 진다. 16 진수들은 흔히 자리수가 큰 이

진수들의크기를줄이기위해사용된다. 16 진수가 16개의숫자를사용하기때문에 6

개의숫자가모자르게된다. 이 때문에 9 다음으로나타나는숫자들은 영어대문자 A,
B, C, D, E, F를사용하게된다.따라서 16진수 A는10진수 10과동일한값을가지며

B는11, 그리고 F는15의값을가진다. 16진수는16을밑으로하여수를나타내는데

아래예를보면

1) 참고로수옆에아래첨자로적힌 2는이숫자가십진법으로나타낸수가아니라이진법으로나타낸
수임을알려준다

십진수 이진수

25 ÷ 2 = 12 r 1 11001 ÷ 10 = 1100 r 1
12 ÷ 2 = 6 r 0 1100 ÷ 10 = 110 r 0
6 ÷ 2 = 3 r 0 110 ÷ 10 = 11 r 0
3 ÷ 2 = 1 r 1 11 ÷ 10 = 1 r 1
1 ÷ 2 = 0 r 1 1 ÷ 10 = 0 r 1

따라서 2510 = 110012

그림 1.2: 십진수의변환

589 ÷ 16 = 36 r 13
36 ÷ 16 = 2 r 4
2 ÷ 16 = 0 r 2

결과적으로 589 = 24D16

그림 1.3:

2BD16 = 2 × 162 + 11 × 161 + 13 × 160

= 512 + 176 + 13

= 701

십진수를 16 진수로 변환하기 위해서는 우리가 앞서 이진수를 십진수로 변환하는데

사용했던방법을이용하면된다.다만,이진수로변환하는것은2로나누는것이였지만

16진수로변환하는것은 16으로나누어주면된다. 예시로아래그림 1.3 을보면알

수있다.

16 진수가 유용한 이유는 바로 2진수를 손쉽게 16 진수로 바꿀 수 있기 때문이다.

이진수들은 그 자리수가 너무 커서 다루기 불편하지만 16 진수들은 그 길이가 훨씬

짧아지므로표현하기가매우쉬워진다.

16진수를 2진수로바꾸는방법은단지 16진수의각자리수를 4비트의이진수로

바꾸어주면된다. 예를들어 24D16 은 0010 0100 11012 으로변환된다. 주의해야할

점은 16 진수의 모든 자리수는 반드시 4 비트로 바꾸어 주어야 하는데 (다만 16 진수

의 첫 번째 자리수는 꼭 그렇게 할 필요는 없다) 왜냐하면 4 를 이진수로 바꾸었을 때

416 = 1002 라고해서그냥 100 으로한다면그결과가완전히달라지게된다. 마찬가

지로이진수를 16진수로바꾸는것도매우쉬운데오른쪽끝에서부터 4비트씩끊어

가면서각각을16진수로바꾸면된다.물론이때반드시오른쪽끝에서부터변환해야

워드(word) 2 바이트
더블워드(double word) 4 바이트
쿼드워드(quad word) 8 바이트
패러그래프(paragraph) 16 바이트

표 1.2: 메모리의단위

한다.2) 예를들면

110 0000 0101 1010 0111 11102

6 0 5 A 7 E16

4 비트의 이진수를 니블(nibble) 이라 하고, 따라서 16 진수의 각 자리수는 1 개의

니블에 대응된다. 또한 2 개의 니블은 1 바이트를 나타내며, 1 개의 바이트는 2 자리

16진수에대응된다. 1 바이트가가질수있는값의범위는이진수로나타내면 0부터

11111111 , 16 진수로 나타내면 0 부터 FF, 십진수로 나타내면 0 부터 255 까지 임을

알수있다.

1.2 컴퓨터의구조

1.2.1 메모리

메모리의기본적인단위는바이트이다. 만약어떤컴퓨터가32메가바이트의메모메모리의 크기를 나타내는 단

위들은 킬로바이트( 210 =

1, 024 바이트), 메가바이트

( 220 = 1, 048, 576 바이트)

그리고 기가바이트( 230 =

1, 073, 741, 824바이트).

리를가진다면그컴퓨터는대략 3200 만바이트의정보를보관할수있다. 메모리의

각 바이트는 그림 1.4 에 잘 나와 있듯이 주소(adress)라 불리는 고유의 값을 가지고

있다.

주소 0 1 2 3 4 5 6 7
메모리 2A 45 B8 20 8F CD 12 2E

그림 1.4: 메모리주소

보통메모리는단일바이트보다는여러개의바이트들의묶음으로많이사용된다.

PC 아키텍쳐에서는그묶음들을표 1.2 에서나타나는것처럼이름붙인다.

메모리에 있는 모든 데이터들은 수 이다. 문자들은 문자 코드(character code) 에
나타난숫자들에대응되어기록된다.문자코드중에서가장유명한것으로는아스키

(ASCII, American Standard Code for Information Interchange) 코드를 들 수 있다. 현재

새롭게아스키코드를대체해나가고있는코드는유니코드(Unicode) 이다. 유니코드

와아스키코드의가장큰차이점은아스키코드에서는한개의문자를나타내기위해

1바이트를사용하지만유니코드는 2바이트(혹은워드)를사용한다.예를들어서아

스키코드에서는대문자 A 를나타내기위해바이트 4116(6510) 를 사용하지만유니코

드에서는워드 004116 을사용한다.아스키코드가1바이트만을사용하기때문에오직

256개의서로다른문자들만사용할수있으나 3) 반면에유니코드는아스키코드를확

2) 왜 그래야하는지이해가잘가지않는다면직접어떤이진수를골라서왼쪽부터 16 진수로변환해
보아라

3)사실아스키코드는오직하위7비트만을사용하므로오직128개의서로다른값들만나타낼수있다.

장하여훨씬많은문자들을사용할수있게하였다.이를통해전세계의많은언어들의

문자를나타낼수있게되었다.

1.2.2 CPU

중앙처리장치(CPU)는명령을수행하는장치이다.보통CPU들이수행하는명령들

은매우단순하다.명령들은대부분CPU자체에있는특별한데이타저장소인레지스터

에들어있는값들을필요로한다. CPU는레지스터에있는데이터에메모리에있는데

이터보다훨씬빠른속도로접근할수있다.그러나CPU한개에있는레지스터의수는

제한적이기때문에프로그래머는반드시현재사용되고있는데이터들만레지스터에

보관하도록해야한다. CPU가실행하는명령들은 CPU의기계어(Machine language)
에의해작성되어있다.기계프로그램은고급언어들보다훨씬간단한구조를가지고

있다.기계어의명령들은숫자들에대응이된다. CPU는이러한명령들을빠른속도로

해석하여효율적으로실행해야만한다.따라서기계어는이러한점을항상염두에두어

디자인되기때문에인간이해석하기에는매우난해한면이있다.다른언어로쓰여진

프로그램들은 실행되기 위해서는 반드시 CPU 가 해석할 수 있는 기계어로 변환되어

야만한다.따라서특정한컴퓨터아키텍쳐에대한기계어코드로소스를바꾸어주는

프로그램을컴파일러(compiler) 라고한다.보통각 CPU마다특정한기계어를가지고

있다.이때문에 Mac 에서쓰여진코드가 IBM PC 에서잘작동이되지않는다.

컴퓨터는클록(clock) 을이용하여명령들의실행을동기화한다.클록은특정한진 GHz 는 기가헤르츠의 약자로

초당 10억 번의 펄스가 발생

했다는 뜻이다. 예로 1.5 GHz

CPU는초당15억번의펄스가

발생한다.

동수에따라펄스를방출하는데이진동수를클록속도라고한다.당신이만약1.5GHz
의 컴퓨터를 구매 한다면 1.5Ghz 가 바로 이 컴퓨터의 클록의 진동수이다. 4) 우리가

메트로놈을이용하여음악을정확한리듬으로연주할수있는것처럼 CPU 도언제나

일정하게 클록에서 방출되는 펄스를 이용하여 명령을 정확하게 수행 할 수 있다. 각

명령이 필요로 하는 클록의 진동 횟수(보통 사이클 이라 부른다) 는 CPU 의 세대와

모델에따라차이가난다.

1.2.3 CPU의80x86계열

IBM형의 PC는인텔의80x86계열의 CPU를장착하고있다.이계열의 CPU들은

모두 동일한 기반의 기계어를 사용한다. 최근에 출시된 것들의 경우 그 기능이 대폭

강화되었다.

8088,8086: 이 CPU 들은 초기의 PC 에서 사용되었던 것들이다. 이들은 AX, BX, CX,
DX, SI, DI, BP, SP, CS, DS, SS, ES, IP, FLAGS 와같은 16비트레지스터들을사용

한다. 뿐만아니라최대 1 MB 의메모리만지원하며오직실제모드에서만실행

된다. 이 모드에서는 하나의 프로그램이 메모리의 모든 부분에 접근할 수 있다.

심지어다른프로그램이그부분을이미사용하고있다고해도말이다.이때문이

디버깅과보안을유지하기가매우힘들어지게된다.뿐만아니라프로그램메모

리는반드시세그먼트(segments) 들로나누어져야하는데각각의세그먼트들의

크기는 64KB 를넘을수없다.

80286: 이 CPU 는 AT 계열의 PC 들에서사용되었다.이 CPU 는몇개의새로운명령

들이추가된8088/86기계어를사용하였다.그러나이 CPU의가장중요한새기

4) 사실클록펄스는컴퓨터의여러다른부분들에서도많이사용되는데그부분들의클록속도는 CPU
와다르다.

AX
AH AL

그림 1.5: AX 레지스터

능은바로16비트보호모드(16-bits protected mode)이다.이모드에서는최대16

MB 의메모리에접근할수있고특정한프로그램이다른프로그램의메모리에

함부로접근하는것을방지해준다. 하지만아직도각각의프로그램들은 64 KB
보다작아야하는세그먼트들에서만실행될수있었다.

80386: 이 CPU 는 80286 CPU 를 크게 발전시킨 것이다. 첫번째로, 기존의 16 비트

레지스터들을 32비트레지스터 (EAX, EBX, ECX, EDX, ESI, EDI, EBP, ESP, EIP)
로확장하였고새로운16비트레지스터인 FS와GS를추가하였다.또한이 CPU
는 32 비트 보호 모드(32-bit protected mode) 를 지원한다. 이 모드에서는 최대

4 GB 의 메모리를 사용할 수 있다. 아직도 프로그램들은 세그먼트들에 나뉘어

들어가야만하지만각세그먼트의크기는최대 4 GB 까지될수있다.

80486/펜티엄/펜티엄프로 : 이 CPU 들은기존의 CPU 들과는크게달라진것이없다.

다만명령들의연산속도가매우빨라졌다.

펜티엄MMX: 이프로세서에는MMX(멀티미디아확장,MultiMedia eXtensions) 명령

이추가되었다.이명령은자주실행되는그래픽관련명령들의실행속도를향상

시켰다.

펜티엄 II: 이는펜티엄프로프로세서에MMX명령들이추가된프로세서이다. (펜티

엄 III 의경우단지 II 보다빨라진것이다.)

1.2.4 8086 16비트레지스터

기존의 8086 CPU 는 4개의 16 비트범용프로세서들을지원했는데이는 AX, BX,
CX, 그리고 DX 이다. 각각의레지스터들은 2개의 8비트레지스터들로나뉘어질수

있다. 예를들어 AX 레지스터는 AH, AL 레지스터로나뉠수있는데이는그림 1.5 에

잘나타나있다. AH레지스터는 AX의상위8비트를나타내고, AL은AX의하위8비

트를나타낸다. AH와 AL을종종독립적인1바이트레지스터들로사용된다.그러나

두레지스터들이 AX 와다른것이라고생각하면안된다. AX 레지스터의값을바꾸게

되면 AH 와 AL 의 값도 바뀌고 AH 와 AL 의 값을 바꾸면 AX 의 값이 바뀐다. 범용

레지스터들은산술연산과데이터이동명령들에서잘쓰인다.

CPU에는2개의 SI와DI라불리는16비트색인레지스터가있다.이들은포인터

로자주쓰이지만범용레지스터들과같은용도로종종쓰인다.그러나색인레지스터

들은범용레지스터와는달리 2개의 8비트레지스터들로나누어질수없다.

16 비트 BP 와 SP 레지스터들은기계어스택에들어있는데이터의위치를가리키

는데쓰인다. 이 2 개의레지스터를각각기준포인터와스택포인터라고부른다. 이

들에대해선나중에다루기로하자.

16 비트 CS, DS, SS, ES 레지스터들은세그먼트레지스터(segment register) 라고부

른다. 이 프로그램의 각 부분에서 메모리의 어떤 부분이 사용되는지 가리킨다. CS 는
코드세그먼트(Code Segment),DS는데이타세그먼트(Data Segment), SS는스택세그

먼트(Stack Segment), 그리고 ES는보조세그먼트(Extra Segment) 의약자이다. ES는
임시적인세그먼트레지스터로사용된다. 이레지스터들에대한자세한내용은에서

다룰것이다.

명령 포인터(IP) 레지스터는 CS 레지스터와 함께 사용되는데 이는 CPU 에서 다

음으로 실행될 명령의 주소를 저장한다. 보통 CPU 에서 명령이 하나 실행되면 IP 는

메모리상에서그다음으로실행될명령을가리키게된다.

플래그(FLAG)레지스터들은이전에실행된명령들의중요한결과값들을저장하고

있다.이값들은플래그레지스터의각각의비트에저장되는데예를들며이전연산의

결과가0이였다면플래그레지스터의 Z비트의값은1이되고0이아니라면0이된다.

모든명령이플래그레지스터의비트값들을바꾸는것은아니며이책의부록에보면

어떠한명령들이플래그레지스터의값에영향을주는지알수있다.

1.2.5 80386 32비트레지스터

80386과그이후에출시된프로세서들은모두확장레지스터를가지고있다.예를

들어 16 비트 AX 레지스터는 32 비트로확장되었다. 이때하위호환성을위해 AX 는

계속 16 비트레지스터로남아있고그대신 EAX 가확장된 32 비트레지스터를가리

키게된다.AL이단지AX의하위8비트인것처럼AX도단지 EAX의하위16비트를

나타낸다.이때, EAX의상위16비트에직접적으로접근할수있는방법은없다.다른

확장된레지스터들은 EBX, ECX, EDX, ESI, 그리고 EDI 가있다.

다른 많은 수의 레지스터들도 확장되었다. BP 는 EBP로, SP 는 ESP로, FLAGS 는

EFLAGS로, 그리고 IP 는 EIP가되었다. 그러나 32 비트보호모드에서는범용레지스

터를뺀나머지레지스터들은모두확장된것으로만사용해야한다.

80386 에선 아직 세그먼트 레지스터들이 16 비트로 남아 있다. 하지만 새로운 2

개의세그먼트레지스터들이추가되었는데바로 FS와GS이다.이레지스터들은단지

ES 와같이임시적인세그먼트레지스터들로사용된다.

워드(word) 라는단어는 CPU 데이터레지스터의크기로정의된다. 80x86 계열에

서는위정의가약간혼란스럽다.아래표 1.2에나온것처럼워드는2바이트(혹은16

비트)로정의됨을알수있다.이는처음8086이발표되었을때정해진것이다.하지만

80386 이 개발되었을때레지스터의크기가바뀌었는데도불구하고기존의워드의

정의를그대로남기기로했다.

1.2.6 실제모드

실제 모드 에선 메모리 사용이 오직 1 MB (220 바이트) 로 제한된다. 이는 00000 그렇다면 DOS 의 640KB 제

한은 어디서 나타난 것일까?

BIOS는비디오스크린과같은

하드웨어를위해 1MB의일부

분을필요로한다

부터 FFFFF 에해당하는값이다. 따라서메모리의주소를나타내기위해선 20비트의

숫자를저장할수있는레지스터가필요한데안타깝게도 8086의레지스터는모두 16

비트 밖에 되지 않으므로 그럴 수 없다. 하지만 인텔은 하나의 메모리 주소를 정하기

위해 2 개의 16 비트 값을 사용하여 문제를 해결하였다. 처음의 16 비트 값을 실렉터

(selector) 라고부른다.실렉터의값은반드시세그먼트레지스터에저장되어야만한다.

두번째 16 비트값을오프셋(offset) 이라부른다. 메모리의물리주소는 32 비트의실

렉터 :오프셋쌍으로나타나며그값은

16 ∗ selector + offset

KJH
강조

이다. 16 을 곱하는것은 16 진수에서는단순히오른쪽끝에 0 을하나추가하는것이

므로매우쉬운작업이다.예를들어특정한메모리공간의주소가047C:0048로주어

진다면이는

047C0
+0048
04808

에데이터가저장되어있다는것이다. 사실상실릭터의값은언제나 16 의배수(패러

그래프수)가된다. (표 1.2 참조)

그러나위와같은방법도몇가지단점이있는데 :

• 하나의실렉터를사용하면최대 64 KB 의메모리까지밖에사용할수없다. 따

라서만일프로그램이 64KB 보다더큰메모리를필요로한다면하나의실렉터

만으로는충분하지못할것이다. 이때문에프로그램이 64KB 이상의메모리를

사용시 프로그램은 반드시 64KB 이하의 조각들로 쪼개져야 한다 (이 조각들을

세그먼트 라부른다) 만약 프로그램의명령이다른세그먼트로넘어갈경우 CS
의값도반드시달라져야한다.큰데이터를다룰때 DS 레지스터에서도비슷한

일이발생하게된다.상당히짜증나는일이다!

• 위와같이주소지정을해준다면메모리의공간들은유일한주소값을가지지않는

다.예를들어물리주소04808은047C:0048, 047D:0038, 047E:0028, 047B:0058
등과같이나타날수있다.이때문에두개의주소값이같은지비교시복잡해진다.

1.2.7 16비트보호모드

80286 의 16비트 보호 모드에서 실렉터의 값들은 실제 모드에서의 값과 다른 방

식으로해석된다.실제모드에서는실릭터의값은언제나패러그래프수이였다.보호

모드에서는실릭터의값은디스크립터테이블(descriptor table) 의인덱스(index) 이다.

두모드모두프로그램들이세그먼트들로나뉘어진다.실제모드에서는이러한세그

먼트가 물리 메모리에 고정된 위치에 저장되며 실릭터는 그 세그먼트의 시작 부분의

패러그래프값을가리키게된다.그러나보호모드에서는세그먼트들이물리메모리에

고정된위치에저장되는것이아니다.사실메모리상에존재할필요도없다!

보호모드는가상메모리(virtual memory) 라는기술을이용한다.가상메모리시스

템의기본적인개념은프로그램의현재사용하고있는데이터와코드만메모리상에

저장하는것이다. 다른데이터나코드들은나중에필요할때까지임시적으로디스크

에보관하게된다. 16비트보호모드에선세그먼트들이메모리와디스크사이를왔다

갔다하게되므로세그먼트들이언제나메모리상의고정된부분을점유하고있을수

가없게된다.이러한작업은운영체에의해제어되므로프로그램이가상메모리에서의

작업을위해특별히해주어야될것은없다.

보호모드에서는개개의세그먼트가디스크립터테이블의엔트리에할당되는데이

디스크립터테이블에는시스템이그세그먼트에대해알아야할모든정보가수록되어

있다. 예를들면현재메모리상에있는지의유무, 메모리상에있다면어디에있는지,

접근가능한지(e.g.,읽기전용). 세그먼트엔트리의인덱스값이바로세그먼트레지스

터에저장되어있는실렉터값이다.

16비트 보호 모드의 가장 큰 단점은 오프셋이 아직도 16비트 이라는 점이다. 이

때문에세그먼트의크기는아직도 64KB 로제한된다. 이는큰배열을만들기매우골

치아프게한다. 한 유명한 PC 칼럼니스트는

286 CPU를가리켜 ’뇌사상태’

라고말했다1.2.8 32비트보호모드

80386 이 처음으로 32비트 보호 모드의 장을 열었다. 386 32비트와 286 16비트

보호모드에는 2개의큰차이점이있는데

1. 오프셋이32비트로확장되었다.이는오프셋의값이최대40억까지가능하다는

뜻이다.따라서,하나의세그먼트는최대 4GB 까지가질수있다.

2. 세그먼트들은작은 4KB 크기의페이지(page) 라 부르는단위들로나뉘어질수

있다.가상메모리시스템도세그먼트가아닌페이지를기준으로작업하게된다.

이말은세그먼트의일부분도메모리에존재할수있게되었다는뜻이다.이전의

286 16 비트모드에서는전체세그먼트가메모리상에한꺼번에존재하던지,아

니면메모리상에아무것도존재못하는2가지경우밖에없었었다.이를거대한

32비트모드의세그먼트에그대로적용하는것은이치에맞지않다.

윈도우즈3.x버전에서표준모드(standard mode) 는 286 16비트보호모드를가리

키고향상모드(enhanced mode) 는 32 비트모드를가리킨다. 윈도우즈 9X, 윈도우즈

NT/2000/XP, OS/2, 그리고리눅스에선모든프로그램이페이지가능한 32비트보호

모드에서실행된다.

1.2.9 인터럽트

종종보통의프로그램의흐름이즉시응답이필요한프로세스이벤트(event)로부터

중지되는경우가있다. 컴퓨터하드웨어는인터럽트(interrupt) 라는메커니즘을통해

이러한이벤트를처리한다.예를들어마우스가움직였을때,마우스하드웨어는마우

스움직임을처리(마우스커서를움직이기위해)하기위해현재실행되는일시적으로

프로그램을중지시킨다. 인터럽트는제어흐름이인터럽트핸들러(interrupt handler)
을 통과하게한다. 인터럽트핸들러는인터럽트를처리하는루틴이다. 각각의인터럽

트는하나의정수에대응이되어있다.물리메모리상단에있는인터럽트벡터테이블

(interrupt vector table)에인터럽트핸들러들의주소값을보관한다.인터럽트의번호는

본질적으로테이블의인덱스가되는것이다.

CPU 외부에서오는인터럽트를외부인터럽트(External inturrpt) 라한다. 앞에서

예로나온마우스가외부인터럽트의한예이다. 대다수의입출력(I/O) 장치들, 예를

들어키보드,타이머,디스크드라이브, CD-ROM, 그리고사운드카드들은모두외부

인터럽트를발생시킨다.내부인터럽트는CPU내부에서발생되는인터럽트로오류나

인터럽트 명령에 의해 발생된다. 오류 인터럽트는 트랩(trap) 이라고도 불린다. 인터

럽트명령을통해발생되는인터럽트는소프트웨어인터럽트(soware interrupt) 라부

른다. DOS 는자체 API(Application Programming Interface) 를 구현하기위해이러한

인터럽트를 사용하였다. 많은 수의 현대 운영체제 (윈도우즈, 유닉스 등등) 에서는 C
기반의인터페이스(interface) 를사용한다. 5)

5) 그러나이들도커널수준에서는저급수준의인터페이스를사용할것이다.

KJH
스티커 노트

많은인터럽트핸들러들은자신의역할이끝나면중단되었던프로그램으로제어

흐름이돌려진다. 그들은인터럽트되기전에레지스터에저장되어있던값들을모두

복원한다. 따라서 중단되었던 프로그램도 마치 아무 일이 없었던 것 처럼 다시 원상

태로 실행 될 수 있다. (다만, 일부 CPU 사이클들을 잃어버렸다) 트랩의 경우 대부분

리턴되지않으며종종프로그램을종료시킨다.

1.3 어셈블리어

1.3.1 기계어

CPU 들은자기들만이이해할수있는기계어를가지고있다. 기계어의명령들은

메모리에바이트로저장되어있는숫자들이다.각명령들은자기만의유일한수코드를

가지고있고이것을연산부호(operation code) 또는줄여서 opcode라부른다. 80x86프
로세서명령의크기는각각다르며연산부호는언제나명령앞부분에위치한다.많은

명령들은명령에의해사용되는데이터를포함한다(e.g., 상수나주소)

기계어는직접적으로프로그래밍하기가매우힘들다.수치화된각각의연산부호

들을 해독하는 일이 인간에게는 매우 힘든 일이기 때문이다. 예를 들어, EAX 와 EBX
레지스터의 값을 더해 다시 EAX 레지스터에 대입하는 문장은 아래와 같이 기계어로

나타내진다.

03 C3

이는 정말로 이해하기 힘들다. 다행이도 어셈블러(assembler) 라 불리는 프로그램이

위와같은지루한일을프로그래머가안해도되게해준다.

1.3.2 어셈블리어

어셈블리언어프로그램은고급언어프로그램처럼문자형태로저장된다.각각의

어셈블리명령들은하나의기계명령에대응된다.예를들어서위에예로든덧셈연산

은어셈블리어로아래와같이표현된다.

add eax, ebx

이를 통해 단순히 숫자로 나타낸 기계어 보다 훨씬 알기 쉽게 명령을 나타낼 수 있다.

add 는 덧셈을 한다라는 명령을 인간이 알기 쉬운 문자열로로 대응 시킨 연상 기호

(mnemonic) 이다. 어셈블리어의일반적인형식은아래와같다

mnemonic operand(s)

어셈블러(assembler) 는 어셈블리 명령들로 작성된 파일을 읽어 들어서 기계어로

작성된코드로변환해주는프로그램이다.컴파일러(Compiler) 는위와비슷한작업을

하지만다만고급언어로작성된코드를기계어로변환해준다.모든어셈블리문장은컴퓨터 과학자들은 컴파일러

자체를 어떻게 작성할지에 대

해수년간연구하였다!

하나의 기계어 문장을 직접 적으로 나타낸다. 한편 고급 언어의 경우 하나의 문장이

훨씬많고복잡한기계명령을나타낸다.

어셈블리어와고급언어의또다른차이점은모든 CPU 가각각의고유한기계어를

가지고있고또고유의어셈블리어를가지고있다는점이다.다른컴퓨터아키텍쳐에

어셈블리 프로그램을 포팅(porting) 하는 이는 고급 언어에서 처리하는 것 보다 훨씬

어려운일이다.

이 책에서는 예제들은 Netwide Assembler 또는 NASM 이라 불리는 어셈블리어를

사용할것이다.이는인터넷에서무료로받을수있다. (머리말참조)다른어셈블러들

로는 마이크로소프트 사의 MASM 이나 볼랜드 사의 TASM 을 들 수 있다. NASM 과

MASM/TASM 의문법은약간다르다.

1.3.3 피연산자

기계코드명령들에따라피연산자의형태와개수가다르다.그러나대부분각각의

명령들은고정된수의피연산자수(보통0 - 3)를갖는다.피연산자로올수있는것들은

아래와같다.

레지스터 : 이피연산자들은 CPU 레지스터들에직접적으로접근할수있다.

메모리 : 이는메모리에저장된데이터를가리킨다. 이때, 메모리의주소값은명령에

직접적으로사용하거나,레지스터에저장하여사용할수도있다.언제나세그먼

트최상단부터의오프셋값으로나타낸다.

즉시피연산자 : 즉시피연산자(immediate) 는 명령자체에있는고정된값들이다. 이

들은명령자체에저장된다(즉,데이터세그먼트가아닌코드세그먼트에저장됨)

묵시적피연산자 : 묵시적피연산자(implied) 는정확히나타나지않는다.예를들어서

증가명령은레지스터나메모리에1을더한다.이때 1을묵시적피연산자라고

부른다.

1.3.4 기초명령

어셈블리어에서가장기본이되는명령은 MOV명령이다. 이는특정한지점의데이

터를다른지점으로이동시킨다. (이는고급언어의대입연산자와같다)이명령은두

개의피연산자를필요로한다

mov dest, src

src 에있는데이터가 dest 로복사된다. 이때, 두 피연산자가모두메모리이면안된다.

이는어셈블리어의한가지흠이라볼수있다.어셈블리에서는명령문에따라특정한

기준없이규칙이달라지는경우가있다.두개의피연산자의크기가같아야한다.예를

들어 AX (16 비트)에있는값은 BL (8비트)로저장될수없다.

여기예제가있다. (세미콜론은주석의시작을나타낸다):

mov eax, 3 ; eax 레지스터에 3 을 대입한다 (이 때 3 은 즉시 피연산자 이다)

mov bx, ax ; ax 레지스터의 값을 bx 레지스터에 대입한다.

ADD명령은두개의정수를더할때사용된다.

add eax, 4 ; eax = eax + 4

add al, ah ; al = al + ah

SUB명려은한정수에서다른정수를뺄때사용된다.

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

sub bx, 10 ; bx = bx - 10

sub ebx, edi ; ebx = ebx - edi

INC와 DEC는 1을증가또는감소시키는명령이다. 1 이묵시적피연산자이므로

INC 와 DEC 의 기계 코드는 동일한 작업을 수행하는 ADD 와 SUB 의 기계 코드의 크기

보다작다.

inc ecx ; ecx++
dec dl ; dl--

1.3.5 지시어

지시어(directive) 는 CPU 가아닌어셈블러를위해만들어진것이다. 이것들은주

로어셈블러로하여금무언가를하게하거나아니면어셈블러에게무언가를알려주는

역할을한다.지시어는기계코드로변환되지않는다.지시어들의주용도로는 :

• 상수를정의한다

• 데이터를저장할메모리를정의한다

• 메모리를세그먼드로묶는다

• 조건적으로소스코드를포함시킨다

• 다른파일들을포함시킨다.

NASM은 C와비슷한전처리기명령들이있다.그러나NASM의전처리기지시어

는 C 에서의 # 이아닌 % 을사용한다.

equ지시어

equ지시어는심볼(symbol) 을정의할때사용된다.심볼은어셈블리프로그래밍시

사용되는상수를뜻한다.이는아래와같이사용한다

symbol equ value

한번정의된심볼의값은절대로재정의될수없다.

e%정의지시어

이지시어는C에서의 #define지시어와비슷하다.이는C에서처럼상수매크로를

정의할때사용된다

%define SIZE 100

mov eax, SIZE

위의나온코드는 SIZE라는이름의매크로를정의하고이를 MOV명령에서어떻게

사용하는지보여주고있다.매크로는심볼들보다좀더유연한다.왜냐하면매크로는

심볼들과는다르게재정의될수있고단순한수가아니여도되기때문이다.

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH

KJH
강조

KJH
스티커 노트
symbol equ 이름

KJH
강조

KJH
스티커 노트
eax = SIZE ;

Unit Letter
바이트 B
워드 W

더블워드 D
쿼드워드 Q
10 바이트 T

표 1.3: RESX와 DX지시어를위한글자

데이터지시어

데이터지시어들은데이터세그먼트에서메모리상의공간을정의하는데사용된다.

메모리공간이정의되는데에는2가지방법이있다.첫번째방법은오직데이터를위한

공간들만정의한다. 두 번째방법은초기의값들을위한방들을정의한다. 첫 번째방

법의경우 RESX지시어를사용한다. X 에는저장될데이터의크기를나타내는문자가

들어간다.이는표 1.3를참조하여라.

두번째방법(초기값도정의하는)은 DX지시어를사용한다. X 에는역시 RESX지
시어사용시들어가는문자가들어가게된다.

메모리위치를라벨(label) 로표시하는것은매우흔한일이다.라벨은코드상에서

특정한메모리위치를가리킬수있다.아래예제를참조하면 :

L1 db 0 ; L1 로 라벨 붙혀진 바이트가 0 으로 초기화 됨

L2 dw 1000 ; L2 로 라벨 붙혀진 워드가 1000 으로 초기화 됨

L3 db 110101b ; L3 로 라벨 붙혀진 바이트가 이진수 110101(십진수로 53)로 초기화 됨

L4 db 12h ; L4 로 라벨 붙혀진 바이트가 16진수 12 로 초기화 됨

L5 db 17o ; L5 로 라벨 붙혀진 바이트가 8진수 17 로 초기화됨

L6 dd 1A92h ; L6 로 라벨 붙혀진 더블워드가 16진수 1A92 로 초기화 됨

L7 resb 1 ; L7 로 라벨 붙혀진 초기화 되지 않은 1 바이트

L8 db "A" ; L8 로 라벨 붙혀진 아스키 코드 A(65) 로 초기화 됨

큰따옴표와작은따옴표는모두같은것으로취급된다. 데이터가연속적으로정의

되면 그 데이터들은 메모리 상에 연속되게 존재하게 된다. 따라서, 메모리 상에서 L2
는 L1 바로다음으로위치하게된다.메모리상에서의순서또한정의될수있다.

L9 db 0, 1, 2, 3 ; 4 바이트를 정의한다

L10 db "w", "o", "r", 'd', 0 ; C 문자열 "word" 를 정의한다

L11 db 'word', 0 ; L10 과 같다

DD지시어는정수와단일정밀도부동소수점 6) 상수들을정의할수있다. 그러나

DQ의경우오직 2배정밀도부동소수점상수만정의할수있다.

크기가 큰 것들을 위해선 NASM 의 TIMES 지시어를 이용하면 된다. 이 지시어는

피연산자를특정한횟수만큼반복한다.예를들면

L12 times 100 db 0 ; 이는 100 개의 (db 0) 을 나열하는 것과 같다

L13 resw 100 ; 100 개의 워드를 위한 공간을 정의한다.

6) 단일정밀도부동소수점은 C 언어의 float 와같다

스티커 노트

KJH

KJH
강조

KJH
스티커 노트
포인터와 비슷하다.

KJH
스티커 노트
110101b 는 이진수를 의미1Ah 는 16진수를 의미십진수는 걍 쓰면됨.

KJH
강조

KJH
강조

KJH
강조

KJH
강조

KJH
줄 긋기

KJH
삽입된 텍스트
DQ의 경우는 double 형이라고 봐도 될거같음.그런데 DQ로 라벨을 정의할 경우오직 소수 0.1234 1.455 이런것들로만 라벨을 정의할수 있는거 같음. 즉, 정수는 못온다는 이야기.

KJH
스티커 노트
배열이랑 비슷한거 같음.변수를 한꺼번에 여러개 선언한다는공통점.L12 times 100 db 1



14 제1장 시작하기앞서서

아까 라벨이 코드 상의 데이터를 가리키는데 쓰인다는 말을 상기해보자. 라벨이

사용되는용도는2가지가있다.만약라벨을그냥이용한다면이는데이터의주소 (즉,

오프셋)으로생각된다.만약라벨이대괄호([]) 속에사용된다면이는그주소에위치

한데이터를나타낸다. 다시말해라벨을어떤데이터에대한포인터(pointer) 라 하면

대괄호는 C 언어의 와 같은 역할을 하는 것이다. (MASM/TASM 에서는 다른 방법을

사용한다) 32비트모드에서는주소들의크기는 32비트이다.아래예제를보면 :

1 mov al, [L1] ; AL 에 L1 에 위치한 데이터를 대입한다.

2 mov eax, L1 ; EAX = L1 에 위치한 바이트의 주소

3 mov [L1], ah ; L1 에 위치한 바이트에 ah 를 대입한다

4 mov eax, [L6] ; L6 에 위치한 더블워드를 EAX 에 대입한다

5 add eax, [L6] ; EAX = EAX + L6 에 위치한 더블워드

6 add [L6], eax ; L6 에 위치한 더블워드 += EAX
7 mov al, [L6] ; L6 에 위치한 더블워드의 하위 비트를 AL 에 대입한다

7 번째 행을 보면 NASM 의 중요한 특징을 알 수 있다. 어셈블러는 라벨이 어떠한 데

이터를가리키고있는지전혀신경쓰지않는다. 이는모두프로그래머에달려있기에

언제나라벨을정확히사용하는데신중을기해야할것이다.나중에데이터의주소값을

레지스터에 저장하여 C 언어의 포인터 처럼 쓰는 연산을 많이 하게 된다. 이 때도 이

포인터가 정확하게 사용되고 있는지에 대해선 어셈블러가 전혀 신경쓰지 않는다. 이

때문에어셈블리를이용한프로그래밍은 C 보다도오류가잦아지게된다.

다음과같은명령을고려하자.

mov [L6], 1 ; L6 에 위치한 데이터에 1 을 저장한다

위 문장은 명령의 크기가 정의되지 않았습니다 라는 오류를 발생한다. 왜?

왜냐하면어셈블러가1을바이트로저장할지,워드로저장할지,더블워드로저장할지

모르기때문이다.이문제를해결하기위해선아래와같이해주면된다.

mov dword [L6], 1 ; L6 에 1 을 저장한다

이를통해어셈블러는 1을메모리상의 L6에서시작하는부분부터더블워드로저장

하라는것을알수있다.그외에도 BYTE, WORD, QWORD, TWORD7) 가있다.

1.3.6 입출력

입출력(I/O)은 매우 시스템 종속적인 작업들이다. 이는 컴퓨터 시스템의 하드웨

어와의 소통을 필요로 한다. C 와 같은 고급 언어에선 I/O 를 위한 표준 라이브러리

를제공한다. 어셈블리언어에서는 C 와같은라이브러리가제공되지않는다. 따라서,

어셈블리에선직접적으로하드웨어에접근하거나 (이는보호모드에서특별한권한을

가지는작업이기도한다)운영체제가제공하는저수준의루틴들을사용해야한다.

어셈블리루틴과함께 C언어를사용하는것은매우흔한일이다.왜냐하면어셈블

리 코드가 표준 C 입출력 라이브러리 루틴을 사용할 수 있기 때문이다. 그러나, C 가

사용하는루틴에게정보를전달하는데에는규칙이있다는점을알아야한다. 그규칙

들을여기서직접다루기에는복잡하다 (나중에다룰것이다).저자는입출력을단순화

7) TWORD는메모리상의10바이트를정의한다부동소수점보조프로세서가이데이타형식을사용한다.

print_int EAX 에저장된값을화면에출력한다.
print_char AL 에 저장된 아스키 코드 값에 대응 하는 문자를 화면에

표시한다
print_string EAX에저장된문자열의주소값에해당하는문자열을화면

에출력한다. 문자열은반드시 C 형식이어야한다(i.e. null
로종료).

print_nl 화면에개행문자를출력한다.
read_int 키보드로부터정수를입력받은다음 EAX레지스터에저장

한다.
read_char 키보드로부터한개의문자를입력받은후그문자의아스

키코드값을 EAX 레지스터에저장한다.

표 1.4: 어셈블리 I/O 루틴

하기위해복잡한 C규칙을숨기고훨씬간단한인터페이스를이용하는루틴을제공한

다.표 1.4에저자가개발한루틴들에대해설명하고있다.읽기루틴만을제외한모든

루틴들은모든레지스터의값을저장한다.이루틴들을이용하기위해서는이루틴들에

대한정보가기록된파일들을소스에포함시켜어셈블러가사용할수있게해야한다.

NASM 에서파일을포함하기위해선 %include전처리기지시어를사용하면된다.저

자가만든 I/O 루틴을 8) NASM 에포함시키기위해서는아래와같이하면된다.

%include "asm_io.inc"

위 I/O루틴중출력루틴을사용하고싶다면 EAX레지스터에적당한값을대입하고

CALL명령을사용해호출한다. CALL명령은고급언어에서의함수를호출하는것과동

일하다.이명령사용시코드의다른부분으로실행을분기했다가그루틴이종료된다면

다시 원래 호출했던 부분으로 돌아오게(리턴) 된다. 아래의 예제 프로그램은 이러한

I/O 루틴의호출을보여준다.

1.3.7 디버깅

저자의 라이브러리는 프로그램을 디버깅하는데 유용한 루틴들을 포함하고 있다.

이 디버깅 루틴은 컴퓨터의 상태를 변경하지 않고서 화면에 그대로 상태를 출력하게

해준다. 이 루틴들을 CPU 의 현 상태를 저장하고 서브 루틴을 호출하는 매크로 이다.

이매크로들은 asm_io.inc파일에정의되어있다.매크로들은보통명령들처럼사용

된다.이때매크로들의피연산자들은반점(,)을통해구분한다.

저자의라이브러리에는4개의디버깅루틴이있는데이는각각dump_regs,dump_mem,
dump_stack, dump_math이다.이들은레지스터,메모리,스택,그리고수학보조프로

세서에들어있는값들을모두보여준다

dump_regs 이매크로는레지스터에들어있는값들을화면에(stdout) 16진수로출력

한다.또한플래그레지스터의비트세트9) 를화면에출력한다.예를들어서제로

플래그가1이라면 ZF 가화면에출력된다. 0이라면화면에출력되지않는다.이

매크로는한개의정수를인자로가지는데이를통해서로다른 dump_regs명령

의출력들을구분할수있게한다.

8) asm_io.inc (그리고 asm_io.inc 가 필요로 하는 asm_io 파일) 은 이 책의 웹 페이지인
http://www.drpaulcarter.com/pcasm 에서다운로드가가능하다

9) Chapter 2 에서다룰것이다

dump_mem 이 매크로는 메모리의 일부 영역의 값들을 16 진수나 아스키 문자로 출

력한다. 이 매크로는 반점으로 구분되는 3 개의 매개 변수를 가진다. 첫 번째는

dump_regs 매크로 처럼 출력을 구분하기 위한 정수, 두 번째는 출력하기 위한

메모리의주소, 마지막은주소뒤에출력될 16바이트패러그래프의수이다. 요

청된메모리주소이전의패러그래프경계부터메모리값들이출력될것이다.

dump_stack 이매크로는 CPU 의스택에저장된값들을출력한다. (스택에대해서는

Chapter 4 에서 다룰 것이다) 스택은 더블워드로 구성되어 있으며 이 매크로를

통해스택에들어있는값들을출력할수있다. 3개의반점으로구분되는인자를

가지며 첫 번째는 dump_regs 에서 처럼 출력을 구분하기 위한 정수, 두 번째는

EBP레지스터에저장된주소아래부터출력할더블워드의수,세번째는 EBP레
지스터주소위로출력할더블워드의수이다.

dump_math 이는 수치 부프로세서의 레지스터에 저장되어 있는 값들을 출력한다.

dump_regs에서처럼결과를구분하기위한정수를매개변수로입력받는다.

1.4 프로그램만들기

요즘에는어셈블리어언어만을이용하여프로그램을만드는것은흔치않은일이

다.어셈블리는오직중요한일부루틴에서만이용하게된다.왜냐하면,어셈블러언어

보다고급언어로작성하는일이훨씬편하기때문이다.또한어셈블리언어를이용시

프로그램을다른플랫폼에프로그램을포팅하는작업이매우어렵게된다.사실상,어

셈블리어를사용하는일은드물다.

하지만우리는왜어셈블리언어를배워야하는가?

1. 때때로컴파일러가생성한코드보다어셈블리어로직접쓴코드가훨씬빠르고

작을수있다.

2. 어셈블리어를통해고급언어사용시힘들거나불가능한시스템하드웨어의직접

적인접근을할수있다.

3. 어셈블리언어를공부함을통해서컴퓨터의작동원리를깊게파악할수있다.

4. 어셈블리언어를공부함을통해컴파일러와 C 와같은고급언어가어떻게동작

하는지파악하는데도움이된다.

특히마지막두가지이유를보면우리는어셈블리로직접프로그래밍을하지않아도

많은도움을받을수있다는사실을알수있다.사실저자인나도어셈블리어만으로는

거의프로그램을짜지않지만어셈블리에서얻은아이디어를매일사용하고있다.

1.4.1 첫프로그램

우리가처음에작성할프로그램들은그림 1.6 와 같은단순한 C 드라이버프로그

램들이다. 이는단순히 asm_main함수만을호출한다. 이루틴은우리가나중에어셈

블리로 쓸 것이다. C 드라이버 루틴을 이용하면 몇 가지 장점들이 있다. 첫 번째로 C
시스템은보호모드에서프로그램이잘돌아가도록모든것을설정해준다. C 에의해

모든 세그먼트와 이에 대응 되는 세그먼트 레지스터들은 초기화 된다. 어셈블리에서

1 int main()
2 {

3 int ret_status ;
4 ret_status = asm_main();
5 return ret_status ;
6 }

그림 1.6: driver.c코드

도걱정할필요는없다. 두번째로어셈블리코드를사용하여 C 라이브러리를사용할

수 있다. 저자의 I/O 루틴이 이 사실을 이용한다. 이들은 C 의 I/O 함수를 사용한다 (

printf등등)아래에는간단한어셈블리프로그램을소개했다.

first.asm
1 ; file: first.asm
2 ; 최초의 어셈블리 프로그램. 이 프로그램은 2 개의 정수를 입력받아

3 ; 그 합을 출력한다.

4 ;
5 ; 실행 가능한 프로그램을 만들려면 DJGPP 를 이용해라 :
6 ; nasm -f coff first.asm
7 ; gcc -o first first.o driver.c asm_io.o
8

9 %include "asm_io.inc"
10 ;
11 ; 초기화 된 데이터는 .data 세그먼트에 들어간다.

12

13 segment .data
14 ;
15 ; 아래 라벨들은 출력을 위한 문자열들을 가리킨다.

16 ;
17 prompt1 db "Enter a number: ", 0 ; 널 종료 문자임을 잊지 말라!

18 prompt2 db "Enter another number: ", 0

19 outmsg1 db "You entered ", 0

20 outmsg2 db " and ", 0

21 outmsg3 db ", the sum of these is ", 0

22

23 ;
24 ; 초기화 되지 않은 데이터는 .bss 세그먼트에 들어간다.

25 ;
26 segment .bss
27 ;
28 ; 이 라벨들은 입력 값들을 저장하기 위한 더블워드를 가리킨다.

29 input1 resd 1

30 input2 resd 1

31

32 ;
33 ; 코드는 .text 세그먼트에 들어간다.

34 ;
35 segment .text
36 global _asm_main
37 _asm_main:
38 enter 0,0 ; 셋업(set up) 루틴

39 pusha
40

41 mov eax, prompt1 ; prompt 를 출력

42 call print_string
43

44 call read_int ; 정수를 읽는다.

45 mov [input1], eax ; input1 에 저장.

46

47 mov eax, prompt2 ; prompt 를 출력

48 call print_string
49

50 call read_int ; 정수를 읽는다.

51 mov [input2], eax ; input2 에 저장.

52

53 mov eax, [input1] ; eax = input1 에 위치한 dword
54 add eax, [input2] ; eax += input2 에 위치한 dword
55 mov ebx, eax ; ebx = eax
56

57 dump_regs 1 ; 레지스터의 값을 출력

58 dump_mem 2, outmsg1, 1 ; 메모리를 출력

59 ;
60 ; 아래 단계별로 메세지를 출력한다.

61 ;
62 mov eax, outmsg1
63 call print_string ; 첫 번째 메세지를 출력

64 mov eax, [input1]
65 call print_int ; input1 을 출력

66 mov eax, outmsg2
67 call print_string ; 두 번째 메세지를 출력

68 mov eax, [input2]
69 call print_int ; input2 를 출력

70 mov eax, outmsg3
71 call print_string ; 세 번째 메세지를 출력

72 mov eax, ebx
73 call print_int ; 합을 출력 (ebx)
74 call print_nl ; 개행문자를 출력

75

76 popa
77 mov eax, 0 ; C 로 리턴된다.

78 leave
79 ret first.asm

13행에서프로그램의데이터세그먼트(.data)에저장될메모리공간을정의하였

다. 이세그먼트에는오직초기화된데이터만이올수있다. 17 에서 21 행에서는몇

개의문자열들이선언되었다.이들은 C라이브러리를통해출력되기때문에반드시널

(NULL) 문자로끝마쳐져야한다 (아스키코드값이 0). 참고로 0 과 '0' 에는매우큰

차이가있다는사실을알아두라.

초기화되지않은데이터는반드시 bss 세그먼트(.bss) 에서선언되어야한다. 이

세그먼트는초기유닉스기반의어셈블러에서사용된“심볼을통해시작된블록(block
started by symbol).” 연산자의이름에서따온것이다.이것에대해선나중에다루겠다.

코드세그먼트는역사적으로 .text라불려왔다.이는명령어들이저장되는곳이다.

( 38행)의메인루틴의라벨에접두어 _ (under score)가사용되었음을유의하자.이는

C함수호출규약(C calling convention)의일부분이다.이규약은코드컴파일시에C가

사용하는규칙들을명시해준다. 이는어셈블리어와 C 를같이사용시에이호출규약

을아는것이매우중요하다.나중에이호출규약에대해속속살펴볼것이다.그러나,

지금은오직모든C심볼들(i.e.,함수와전역변수)에는C컴파일러에의해 _가맨앞에

붙는다는사실만알면된다. (이규칙은오직 DOS/Windows 에만해당하며리눅스 C
컴파일러에서는 C 심볼이름에아무것도붙이지않는다)

37행의 global지시어는어셈블러로하여금 _asm_main라벨을전역으로생성하

라고알려준다. C 에서와는달리라벨들은내부지역(internal scope)이 기본으로된다.

이뜻은오직같은모듈에서만그라벨을사용할수있다는뜻이다.하지만전역(global)
지시어를이용하면라벨을외부지역(external scope) 에서도사용할수있게된다.이러

한형식의라벨은프로그램의어떠한다른모듈에서도접근이가능하다. asm_io모듈

은 print_int등을모두전역으로선언한다. 이때문에우리가 first.asm모듈에서

이를사용할수있었다.

1.4.2 컴파일러의존성

위에나온어셈블리코드는오직무료 GNU10) 의 DJGPP C/C++ 컴파일러에서만

돌아간다11) 이컴파일러는인터넷을통해무료로다운로드할수있다.이컴파일러는

386 이상의 PC 를 필요로 하며, DOS 나 Windows 95/98, NT 등에서 잘 실행된다. 이

컴파일러는 COFF(Common Object File Forma) 형식의 목적 파일(Object File) 을 이

용한다. 이형식으로어셈블하기위해서는 nasm과함께 -f coff스위치를사용해야

한다. (위소스코드주석이잘나와있다)생성되는목적파일의확장자는 o가된다.

리눅스 C 컴파일러 또한 GNU 컴파일러이다. 위 코드를 리눅스에서 실행되게 바

꾸려면 단순히 37 에서 38 행의 _ 를 지워주기만 하면 된다. 리눅스에서 목적파일은

ELF(링크및실행가능한포맷, Excutable and Linkable Format) 형식을이용한다.리눅

스에선 -f elf스위치를이용하면된다.또한목적파일의확장자는 o이다 . 이컴파일러를위한예제파일

들은 저자의 홈페이지에서 다

운받을수있다.

10) 무료소프트웨어재단(Free Soware Foundation)
11) http://www.delorie.com/djgpp

볼랜드C/C++는또다른유명한컴파일러이다.이는목적파일을마이크로소프트의

OMF형식을이용한다.볼랜드사컴파일러를위해선 -f obj스위치를이용하면된다.

OMF 형식은다른목적파일포맷들과는달리다른세그먼트지시어를사용한다. 13

행의데이터세그먼트는아래와같이바뀌어야한다.목적파일의확장자는 obj이다.

segment _DATA public align=4 class=DATA use32

26 행의 bss 세그먼트는아래와같이바뀐다 :

segment _BSS public align=4 class=BSS use32

36행의텍스트세그먼트는아래와같이바뀐다.

segment _TEXT public align=1 class=CODE use32

뿐만아니라 36행에아래와같은새로운행이덧붙여져야한다.

group DGROUP _BSS _DATA

마이크로소프트 C/C++ 컴파일러는목적파일으로 OMF 나 Win32 형식을이용할

수있다. (OMF형식을이용한다해도그정보를내부적으로는Win32로바꾸어서처리

한다) Win32 형식을이용하면 DJGPP나리눅스에서처럼세그먼트를정의할수있다.

-f win32스위치를통해위모드로출력할수있다.목적파일의확장자는 obj가된다.

1.4.3 코드를어셈블하기

커맨드라인에아래와같이입력한다.

nasm -f object-format first.asm

이때, object-format 부분에는 coff , elf , obj, win32 중 하나가올수있으며사용할 C 컴

파일러에따라달라진다. (물론리눅스나볼랜드나에따라서도소스파일의내용이바

뀌어야함을잊지말아야한다)

1.4.4 C코드를컴파일하기

DJGPP 를사용한다면 :

gcc -c driver.c

-c스위치는단지컴파일만을하란뜻이다.아직링크는하지않는다.리눅스나볼랜드,

마이크로소프트컴파일러도동일한스위치를사용한다.

1.4.5 목적파일링크하기

링킹은기계어코드와목적파일과라이브러리파일에들어있는데이터를하나로

합쳐서실행가능한파일을만드는과정이다.아래에나타나있듯이이과정은복잡한다.

C코드는표준C라이브러리와특별한개시코드(startup code)를필요로한다.이는

C컴파일러가정확한매개변수와함께링커를호출하고직접적으로링커를호출하는

작업을훨씬쉽게하게해준다.예를들어서위프로그램을 DJGPP 으로링크할때,

gcc -o first driver.o first.o asm_io.o

를쓰면 first.exe (리눅스에선그냥 first) 라는실행가능한파일이만들어진다.

볼랜드컴파일러는

bcc32 first.obj driver.obj asm_io.obj

와같이하면된다.볼랜드는실행가능한파일의이름을정할때나열된첫번째파일의

이름을따오게된다.따라서,위경우프로그램의이름은 first.exe가된다.

컴파일과링크과정을하나로같이묶어서할수도있다.예를들어서

gcc -o first driver.c first.o asm_io.o

그러면 gcc는 driver.c를컴파일한후링크할것이다.

1.4.6 어셈블리리스트파일이해

-l listing-file스위치는 nasm으로하여금주어진이름의리스트파일을생성

하라고명령한다.이파일은코드가어떻게어셈블되었는지보여준다.아래 17에서18

행 (데이타세그먼트)이리스트파일에어떻게나타났는지보여준다. (리스트파일의

행번호를잘보면실제소스파일의행번호와다름을알수있다)

48 00000000 456E7465722061206E- prompt1 db "Enter a number: ", 0

49 00000009 756D6265723A2000
50 00000011 456E74657220616E6F- prompt2 db "Enter another number: ", 0

51 0000001A 74686572206E756D62-
52 00000023 65723A2000

첫번재열은각행의번호를표시하고두번째열은세그먼트의데이타의오프셋을16

진수로나타낸다.세번째열은저장될 16진수값을보여준다.위경우그값은아스키

코드값과대응이된다. 마지막으로소스파일의코드가나타나게된다. 두 번째열에

나타나는오프셋은실제프로그램에서의데이터의오프셋과다를가능성이매우높다.

각각의모듈은데이터세그먼트(를포함한다른세그먼트들에서도)에서의자신만의

라벨을가지고있다.링크과정에선 ( 1.4.5참조)에선모든데이터세그먼트의라벨들

의정의가하나의데이터세그먼트로통합된다.그리고마지막으로정리된오프셋들은

링커에의해계산된다.

아래 소스코드의 54 에서 56 행의 텍스트 세그먼트의 리스트 파일에서 해당되는

부분을보여주고있다.

94 0000002C A1[00000000] mov eax, [input1]
95 00000031 0305[04000000] add eax, [input2]
96 00000037 89C3 mov ebx, eax

3 번째열은어셈블리에의해만들어진기계어코드를보여준다. 보통명령어들의완

전한 코드는 아직 계산되지 못한다. 예를 들어서 94 행에서 input1 의 오프셋(혹은

주소) 은코드가링크되기전까지는확실히알수없다. 어셈블러는 mov명령의연산

부호(Opcode) 는 계산할수있지만 (이는 A1 에 나타남) 대괄호안의정확한값은알

수 없다. 위 경우 어셈블러는 위 파일에서 정의된 bss 세그먼트의 시작을 임시적으로

오프셋 0 으로하였다. 이것이프로그램의마지막 bss 세그먼트의시작을의미한다는

것이아니라는사실을명심하여라.코드가링크가된다면링커는정확한오프셋값을

그 자리에 집어 넣을 것이다. 다른 명령들의 경우, 예를 들어 96 행을 본다면 어떠한

라벨도참조하지않음을알수있다.여기에선어셈블러가정확한기계어코드를계산할

수있다.

빅,리틀엔디안표현

95 행을 자세히 본다면 기계어 코드의 대괄호 속에 있는 오프셋 값이 매우 이상

하다는 것을 알 수 있다. input2 라벨은 오프셋 4 (파일에 정의되어 있는 대로) 이다.

그러나, 메모리에서의오프셋은 00000004 가아니라 04000000 이다! 왜냐하면프로

세서가정수를메모리상에다른순서로저장하기때문이다.정수를저장하는방법에는

대표적으로 2 가지 방법이 있다. 빅 엔디안(Big endian) 과 리틀 엔디안(Little endian)
이 그것이다. 빅 엔디안은 우리가 생각하는 그대로 정수를 저장한다. 가장 큰 바이트

(i.e. 가장 상위 바이트) 가 먼저 저장되고 가장 하위 바이트가 나중에 저장된다. 예를

들어서 dword 00000004는 4개의바이트00 00 00 04로저장된다. IBM메인프레임,

대부분의 RISC 프로세서,그리고모토롤라프로세서들은빅엔디안방법을이용한다.

그러나 인텔 기반의 프로세서 는 모두 리틀 엔디안 방식을 이용한다. 여기서는 하위

바이트가먼저저장된다.따라서00000004는메모리상에04 00 00 00으로저장된다.

이형식은 CPU 에따라정해져있어서바뀔수없다. 보통프로그래머는어떤방식을

사용하느냐를구별한필요가없다.그러나이사실을아는것이중요한이유가몇가지

있는데

1. 이진데이터가서로다른컴퓨터로전송될때 (파일또는네트워크를통해서 ).

2. 멀티바이트(multibyte) 정수데이터가메모리에쓰이고개개의바이트로읽어질

때(역도같음)

엔디안표현은배열의원소들의순서에대해서는영향을주지않는다.배열의첫번

째원소는언제나최하위의주소를갖는다.이는문자열(단지문자들의배열인)에서도

해당된다.엔디안표현은단지배열의개개의원소들에대해서만영향을줄뿐이다.

1.5 뼈대파일

그림 1.7 은뼈대파일(Skeleton ĕle) 을보여준다. 어셈블리프로그램코딩시매번

같은내용을작성하기보단위코드를기준으로시작하면편리하다.

skel.asm
1 %include "asm_io.inc"
2 segment .data
3 ;
4 ; 초기화된 데이터는 여기 데이타 세그먼트에 들어간다.

5 ;
6

7 segment .bss
8 ;
9 ; 초기화 되지 않은 데이터는 여기 bss 세그먼트에 들어간다.

10 ;
11

12 segment .text
13 global _asm_main
14 _asm_main:
15 enter 0,0 ; 셋업 루틴

16 pusha
17

18 ;
19 ; 코드는 텍스트 세그먼트에 들어간다. 이 주석 앞 뒤의 코드를

20 ; 수정하지 마세요

21 ;
22

23 popa
24 mov eax, 0 ; C 로 리턴된다.

25 leave
26 ret skel.asm

그림 1.7: 뼈대프로그램

어셈블리언어의기초

2.1 정수들다루기

2.1.1 정수의표현

정수는두가지종류로나눌수있다 : 부호없는정수(unsigned) 와부호있는정수

(signed) 이다.부호없는정수 (즉,음이아닌)는이진수로직관적으로표현된다.예를

들어 200은 1바이트부호없는정수로, 11001000 (또는 16진수로 C8) 로나타난다.

부호있는정수 (이는양수또는음수가될수있다) 는좀더복잡한방법으로표현

된다.예를들어,−56과+56은바이트로똑같이00111000으로표현될것이다.종이

에다가는우리가−56을−111000이라쓸수있지만컴퓨터메모리상에선어떻게할

것인가?어떻게마이너스기호가저장될수있는가?

컴퓨터메모리에부호있는정수를나타내는방법은대표적으로 3가지가있다.이

모든방법들은공통적으로정수의최상위비트를부호비트(sign bit) 로 사용한다. 이

비트는만약이정수가양수이면 0, 음수이면 1의값을가진다.

부호있는크기표현법

첫 번째 방법은 가장 단순한 방법인데, 부호 있는 크기(signed magnitude) 표현법

이라부른다.이는정수를두부분으로나누는데,첫번째부분은부호비트이고나머지

부분그정수의절대값(크기)을나타낸다.따라서56은바이트로 00111000으로나타

내지고 (부호비트에밑줄쳐짐)−56 은바이트로 110111000 이된다.가장큰바이트

값은 01111111, 혹은 +127 이 되고 가장 작은 바이트 값은 11111111 혹은 −127 이

된다. 어떠한값의부호를바꾸려면단순히부호비트의값만바꾸어주면된다. 이표

현법은상당히직관적이지만몇가지단점이있다. 0은양수도,음수도아니기때문에

0을두가지로표현할수있게된다.+0(00000000) 과−0(10000000) 이다.이때문에

CPU 의산술연산로직이복잡하게된다. 두번째로보통의산술연산로직도복잡해

지는데 예를 들어 10 에 −56 이 더해진다면 이는 다시 10 에서 56 을 빼었다고 생각

되어야하기때문이다. 결과적으로이표현법을사용하면 CPU 의로직이복잡해지게

된다.

1의보수표현법

두번째방법은 1의보수(one’s complement) 표현법이다. 어떠한수의 1의보수는

단순히각각의비트를반전(1을0, 0을1로)시키면얻을수있다. (이는새비트의값을

(1−이전의비트)로하는것으로도생각할수있다.) 예를들어 00111000 (+56) 의 1

의보수는 11000111 이된다. 1 의 보수표현법이서, 어떤수의 1 의보수를계산하는

것은단순히 -를붙이는것과같다.따라서 11000111은−56을표현한것과같다. 1의

보수표현에서부호비트의값이자연스럽게바뀌었다.첫번째방법에서처럼 0을나

타내는방법은두가지가있는데 00000000 (+0)과 11111111 (−0)이된다. 16진수를

이진수로바꾸지않고도보수를계산하는방법은단순히각자리를 F (또는십진수로

15)에서빼는것이다.예를들어+56의경우16진수로38로나타나는데1의보수를

찾기위해각자리를 F 에서빼면 C7 이된다.이는위에서나타난결과와일치한다.

2의보수표현법

위두방법들은모두초기의컴퓨터에서만사용되었던방법들이다.현대의컴퓨터는

이세번째방법을사용하는데이를 2의보수표현법이라한다.어떤수의 2의보수는

아래두단계를통해찾을수있다 :

1. 어떤수의 1의보수를계산한다.

2. 위단계에서나온결과에 1을더한다.

아래는 00111000(56)을이용한방법이다.먼저,56의1의보수를계산하면 : 11000111.
그리고 1을더하면 :

11000111
+ 1

11001000

2 의보수표현법에서 2의보수를계산하는것은숫자에 - 를붙이는것과같다.따

라서 11001000 은−56 를 2의보수표현법으로나타낸것이다.어떤수에음을두번

취하면원래수로돌아와야하는데놀랍게도2의보수표현법에서는이를잘만족한다.

한번 11001000 의 2의보수표현을취해보자.

00110111
+ 1

00111000

2의보수표현에서1을더할때,이덧셈에서최상위비트에서캐리(carry)가있을

것이다. 이캐리는쓰이지않는다. 컴퓨터의모든데이터들은정해진크기(비트의수)

로보관된다. 두개의바이트를더하면언제나결과는바이트로나온다. (또한워드에

워드를더하면워드가나오고,다른것들도매한가지)이는 2의보수표현에서가장중

요한부분이다.예를들어서1바이트의0을생각하자. (00000000).이것의2의보수를

계산하면

11111111
+ 1
c 00000000

이 때, c 는 캐리(carry)를 나타낸다. (이 장 뒷부분에서, 결과에 저장되지 않는 올

림이발생되었음을어떻게알아채는지에대해이야기하겠다.) 따라서 0 을나타내는

2의보수표현은오직한가지방법밖에없다.이때문에 2의보수표현을이용한산술

연산을이전의방법을이용하였을때보다훨씬쉽게수행할수있게된다.

수 16진수표현 n
0 00
1 01

127 7F
-128 80
-127 81
-2 FE
-1 FF

표 2.1: 2의보수표현

2 의 보수 표현을 이용하여 한 개의 바이트를 통해 −128 에서 +127 까지 나타낼

수있게되었다. 표 2.1 는 2의보수표현으로나타낸몇개의수를보여준다. 만약 16

비트가 이용된다면 −32, 768 에서 +32, 767 까지 표현할 수 있게 된다. +32, 767 은

7FFF로나타나고,−32, 768은8000으로, -128은 FF80, -1은 FFFF로나타나게된다.

32 비트를이용한다면대략−20 억에서 +20 억의수를모두나타낼수있게된다.

CPU 는어떤바이트 (혹은워드나더블워드)가어떠한형식으로표현되어있는지

알수없다. 어셈블리어는고급언어가가지고 있는형(type) 에 대한개념이없다. 데

이터가어떻게생각되느냐는그데이터에어떠한명령이사용되는지에달려있다. 16

진수값 FF가−1로생각되든지 (부호있는경우)아니면+255로생각되든지는 (부호

없는경우) 오직프로그래머에달려있다. 반면에 C 언어는부호있는정수(signed)와
부호 없는 정수(unsigned)를 각각 정의한다. 이는 C 컴파일러로 하여금 그 데이터에

정확한명령들을내리게해준다.

2.1.2 부호확장

어셈블리어에서모든데이터는정해진크기가있다.다른데이터와사용하기위해

데이터의크기를줄이는작업을자주하게된다.크기를감소시키는것이가장쉽다.

데이터의크기줄이기

데이터의크기를줄이기위해단순히데이터의가장자리비트를줄이면된다.아래

는그예이다.

mov ax, 0034h ; ax = 52 (16 비트로 저장됨)

mov cl, al ; cl = ax 의 하위 8 바이트

당연하게도,크기를줄였는데수의값이정확하게표현이되지않으면크기를줄이

는방법은소용이없을것이다. 예를들어서 AX에 0134h (혹은 10 진수로 308, 이 때

끝에 붙은 h 는 이 수가 16 진수 (hex) 로 나타내져 있음을 알려준다.) 이 있었는데 위

코드를실행하면 CL에 34h 가된다. 이방법은부호있는정수와부호없는정수모두

사용가능하다.부호있는정수를고려해보면 AX에 FFFFh (워드로−1)가있었다면 CL
은 FFh (바이트로−1) 이된다. 그러나, AX의값이부호가없는경우라면맞지않음을

유의해라.

부호없는정수의경우변환이올바르게일어나기위해선제거되는비트들이반드시

0이어야한다.반면에부호있는정수들의경우제거되는비트들은모두0이거나모두

1이어야한다.또한,제거되지않은첫번째비트는반드시제거된비트들과같은값을

가져야한다. 이 비트는나중에작아진값의새로운부호비트가될것이다. 이 비트가

이전의부호비트와같아야한다는점이중요하다.

데이터의크기늘리기

데이터의 크기를 늘리는 작업은 줄이는 작업보다 좀 더 복잡하다. 예를 들어서 1

바이트 FF 를 생각해보자. 만약 이것이 워드로 늘 려진다면, 그 워드가 가질 수 있는

값은 무엇이 있을까? 이는 FF 가 어떻게 해석되기에 따라 달라진다. 만약 FF 가 부호

없는정수였다면 (십진수로255), 워드로확장시,그값은00FF가된다.그러나,만약

이 FF가부호있는정수 (십진수로−1) 였다면워드로확장시 FFFF가되어야만한다.

보통 부호 없는 정수를 확장한다면 확장된 새 비트가 모두 0 이 된다. 따라서 FF
는 00FF 가된다.그러나,부호있는정수를확장하려면반드시부호비트를확장하게

된다. 이 말은 새로운 비트가 이전의 부호 비트의 값과 같다는 뜻이다. 따라서, FF 의

부호 비트의 값이 1 이였으므로 새롭게 확장된 수에 추가된 비트들은 모두 1 이 되어

FFFF가된다.만약부호있는정수의값이5A (십진수로90)이였다면확장시005A가

될것이다.

80386은몇가지수를확장하는명령들을제공한다.컴퓨터는그수가부호가있는

정수인지, 부호없는정수인지알수없다는사실을명심하라. 이모든것은프로그래

머가정확한명령을사용하는데달려있다.

부호 없는 정수들에 한에서 MOV 연산을 통해 상위 비트에 0 을 집어 넣을 수 있다.

예를들어서바이트 AL 을워드 AX 로확장시키기위해선 :

mov ah, 0 ; 상위 8 비트들을 모두 0 으로 만듦

그러나 MOV명령을통해AX에저장된부호없는워드를부호없는더블워드인 EAX
에저장할순없다.왜냐하면, EAX의상위16비트를가리키는말이 MOV명령에선없기

때문이다. 80386 은 이 문제를 MOVZX 명령을 추가함으로써 해결하였다. 이 명령은 2

개의 피연산자를 가진다. 값이 들어가는 피연산자(데스티네이션, destination) 는 첫

번째 피연산자로, 16 비트나 32 비트 레지스터여야 한다. 값을 주는 피연산자(소스,

source) 는 두 번째 피연산자로 8 또는 16 비트 레지스터, 혹은 메모리상의 바이트나

워드이어야만한다. 다른조건으로는데스티네이션이반드시소스의크기보다커야

한다는점이다. (대부분의명령들은모두소스의크기와데스티네이션의크기가같아야

한다)아래그예가있다 :

movzx eax, ax ; ax 를 eax 로 확장

movzx eax, al ; al 을 eax 로 확장

movzx ax, al ; al 을 ax 로 확장

movzx ebx, ax ; ax 를 ebx 로 확장

부호있는정수들의경우 MOV연산을적용하리란쉽지않다. 8086 은부호있는정

수들의확장을위해몇가지명령들을지원하였다. CBW (바이트를워드로변환)명령은

AL레지스터를AX로확장해준다.이때,피연산자는쓸필요가없다. CWD (워드를더블

워드로 변환) 명령은 AX 를 DX:AX 로 확장해 준다. 이 때, DX:AX 의 뜻은 DX 와 AX
레지스터를묶어서하나의32 비트레지스터로생각하란뜻이다.이때상위16비트는

DX레지스터가,하위 16비트는 AX레지스터가갖게된다. (8086은 32비트레지스터

가없었음을상기하자!) 80386 에서는몇가지새로운명령들을사용할수있다. CWDE

1 unsigned char uchar = 0xFF;
2 signed char schar = 0xFF;
3 int a = ( int) uchar; /* a = 255 (0x000000FF) */
4 int b = ( int) schar ; /* b = −1 (0xFFFFFFFF) */

그림 2.1:

char ch;
while( (ch = fgetc (fp)) != EOF ) {

/* ch로무언가작업을한다 */
}

그림 2.2:

(워드를 더블워드 확장으로 변환) 명령은 AX 를 EAX 로 확장한다. CDQ 명령은 EAX
를 EDX:EAX ( 64 비트!) 로확장해준다. 마지막으로 MOVSX명령은부호있는정수에

대해서 MOVZX와똑같은작업을한다.

C프로그래밍에응용하기

C 에서도부호있는/없는정수들의확장이사용된다. C 의변수들은부호가있다/ANSI C는 char가부호가있는

지없는지정의하지않고개개

의컴파일러에게정하라고한

다.이는타입이그림 2.1에서

명확히정의되는이유이다.

없다로정의된다. (int는부호가있다). 그림 2.1 의 코드를생각해보자. 행 3 에서

변수 a는부호없는정수들을위한확장규칙에따라(MOVZX를사용하여)확장되었지만

행 4를보면부호있는정수들을위한확장규칙에따라 b가확장되었음을알수있다.

(MOVSX를사용하여)

이주제와관련하여대표적인 C프로그래밍버그가있다.그림 2.2 의코드를보자.

fgetc()함수의원형은 :

int fgetc( FILE * );

이다.아마여러분은 이함수가문자를읽어들이는데왜 int를반환하는지궁금해할

것이다.사실이함수는 char형태로반환한다. (다만, int로앞에 0이붙는방법으로

확장되어서)그러나 char형태로반환할수없는것이딱하나있는데바로 EOF이다.이

매크로는보통−1 로정의된다. 따라서, fgetc()는 char이 int형태로확장된값을

반환하거나 (이는 000000xx 와 같은 형태로 16 진수로 나타날 것이다) EOF (이는 16

진수로 FFFFFFFF로나타난다)를반환할것이다.

그림 2.2 의 기본적인 문제는 fgetc() 가 int 를 반환하지만 그 값은 char 에 저

장되어있다는것이다. C는 char에값을맞추기위해 int값의상위비트들을잘라낼

것이다.문제는 (16진수로) 000000FF와 FFFFFFFF모두 FF로잘려진다는것이다.따

라서 while 루프는파일의바이트 FF를읽어들이고있는지, 아니면파일의마지막을

읽어들이고있는지알수없게된다.

위상황에서코드가어떤일을하느냐는 char이부호가있느냐,없느냐에달려있다.

왜냐하면 2행에서 ch는 EOF와비교되기때문이다. EOF가 int의값을가지고있으

므로 1) ch의값은 int로확장되어두값이같은크기로비교될수있게한다 2) 그림

1) 파일의끝에 EOF 문자가있다는것은흔한잘못된생각이다.이는사실이아니다
2) 이에대한이유는나중에다루도록하자

2.1 에서나타나듯이,변수에부호가있는지없는지에대한유무는매우중요하다.

만약char이부호가없다면FF는 000000FF로확장될것이다.이는EOF (FFFFFFFF)
와비교되어값이다르다고나타날것이다.따라서,루프에서절대로빠져나갈수없다!

만약 char부호가있다면 FF는 FFFFFFFF로확장될것이다. 이는 EOF 문자를입

력된다면성공적으로루프를빠져나갈수있게된다. 그러나, FF는파일의중간에서

입력될수도있다.따라서,루프가파일을다읽어들이지못한채끝날수도있다.

위문제에대한해결책은 ch변수를 char이아닌 int로선언하는것이다.이렇게

된다면 2 행에서 값의 확장이나 잘리는 일이 없게 된다. 루프 내부에선 값을 잘라도

아무런문제가없는데왜냐하면 ch가반드시단순한바이트로행동하기때문이다.

2.1.3 2의보수의산술연산

이미 앞에서 보았듯이 add 명령은 덧셈을 수행하고 sub 명령은 뺄셈을 수행한다.

이 두개의명령들은플래그레지스터의 2 개의비트를변경하는데이비트들은각각

오버플로우(overĘow) 와캐리플래그(carry Ęag) 이다.오버플로우플래그는부호있는

정수연산시,결과값이너무나커서데스티네이션에들어갈수없을때세트(set) 된다.

캐리플래그는부호없는정수연산시,덧셈이나뺄셈에서받아올림이나받아내림이

최상위/하위 (MSB, Most Signiĕcant Bit) 비트에있을때세트된다.부호있는연산에서

의캐리플래그의역할은조금있다가살펴볼것이다. 2의보수의가장큰장점은부호

없는정수에대해서덧셈과뺄셈이정확히일치한다는것이다. 따라서 add와 sub는
부호있는/없는정수들에게서사용될수있다.

002C 44
+ FFFF + (−1)

002B 43

여기에선캐리가발생하지만결과엔포함되지않는다.

곱셈과 나눗셈 연산에는 각각 2 개의 명령들이 있다. 첫째로, 곱셈에선 MUL 이나

IMUL연산을사용할수있다. MUL연산은두개의부호없는정수를곱할때사용하고

IMUL연산은부호있는정수를곱할때사용한다.왜두개의다른명령들이필요하냐면,

곱셈을하는방법이부호없는정수와2의보수의부호있는정수들과다르기때문이다.

어떻게다르냐면, 바이트 FF 를제곱해워드로결과를얻어보자. 일단, 부호없는경우

255곱하기 255로그값은65025 (혹은16진수로 FE01)이된다.만약부호있는곱셈

을하였다면−1 곱하기−1 가되어 1 (혹은 16진수로 0001) 이된다.

곱셈명령에는여러형태가있다.가장오래된형태는아래와같다.

mul source

source 는레지스터나메모리의한값이될수있다. 이 값은즉시값 (immediate value)
이 될 수 없다. 피연산자(source)의 크기에 따라 곱셈 연산이 어떻게 실행될 지는 달

라진다. 만약 피연산자가 바이트 크기라면 이는 AL 레지스터의 바이트와 곱해져 그

값은 16 비트 AX 에 저장된다. 만약 16 비트라면 이는 AX 의 워드와 곱해져 32 비트

결과값으로 DX:AX 에 저장된다. 만약 소스가 32 비트라면 그 값은 EAX 에 곱해져서

64비트결과값으로 EDX:EAX 에저장된다.

IMUL명령은 MUL명령과형식이같지만몇개의다른명령형식들을추가하였다. 2

또한 3개의피연산자를가질수있는데 :

destination source1 source2 작업
reg/mem8 AX = AL*source1
reg/mem16 DX:AX = AX*source1
reg/mem32 EDX:EAX = EAX*source1

reg16 reg/mem16 dest *= source1
reg32 reg/mem32 dest *= source1
reg16 immed8 dest *= immed8
reg32 immed8 dest *= immed8
reg16 immed16 dest *= immed16
reg32 immed32 dest *= immed32
reg16 reg/mem16 immed8 dest = source1*source2
reg32 reg/mem32 immed8 dest = source1*source2
reg16 reg/mem16 immed16 dest = source1*source2
reg32 reg/mem32 immed32 dest = source1*source2

표 2.2: imul명령들

imul dest, source1
imul dest, source1, source2

표 2.2 이가능한모든조합을보여준다.

나눗셈명령으로는 DIV와 IDIV로 2개가있다. 이들은부호없는/있는정수들의

나눗셈을각각수행한다.통상적인형식은 :

div source

만약소스가8비트라면AX는피연산자로나뉘어진다.그몫은AL에저장되고나머지

는AH에저장된다.만약소스가16비트라면DX:AX는소스에의해나누어지고몫은

AX에,나머지는DX에저장된다.만약소스가32비트라면 EDX:EAX는소스로나누어

지고몫이 EAX 에, 나머지가 EDX 에저장된다. IDIV명령도같은방식으로작동한다.

IDIV의경우 IMUL과는달리특별한명령들이추가된것은없다.만약몫이레지스터에

저장되기에너무크거나, 제수가 0 이라면프로그램은인터럽트되고종료된다. 가장

흔한애러는 DX와 EDX 를나눗셈하기전에초기화하지않는것이다.

NEG 명령은 한 개의 피연산자를 가지며 피연산자의 2 의 보수를 계산함으로써 부

호를 바꾸어 버린다. 이 때 피연산자는 8비트, 16비트, 32비트 레지스터거나 메모리

값이여야한다.

2.1.4 예제프로그램

math.asm
1 %include "asm_io.inc"
2 segment .data ; 출력되는 문자열

3 prompt db "Enter a number: ", 0

4 square_msg db "Square of input is ", 0

5 cube_msg db "Cube of input is ", 0

6 cube25_msg db "Cube of input times 25 is ", 0

7 quot_msg db "Quotient of cube/100 is ", 0

8 rem_msg db "Remainder of cube/100 is ", 0

9 neg_msg db "The negation of the remainder is ", 0

10

11 segment .bss
12 input resd 1

13

14 segment .text
15 global _asm_main
16 _asm_main:
17 enter 0,0 ; 셋업 루틴

18 pusha
19

20 mov eax, prompt
21 call print_string
22

23 call read_int
24 mov [input], eax
25

26 imul eax ; edx:eax = eax * eax
27 mov ebx, eax ; 그 결과를 ebx 에 저장

28 mov eax, square_msg
29 call print_string
30 mov eax, ebx
31 call print_int
32 call print_nl
33

34 mov ebx, eax
35 imul ebx, [input] ; ebx *= [input]
36 mov eax, cube_msg
37 call print_string
38 mov eax, ebx
39 call print_int
40 call print_nl
41

42 imul ecx, ebx, 25 ; ecx = ebx*25
43 mov eax, cube25_msg
44 call print_string
45 mov eax, ecx
46 call print_int
47 call print_nl
48

49 mov eax, ebx
50 cdq ; 부호 확장을 위해 edx 를 초기화

51 mov ecx, 100 ; 즉시값으로 나눌 수 없다.

52 idiv ecx ; edx:eax / ecx
53 mov ecx, eax ; 몫을 ecx 에 저장

54 mov eax, quot_msg
55 call print_string
56 mov eax, ecx
57 call print_int
58 call print_nl
59 mov eax, rem_msg
60 call print_string
61 mov eax, edx
62 call print_int
63 call print_nl
64

65 neg edx ; 나머지에 음을 취한다

66 mov eax, neg_msg
67 call print_string
68 mov eax, edx
69 call print_int
70 call print_nl
71

72 popa
73 mov eax, 0 ; C 로 리턴

74 leave
75 ret math.asm

2.1.5 확장정밀도산술연산

어셈블리어는또한더블워드보다큰수들을위한덧/뺄셈명령을지원한다.이명령

들을캐리플래그를이용한다.위에서설명했던대로 ADD와 SUB명령은모두만약받아

올림/내림 이 발생했을 경우 캐리 플래그의 값을 변경한다. 이 캐리 플래그에 저장된

정보를통해큰수를작은더블워드조각(혹은더작게)들로쪼개에서연산을수행해

덧/뺄셈을수행할수있게한다.

ADC와 SBB명령은이캐리플래그의저장된정보를이용한다. ADC명령은아래의

작업을수행한다 :

operand1 = operand1 + carry flag + operand2

SBB명령은아래의작업을수행한다 :

operand1 = operand1 - carry flag - operand2

이들이어떻게사용될까? EDX:EAX와 EBX:ECX레지스터에저장된64비트정수들의

합을계산한다고하자.아래의코드는그합을 EDX:EAX 레지스터에넣을것이다.

1 add eax, ecx ; 하위 32 비트를 더한다

2 adc edx, ebx ; 상위 32 비트와 이전의 덧셈에서의 캐리를 더한다.

뺄셈도매우비슷하다.아래의코드는 EBX:ECX 를 EDX:EAX 에서뺀다.

KJH
강조

KJH
강조

KJH
강조

KJH
강조



제어구조 33

1 sub eax, ecx ; 하위 32 비트를 뺀다.

2 sbb edx, ebx ; 상위 32 비트를 빼고 받아 내림을 뺀다.

정말로큰수들의경우, 루프를사용할수있다. ( 2.2 참조) 덧셈루프에서, 되풀이

되는부분마다 ADC명령을사용한다면편리할것이다.그대신첫번째루프에는 ADD를
이용해야되는데이는 CLC (캐리클리어, Clear Carry) 명령을통해통해루프시작하기

직전에캐리플래그를0으로만들어버림으로써시행될수있다.만약캐리플래그가0

이라면 ADD와 ADC명령사이에는아무런차이가없게된다.같은방식으로뺄셈도할

수있다.

2.2 제어구조

고급언어들은실행의흐름을제어하기위해높은수준의제어구조 (e.g. if 문이나

while문)를지원한다.어셈블리는그와같은높은수준의제어구조를제공하진않는다.

그대신에악명높은 goto 와잘못사용시난잡한코드를만들수있는것들을지원한다.

그러나어셈블리를통해서도잘짜인프로그램을만드는것은가능하다.이를위해먼

저친숙한고급언어를이용하여프로그램의구조를디자인한뒤에동일한어셈블리로

바꾸어주면된다. (이는컴파일러가하는일과비슷하다)

2.2.1 비교

제어구조는데이터의비교를통해무엇을할지결정한다.어셈블리어에서그비교

의결과를나중에사용하기위해플래그레지스터에저장된다. 80x86 은비교를위해

CMP명령을지원한다.플래그레지스터는 CMP연산의두개의피연산자의차이에의해

설정된다. 피연산자가서로뺄셈을수행하고그결과에따라플래그레지스터가설정

되는데그결과는플래그레지스터의정해진부분에저장된다.만약뺄셈의결과를얻고

싶다면 CMP명령대신 SUB 을수행해야한다.

부호없는정수들의경우 2 개의플래그(플래그레지스터의개개의비트)가중요

하다 : 제로(ZF) 와 캐리(CF) 플래그 이다. 제로 플래그는 두 피연산자의 차이가 0 일

때세트3) 된다.캐리플래그는뺄셈에서의받아내림플래그와같이사용된다.아래와

같은비교문을고려하자.

cmp vleft, vright

vleft - vright의차이가계산되어이에따라플래그들이세팅된다.만약 CMP연산

결과가 0 이라면 vleft = vright, ZF 플래그가 세트(i.e. 1) 되고 CF 는 언세트(받

아내림 없음) 된다. 만일 vleft > vright 이라면, ZF 는 언세트 되고 CF 도 언세트

(받아내림없음) 된다. 만약 vleft < vright인경우 ZF 는언세트되고, CF 는세트

(받아내림있음)된다.

부호있는정수들의경우, 3개의플래그들이중요한역할을한다 :제로 (ZF) 플래그,

오버플로우(OF) 플래그, 부호 (SF) 플래그이다. 오버플로우플래그는연산의결과에 왜 vleft > vright 이면

SF = OF 일까? 만약 오버플로

우가없다면그차이는양수이

고정확한값을가질것이다.따

라서 SF = OF = 0 이다. 그러

나오버플로우가있다면그차

이는정확한값이아닐것이다.

(그리고 사실 음수이다.) 따라

서, SF = OF = 1이된다.

오버플로우(또는 언더플로우(underĘow) 발생시) 세트 된다. 부호 플래그는 연산의

결과가음수이면세트된다. 만약 vleft = vright라면 ZF 가세트된다. (부호없는

3) 역자주)여기서세트된다는의미는1로만든다는것이다.뒤에나타날언세트된다는의미는0으로
만든다는것이다

정수들의경우와마찬가지)만약 vleft > vright라면 ZF가언세트되고 SF = OF가
된다.만약 vleft < viright라면 ZF 는언세트되고 SF 6= OF 이다.

CMP 만이 아닌 다른 연산들 역시 플래그 레지스터의 값을 바꿀 수 있다는 사실을

명심하라.

2.2.2 분기명령

분기명령(Branch instruction) 은프로그램의임의의부분으로실행의흐름을옮길

수있다.다시말해,이들은 goto문과같다. 2가지종류의분기명령들이있는데 :무조건

분기 명령과 조건 분기 명령이다. 무조건 분기 명려은 goto 와 같이 반드시 분기를 한

다.조건분기명령은플래그레지스터의플래그의상태에따라서분기를할지안할지

결정한다.만약조건분기명령에서분기가일어나지않으면그다음명령이실행되게

된다.

JMP (jump 의준말) 명려은무조건분기명령을실행한다. 이는하나의인자를가

지며보통분기할코드의라벨(code label)을가진다. 어셈블러, 혹은링커는이라벨을

정확한주소로대체하게된다. 이는어셈블러가프로그래밍을편리하게해주는부분

이기도한다.한가지명심할점은 JMP명령의바로다음명령으로분기하지않는다면

그명령은수행되지않는다는것이다.

분기명령에몇가지형태들이있다.

SHORT 이는매우한정된범위에서분기를한다. 이는메모리상의최대 128 바이트

까지 분기 할 수 있다. 이 분기에 장점은 다른 것들에 비해 메모리를 적게 차지

한다는점이다.이는점프할변위(displacement)를저장하기위해 1부호있는바

이트를사용한다. 이변위는얼마나많은바이트를분기할지나타낸다. (EIP 에

이변위가더해진다) short 분기를명확히지정하기위해선 SHORT키워드를 JMP
명령에서의라벨바로앞에써주면된다.

NEAR 이분기는무조건/조건분기의기본으로지정된형태로,세그먼트의어떠한부

분으로도점프를할수있다.사실, 80386은 near분기의두가지형태를지원한다.

첫번째는변위를위해2바이트를사용한다.이는대략32,000바이트위,아래로

분기할수있게한다.다른형태는변위를위해4바이트를사용하는데이를통해

코드세그먼트의어떠한부분으로도분기할수있게해준다. 386보호모드에선

4 바이트유형이기본값으로설정되어있다. 만약 2 바이트만사용하는것으로

하려면 JMP명령에서의라벨바로앞에 WORD를써주면된다.

FAR 이 분기는 다른 코드 세그먼트로의 분기를 허용한다. 이를 이용하는 것은 386

보호모드에선매우드문일이다.

올바른코드라벨들은데이터라벨들과같은규칙이적용된다.코드라벨은코드세

그먼트에서그들이가리키는(label)문장바로앞에놓임으로써정의된다.라벨뒤에는

콜론(:) 이오게되는데이는라벨이름에포함되지않는다.
많은수에조건분기명령들이있다.이들은코드라벨만을인자로갖는다.단순한

것들은 단지 플래그 레지스터의 하나의 플래그 값만 보고 분기를 할 지 안할 지 결정

한다. 표 2.3 은이러한명령들을정리했다. (PF 는패리티플래그(parity Ęag) 로 결과

값의하위 8비트의수의홀짝성을나타낸다.)

JZ ZF 가세트될때에만분기
JNZ ZF 가언세트될때에만분기
JO OF 가세트될때에만분기
JNO OF 가언세트될때에만분기
JS SF 가세트될때에만분기
JNS SF 가언세트될때에만분기
JC CF 가세트될때에만분기
JNC CF 가언세트될때에만분기
JP PF 가세트될때에만분기
JNP PF 가언세트될때에만분기

표 2.3: 단순한조건분기들

다음의사코드 (pseudo code,알고리즘을이해하기쉽게프로그래밍문법으로표현

한것)는 :

if ( EAX == 0 )

EBX = 1;
else

EBX = 2;

어셈블리로아래와같이쓰일수있다.

1 cmp eax, 0 ; 플래그를 세팅한다 (eax - 0 = 0 이면 ZF 가 세트)

2 jz thenblock ; ZF 가 세트되면 thenblock 분기한다.

3 mov ebx, 2 ; if 문의 else 부분

4 jmp next ; if 문의 then 부분. 그냥 분기한다

5 thenblock:
6 mov ebx, 1 ; if 문의 then 부분.

7 next:

다른 명령들의 경우 표 2.3 에 있는 조건 분기를 사용하기에는 너무 어렵다. 이를

설명하기위해아래의사코드를보자

if ( EAX >= 5 )

EBX = 1;
else

EBX = 2;

만약 EAX 가 5 이상이라면, ZF 는 세트 되거나 언세트 되고 SF 는 OF 와 같을 것이다.

아래는이를확인하는어셈블리코드이다. (이때, EAX가부호있는정수라고생각하

자)

1 cmp eax, 5

2 js signon ; SF 가 1 이면 signon 으로 분기

3 jo elseblock ; OF = 1, SF = 0 면 elseblock 으로 분기

4 jmp thenblock ; SF = 0, OF = 0 이면 thenblock 으로 분기

5 signon:

부호있음 부호없음
JE vleft = vright이면분기 JE vleft = vright이면분기
JNE vleft 6= vright이면분기 JNE vleft 6= vright이면분기
JL, JNGE vleft < vright이면분기 JB, JNAE vleft < vright이면분기
JLE, JNG vleft ≤ vright이면분기 JBE, JNA vleft ≤ vright이면분기
JG, JNLE vleft > vright이면분기 JA, JNBE vleft > vright이면분기
JGE, JNL vleft ≥ vright이면분기 JAE, JNB vleft ≥ vright이면분기

표 2.4: 부호있는/없는경우의비교명령들

6 jo thenblock ; SF = 1, OF = 1 이면 thenblock 으로 분기

7 elseblock:
8 mov ebx, 2

9 jmp next
10 thenblock:
11 mov ebx, 1

12 next:

위의코드들은매우난해하다. 다행이도, 80x86 은 위와같은작업들을훨씬쉽게

할수있도록여러종류의분기명령들을더지원한다.이들은각각부호있는/없는경우

가각각나뉘어있다.표 2.4 에이명령들이잘나타나있다.같다/다르다분기명령은

(JE, JNE) 는부호가있는/없는정수에모두동일하다. (사실, JE와 JNE는 JZ와 JNZ와

각각동일하다)각각의다른분기명령들은2개의동의어를갖고있다.예를들어서 JL
(미만이면분기, jump less than) 과 JNGE(같거나초과가아니면분기, jump not greater
than or equal to) 이다. 이들은모두동일한명령들인데왜냐하면

x < y =⇒ not(x ≥ y)

부호없는정수의분기명령의경우초과를 A(above), 미만을 B(below) 로대응시켰고

부호있는정수의분기명령의경우 G 와 L 에대응시켰다.

위새로운분기명령들을이용하여위의의사코드는다음과같이좀더쉽게어셈블

리어로바뀔수있다.

1 cmp eax, 5

2 jge thenblock
3 mov ebx, 2

4 jmp next
5 thenblock:
6 mov ebx, 1

7 next:

2.2.3 루프명령

80x86 은 for 과같은명령을위해몇가지명령들을지원한다. 이명령들은한개의

인자만가지며이는코드라벨이다.

LOOP ECX 의값을감소시키고, ECX 6= 0 이면라벨로분기한다.

LOOPE, LOOPZ ECX 의값을 감소시키고 (플래그 레지스터의 값은 바뀌지 않는다)

만일 ECX 6= 0 이고 ZF = 1 이면분기한다.

LOOPNE, LOOPNZ ECX 의값을감소시키고 (플래그레지스터의값은바뀌지않는

다)만일 ECX 6= 0 이고 ZF = 0, 이면분기한다.

아래두개의루프들은순차적인검색루프에매우효과적이다. 아래는그의사코

드이다.

sum = 0;

for( i=10; i >0; i−− )

sum += i ;

어셈블리로다음과같이바뀐다 :

1 mov eax, 0 ; eax 는 합

2 mov ecx, 10 ; ecx 는 i
3 loop_start:
4 add eax, ecx
5 loop loop_start

2.3 보통의제어구조를번역하기

이부분에서는고급언어의제어구조가어셈블리언어로어떻게바뀌는지살펴볼

것이다.

2.3.1 If문

아래는의사코드는 :

if ( condition )

then_block;

else
else_block ;

와같이바뀐다 :

1 ; code to set FLAGS
2 jxx else_block ; xx 를 적절히 선택하여 조건이 거짓이면 분기하게 한다.

3 ; then block을 위한 부분

4 jmp endif
5 else_block:
6 ; else block 을 위한 부분

7 endif:

만약 else 가없다면 else_block분기는 endif로의분기로바뀔수있다.

1 ; code to set FLAGS
2 jxx endif ; xx 를 적절히 선택하여 조건이 거짓이면 분기하게 한다.

3 ; then block 을 위한 부분

4 endif:

2.3.2 While루프

while 루프는조건을상단에서검사하는루프이다 :

while( condition ) {

body of loop;

}

이는어셈블리어로다음과같이바뀐다 :

1 while:
2 ; code to set FLAGS based on condition
3 jxx endwhile ; xx 를 적절히 선택해 거짓이면 분기하게 함

4 ; body of loop
5 jmp while
6 endwhile:

2.3.3 Do while루프

do while 루프는조건을하단에서검사한다 :

do {

body of loop;

} while( condition );

이는어셈블리어로다음과같이바뀐다 :

1 do:
2 ; body of loop
3 ; 조건에 따라 플래그 레지스터를 세트한다.

4 jxx do ; xx 를 적절히 선택해 거짓이면 분기하게 함

2.4 예제 :소수찾는프로그램

이부분에서는소수를찾는프로그램에대해살펴볼것이다. 소수는 1 과자기자

신만을 약수로 갖는 수 임을 기억해라. 이 소수를 찾는 공식은 알려져 있지 않다. 이

프로그램이사용하는기본적인방법은특정한값이하의모든홀수들의소인수를찾는

것이다 4) 만약 어떤 홀수의 소인수가 없다면 그 홀수는 소수가 된다. 그림 2.3 은 위

내용을 C 로쓴기본적인알고리즘을나타낸다.

여기어셈블리어버전이있다 :

prime.asm
1 %include "asm_io.inc"
2 segment .data
3 Message db "Find primes up to: ", 0

4

5 segment .bss

4) 2 는유일한짝수소수이다

1 unsigned guess; /*현재소수라추측되는것 */
2 unsigned factor; /* guess의가능한소인수 */
3 unsigned limit ; /* 이값까지소수를찾는다 */
4

5 printf (”Find primes up to : ” );
6 scanf(”%u”, & limit );
7 printf (”2\n” ); /* 처음두소수를특별한경우로*/
8 printf (”3\n” ); /* 취급한다 */
9 guess = 5; /*guess의초기값*/

10 while ( guess <= limit ) {

11 /* guess의소인수를살펴본다*/
12 factor = 3;
13 while ( factor * factor < guess &&
14 guess % factor != 0 )

15 factor += 2;
16 if ( guess % factor != 0 )

17 printf (”%d\n”, guess);
18 guess += 2; /*홀수만살펴본다*/
19 }

그림 2.3:

6 Limit resd 1 ; 최대 이 수 까지 소수를 찾는다.

7 Guess resd 1 ; 현재 소수라 추측되고 있는 것

8

9 segment .text
10 global _asm_main
11 _asm_main:
12 enter 0,0 ; 셋업 루틴

13 pusha
14

15 mov eax, Message
16 call print_string
17 call read_int ; scanf("%u", & limit );
18 mov [Limit], eax
19

20 mov eax, 2 ; printf("2\n");
21 call print_int
22 call print_nl
23 mov eax, 3 ; printf("3\n");
24 call print_int
25 call print_nl
26

27 mov dword [Guess], 5 ; Guess = 5;
28 while_limit: ; while ( Guess <= Limit )

29 mov eax,[Guess]
30 cmp eax, [Limit]

31 jnbe end_while_limit ; 숫자들이 부호 없는 정수이므로 jnbe 이용

32

33 mov ebx, 3 ; ebx 은 factor = 3;
34 while_factor:
35 mov eax,ebx
36 mul eax ; edx:eax = eax*eax
37 jo end_while_factor ; 위 계산서 오버플로우 발생시 점프.

38 cmp eax, [Guess]
39 jnb end_while_factor ; if !(factor*factor < guess)
40 mov eax,[Guess]
41 mov edx,0
42 div ebx ; edx = edx:eax % ebx
43 cmp edx, 0

44 je end_while_factor ; if !(guess % factor != 0)

45

46 add ebx,2 ; factor += 2;
47 jmp while_factor
48 end_while_factor:
49 je end_if ; if !(guess % factor != 0)

50 mov eax,[Guess] ; printf("%u\n")
51 call print_int
52 call print_nl
53 end_if:
54 add dword [Guess], 2 ; guess += 2

55 jmp while_limit
56 end_while_limit:
57

58 popa
59 mov eax, 0 ; return back to C
60 leave
61 ret prime.asm



41

제3장

비트연산

3.1 쉬프트연산

어셈블리언어는프로그래머가데이터의개개의비트를바꿀수있게해준다.이중

가장흔한비트연산은바로쉬프트(shi) 연산이다. 쉬프트연산은데이터의비트의

위치를옮겨준다. 쉬프트는왼쪽(i.e 최상위비트쪽으로) 이나오른쪽(최하위비트쪽

으로)일어날수있다.

3.1.1 논리쉬프트

논리쉬프트는쉬프트중가장단순한형태이다.이는매우단순하게쉬프트연산을

한다.그림 3.1 는단일바이트의쉬프트연산을보여준다.

원래값 1 1 1 0 1 0 1 0
왼쪽쉬프트 1 1 0 1 0 1 0 0
오른쪽쉬프트 0 1 1 1 0 1 0 1

그림 3.1: 논리쉬프트

한가지주목할점은새롭게나오는비트는언제나 0이라는사실이다. SHL과 SHR
명령은각각왼쪽, 오른쪽쉬프트연산을수행한다. 이명령들을통해몇자리든지쉬

프트할수있게된다.쉬프트할자리수는상수이거나, CL레지스터에저장된값이여야

한다. 데이터로 부터 쉬프트 된 마지막 비트는 캐리 플래그에 저장된다. 아래는 예제

코드이다.

1 mov ax, 0C123H
2 shl ax, 1 ; 1 비트 왼쪽으로 쉬프트, ax = 8246H, CF = 1

3 shr ax, 1 ; 1 비트 오른쪽으로 쉬프트, ax = 4123H, CF = 0

4 shr ax, 1 ; 1 비트 오른쪽으로 쉬프트, ax = 2091H, CF = 1

5 mov ax, 0C123H
6 shl ax, 2 ; 2 비트 왼쪽으로 쉬프트, ax = 048CH, CF = 1

7 mov cl, 3

8 shr ax, cl ; 3 비트 오른쪽으로 쉬프트, ax = 0091H, CF = 1

3.1.2 쉬프트의이용

쉬프트 연산은 빠른 곱셈과 나눗셈 연산에 가장 많이 사용된다. 십진 체계에서는

10의멱수를곱하거나나누는것이매우단순했다.단지숫자들만쉬프트해주면되기

때문이다. 이는 이진수에서 2 의 멱수의 경우도 마찬가지 이다. 예를 들어서 이진수

10112 (십진수로 11) 에 2 를곱하는것은단지왼쪽으로한자리쉬프트하여 101102

(십진수로22)를얻는것과같다.또한2로나눈것의몫도또한오른쪽으로쉬프트한

결과이다. 2로나누기위해선한자리오른쪽으로쉬프트하면된다 ; 4 (22)로나누기

위해선, 두 자리를오른쪽으로쉬프트 ; 8 (23) 로 나누기귀해선 3 자리를오른쪽으로

쉬프트하면된다.쉬프트연산은 MUL나 DIV명령에비해매우빠르다.

사실,부호없는정수들에서만곱셈과나눗셈을위해쉬프트연산이사용될수있다.

부호가 있는 정수들에게선 적용되지 않는다. 예를 들어서 2 바이트 FFFF (−1) 를 생

각해보면오른쪽으로한번쉬프트되면그결과는 7FFF 로 +32, 767 이된다.따라서

부호있는값에선다른형태의쉬프트연산이사용되어야한다.

3.1.3 산술쉬프트

이 쉬프트들은 부호 있는 정수에 2 의 멱수의 곱셈과 나눗셈을 빠르게 수행하게

위해서만들어졌다.이들은부호비트가올바르게설정되어있다고생각한다.

SAL 산술왼쪽쉬프트 -이는 SHL와동일하다.이는 SHL과정확히같은기계어코드로

번역된다.부호비트가쉬프트연산에의해바뀌지않는이상,결과값은정확하다.

SAR 산술오른쪽쉬프트 -이새로운명령은피연산자의부호비트(i.e.최상위비트)를

쉬프트하지않는다.다른비트들은보통의비트연산들처럼오른쪽으로쉬프트

되며왼쪽에서추가되는새비트들은이전의부호비트들과값이같다. (즉, 부호

비트가 1 이였다면 새롭게 추가되는 비트들도 1 이 된다) 따라서, 바이트가 이

연산에 따라 쉬프트 된다면 하위 7 비트 들만 쉬프트 되는 것이다. 다른 쉬프트

연산들처럼마지막으로쉬프트되는비트는캐리플래그에저장된다.

1 mov ax, 0C123H
2 sal ax, 1 ; ax = 8246H, CF = 1

3 sal ax, 1 ; ax = 048CH, CF = 1

4 sar ax, 2 ; ax = 0123H, CF = 0

3.1.4 회전쉬프트

회전쉬프트는논리쉬프트처럼작동하지만다른점은쉬프트를통해사라진끄트

머리의데이터가새롭게추가되는데이터와같다는점이다. 따라서, 데이터는하나의

순환고리로볼수있다. 두개의가장단순한회전쉬프트명령으로는 ROL과 ROR이
있으며이는각각왼쪽과오른쪽회전을가리킨다.다른쉬프트연산들처럼이명령은

마지막으로쉬프트된비트를캐리플래그에저장한다.

1 mov ax, 0C123H
2 rol ax, 1 ; ax = 8247H, CF = 1

3 rol ax, 1 ; ax = 048FH, CF = 1



불리언비트연산 43

4 rol ax, 1 ; ax = 091EH, CF = 0

5 ror ax, 2 ; ax = 8247H, CF = 1

6 ror ax, 1 ; ax = C123H, CF = 1

데이터의 비트들과 함께 캐리 플래그의 값을 회전시키는 명령들도 있는데, 이는

RCL과 RCR이다. 예를들어서 AX레지스터가이명령을통해회전된다면 AX와캐리

플래그와함께총 17비트가회전되게된다.

1 mov ax, 0C123H
2 clc ; clear the carry flag (CF = 0)

3 rcl ax, 1 ; ax = 8246H, CF = 1

4 rcl ax, 1 ; ax = 048DH, CF = 1

5 rcl ax, 1 ; ax = 091BH, CF = 0

6 rcr ax, 2 ; ax = 8246H, CF = 1

7 rcr ax, 1 ; ax = C123H, CF = 0

3.1.5 간단한프로그램

아래는 EAX 레지스터에얼마나많은수의비트들이켜져있는지 (i.e. 1 인지)세는

소스이다.

1 mov bl, 0 ; bl 에 켜진 비트들의 수를 보관

2 mov ecx, 32 ; ecx 는 루프 카운터

3 count_loop:
4 shl eax, 1 ; 쉬프트 된 비트는 캐리 플래그에 들어감.

5 jnc skip_inc ; 만일 CF == 0 면 skip_inc 로 분기

6 inc bl
7 skip_inc:
8 loop count_loop

위의 코드는 원래 EAX 에 저장되어 있던 값을 파괴시킨다. (EAX 는 루프 끝에선 0 이

되어버린다) 만약 EAX의값을저장하고싶으면 4행을 rol eax, 1로바꾸어주면

된다.

3.2 불리언비트연산

4개의중요한불리언(Boolean) 연산자가있는데 : AND, OR, XOR, NOT 이다.진리

표에는이들의가능한모든피연산자들에대한연산수행시나타나는결과값을정리

해놓았다.

3.2.1 AND연산

오직두비트의값이1일때에만AND연산의결과값이1이된다.나머지경우에는

모두표 3.1 의진리표에잘나와있듯이 0이된다.

프로세서는 이러한 연산들을 데이터의 각 비트에 대해 일일히 적용 시킨다. 예를

들어서 AL과 BL의값에 AND연산을취했을때,기본적인 AND연산은두개의레지스

X Y X AND Y
0 0 0
0 1 0
1 0 0
1 1 1

표 3.1: AND 연산

1 0 1 0 1 0 1 0
AND 1 1 0 0 1 0 0 1

1 0 0 0 1 0 0 0

그림 3.2: 바이트에 AND 연산취하기

X Y X OR Y
0 0 0
0 1 1
1 0 1
1 1 1

표 3.2: OR 연산

터의각각대응되는 8쌍의비트들에대해각각적용된다.이는그림 3.2 에잘나타나

있다.아래는예제코드이다.

1 mov ax, 0C123H
2 and ax, 82F6H ; ax = 8022H

3.2.2 OR연산

두비트가모두0일때에만포함적(inclusive)OR연산 1) 의결과가0이된다.나머

지경우는모두결과가1이되며이는진리표 3.2에잘나타나있다.아래는예제코드

이다 :

1 mov ax, 0C123H
2 or ax, 0E831H ; ax = E933H

3.2.3 XOR연산

두비트가같을때에만배타적 OR 의연산결과가 0이된다. 두비트가다르면연

산의결과는 1이되며이는진리표 3.3 에잘나타나있다.아래는예제코드이다 :

1) (역자주)이는우리가보통 OR 연산이라부르는것이다

X Y X XOR Y
0 0 0
0 1 1
1 0 1
1 1 0

표 3.3: XOR 연산

X NOT X
0 1
1 0

표 3.4: NOT 연산

i 번째비트를켠다 2i 와 OR 연산을한다. (이는 i 번째비트만켜진(1
인)이진수이다.)

i 번째비트를끈다 오직 i 번째 비트만 꺼진(0 인) 이진수 와 AND 연
산을한다. 이러한피연산자를마스크(mask) 라고
부른다.

i 번째비트에보수취하기 2i 와 XOR 연산을한다.

표 3.5: 불리언연산의사용

1 mov ax, 0C123H
2 xor ax, 0E831H ; ax = 2912H

3.2.4 NOT 연산

NOT 연산은단항(unary) 연산이다. (i.e 이는오직한개의피연산자에대해연산

한다. AND 연산같이두개의피연산자를가지는연산이아니다)어떠한비트의 NOT
연산을하게되면그비트의정반대되는값이되며이는진리표 3.4에잘나타나있다.

아래는예제코드이다 :

1 mov ax, 0C123H
2 not ax ; ax = 3EDCH

NOT 연산이 1의보수를찾는다는것을주목해라. 다른비트연산(Bitwise opera-
tion) 들과는달리 NOT명령은 FLAGS레지스터의어떠한값도바꾸지않는다.

3.2.5 TEST 명령

TEST명령은 AND 연산을수행하지만결과는보관하지않는다. 이는오직연산결

과의따라 FLAGS레지스터의값만을바꿀뿐이다. (이는 CMP명령이뺄셈을수행하지

만 오직 FLAGS 레지스터의 값 만을 바꾸었던 것과 일맥상통 하다) 예를 들어, 연산의

결과가 0이라면 ZF가세트된다.

3.2.6 비트연산의사용

비트연산은개개의값을다른비트에는전혀영향을주지않고도수정하는경우에

매우편리하다.표 3.5에는이러한연산들의3가지자주쓰이는경우를보여주고있다.

아래는위아이디어를적용한예제코드이다.

1 mov ax, 0C123H
2 or ax, 8 ; 3 번째 비트를 킨다, ax = C12BH
3 and ax, 0FFDFH ; 5 번째 비트를 끈다, ax = C10BH
4 xor ax, 8000H ; 31 비트를 반전, ax = 410BH
5 or ax, 0F00H ; 니블을 킨다, ax = 4F0BH

1 mov bl, 0 ; bl 은 값이 1 인 비트의 수를 갖는다.

2 mov ecx, 32 ; ecx 는 루프 카운터 이다.

3 count_loop:
4 shl eax, 1 ; 캐리플래그에 비트를 쉬프트

5 adc bl, 0 ; bl 에 캐리 플래그를 더한다.

6 loop count_loop

그림 3.3: ADC를이용해비트세기

6 and ax, 0FFF0H ; 니블을 끈다, ax = 4F00H
7 xor ax, 0F00FH ; 니블을 반전, ax = BF0FH
8 xor ax, 0FFFFH ; 1의 보수, ax = 40F0H

AND 연산은 2 의 멱수로 나눈 나머지를 찾는데 사용할 수 있다. 2i 로 나눈 수의

나머지를 찾기 위해선 값이 2i − 1 인 마스크와 AND 연산을 하면 된다. 이 마스크는

i−1번째비트까지모두1이다.이마스크에나머지가저장된다.AND연산의결과는

이러한비트들을저장하고나머지는모두 0이되게한다.아래는 100을 16으로나눈

나머지와몫을저장하는코드이다.

1 mov eax, 100 ; 100 = 64H
2 mov ebx, 0000000FH ; mask = 16 - 1 = 15 or F
3 and ebx, eax ; ebx = remainder = 4

CL레지스터를이용하여데이터의임의의비트를수정하는것이가능하다. 아래의코

드는 EAX 의 임의의 비트를 세트 (1 로 만듦) 하는 코드 이다. BH 에 세트가 될 비트의

위치가저장되어있다.

1 mov cl, bh ; OR 연산을 할 수를 만듦

2 mov ebx, 1

3 shl ebx, cl ; CL 번 쉬프트

4 or eax, ebx ; 비트를 세트

비트를오프(off, 0 으로만듦)하는일은조금더복잡하다.

1 mov cl, bh ;AND 연산을 할 수를 만듦

2 mov ebx, 1

3 shl ebx, cl ; cl 번 쉬프트

4 not ebx ; 비트 반전

5 and eax, ebx ; 비트를 끔

임의의비트를반전시키는코드는여러분의몫으로남기겠다.

80x86 프로그램에서아래와같은난해한명령을보는일이종종있다 :

xor eax, eax ; eax = 0

어떠한수가자기자신과 XOR 연산을하면0이된다.이명령은동일한 MOV명령보다

기계어코드크기가작기때문에자주사용된다.

3.3 조건분기명령을피하기

현대의프로세서들은코드를빠르게실행하기위해여러가지복잡한방법들을사

용한다.가장흔한방법으로는추론적실행(speculative execution) 이다.이기술은 CPU
의병렬처리기술을이용하여여러개의명령들을동시에실행한다.그런데조건분기

명령에는이방법을적용할수없다.왜냐하면대다수의경우프로세서는분기를할지

안할지알수없기때문이다. 만약분기를하게되면분기를안할때와다른명령들을

실행하게된다.프로세서는분기를할지안할지예측을하려고노력한다.만약예측이

틀리다면프로세서는쓸데없는코드를실행하는데시간을빼앗기게된다.

위문제를피하는방법으로는조건분기명령을되도록사용하지않는것이다.그림

3.1.5의 예제코드는어떻게조건분기명령사용회수를줄이는지잘나타내고있다.

이전의예제에서는 EAX 레지스터의켜진비트들의수를세었다. 이는 INC명령을중

단하기위해조건분기명령을사용했다.그림 3.3는 ADC명령을사용해캐리플래그에

직접적으로더해분기를사용하지않는방법을보여주고있다.

SETxx명령은특정한경우에서분기를이용하지않을수있는방법을제공한다.이

명령들은플래그레지스터의상태에따라바이트레지스터나메모리의값을 0또는 1

로 바꾸어 준다. SET 다음에 오는 문자들은 조건 분기 명령에서의 것들과 같다. 만약

SETxx의조건이참이라면그결과는 1, 거짓이면 0이저장된다.예를들어

setz al ; Z 플래그가 세트 되었으면 1, 아니면 0 이 al 에 저장

이명령들을이용하여여러분은분기를사용하지않고도값들을계산하는놀라운방법

들을만들수있을것이다.

예를들어서두값들중최대값을찾는프로그램을만든다고하자.이문제를해결하

는간단한방법으로는 CMP와조건분기명령을사용해어떠한값이더큰지결정하는

것이다.하지만아래의코드는어떻게조건분기명령을사용하지않고도최대값을찾을

수있는지보여준다.

1 ; file: max.asm
2 %include "asm_io.inc"
3 segment .data
4

5 message1 db "Enter a number: ",0
6 message2 db "Enter another number: ", 0

7 message3 db "The larger number is: ", 0

8

9 segment .bss
10

11 input1 resd 1 ; 첫 번째 숫자 입력

12 segment .text
13 global _asm_main
14 _asm_main:
15 enter 0,0 ; 셋업 루틴

16 pusha
17



48 제3장 비트연산

18 mov eax, message1 ; 첫 번째 메세지 출력

19 call print_string
20 call read_int ; 첫 번째 수 입력

21 mov [input1], eax
22

23 mov eax, message2 ; 두 번째 메세지 출력

24 call print_string
25 call read_int ; eax 에 두 번째 수 입력

26

27 xor ebx, ebx ; ebx = 0

28 cmp eax, [input1] ; 두 번째와 첫 번째 숫자를 비교

29 setg bl ; ebx = (input2 > input1) ? 1 : 0

30 neg ebx ; ebx = (input2 > input1) ? 0xFFFFFFFF : 0

31 mov ecx, ebx ; ecx = (input2 > input1) ? 0xFFFFFFFF : 0

32 and ecx, eax ; ecx = (input2 > input1) ? input2 : 0

33 not ebx ; ebx = (input2 > input1) ? 0 : 0xFFFFFFFF
34 and ebx, [input1] ; ebx = (input2 > input1) ? 0 : input1
35 or ecx, ebx ; ecx = (input2 > input1) ? input2 : input1
36

37 mov eax, message3 ; 결과를 출력

38 call print_string
39 mov eax, ecx
40 call print_int
41 call print_nl
42

43 popa
44 mov eax, 0 ; C 로 리턴

45 leave
46 ret

위 코드에서 최대값을 올바르게 선택하기 위해 비트 마스크를 이용할 수 있다는

사실을이용했다. 30 행에서사용된 SETG명령은두번째입력값이최대값이면 1, 아

니면 0 으로 BL 레지스터를 설정한다. 이는 우리가 원하던 비트 마스크가 아니다. 따

라서올바른비트마스크를만들기위해 31행에서 EBX 레지스터전체에 NEG명령을

사용하였다. (이전의 EBX 레지스터의 값은 0 이 되었음을 주목해라) 만약 EBX 가 0

이라면 아무것도 하지 않는다. 그러나 EBX 가 1 이라면 그 결과는 -1의 2의 보수, 즉

0xFFFFFFFF 가된다. 이는우리가필요했던비트마스크이다. 코드의나머지부분은

비트마스크를이용하여입력값의정확한최대값을계산하게된다.

다른방법으로는 DEC문장을이용하는것이다.위코드에선 NEG를 DEC로바꾸어도

그 결과는 0 이나 0xFFFFFFFF 가 된다. 하지만 NEG 명령을 사용할 때와 달리 경우가

뒤바뀐다.



C 에서비트값바꾸기 49

매크로 의미

S_IRUSR 유저가읽기가능
S_IWUSR 유저가쓰기가능
S_IXUSR 유저가실행가능
S_IRGRP 그룹이읽기가능
S_IWGRP 그룹이쓰기가능
S_IXGRP 그룹이실행가능
S_IROTH 아더(other) 읽기가능
S_IWOTH 아더가쓰기가능
S_IXOTH 아더가실행가능

표 3.6: POSIX 파일실행권한명령들

3.4 C에서비트값바꾸기

3.4.1 C에서의비트연산자

일부고급언어들과는달리 C 는비트연산을위한연산자들을지원한다. AND 연

산은이진 &연산자 2) 로나타난다.OR연산은이진 |연산자로나타난다. XOR연산은

이진 ^연산자로나타난다.그리고 NOT 연산은단항 ~연산자로나타난다.

C에서의쉬프트연산은 <<와 >>이진연산자가수행한다. <<연산자는왼쪽쉬프트,

>>연산자는오른쪽쉬프트를수행한다. 이들은 2개의피연산자를갖는다. 왼쪽의피

연산자는쉬프트할값이고오른쪽의피연산자는쉬프트할비트수를나타낸다.만약

부호없는형식의값을쉬프트한다면논리쉬프트가이루어진다.만약부호있는형식

(예를들면 int) 의 쉬프트를하면산술쉬프트가이루어진다. 아래는이러한연산자

들을이용한예제코드이다.

1 short int s ; /* short int가16비트라고생각 */

2 short unsigned u;

3 s = −1; /* s = 0xFFFF (2의보수) */

4 u = 100; /* u = 0x0064 */

5 u = u | 0x0100; /* u = 0x0164 */

6 s = s & 0xFFF0; /* s = 0xFFF0 */

7 s = s ^ u; /* s = 0xFE94 */

8 u = u << 3; /* u = 0x0B20 (논리쉬프트) */

9 s = s >> 2; /* s = 0xFFA5 (산술쉬프트) */

3.4.2 C에서비트연산자사용하기

C에서사용되는비트연산자들은어셈블리에서사용되는경우와동일하게사용할

수있다.이를통해빠른곱셈과나눗셈을수행하고개개의비트를조작할수있다.사실,

똑똑한 C 컴파일러들은 x*=2와같은문장에선쉬프트연산을자동적으로사용한다.

많은수의운영체제 API3) 들은 (예를들어서 POSIX4) 와 Win32 가 대표적이다.)

2) 이연산자는 & & 와단항 & 연산자와는다르다!
3) 응용프로그램프로그래밍인터페이스(Application Programming Interface)
4) 이식가능운영체제인터페이스 (Portable Operating System Interface for Computer Environment) 의

준말로, UNIX 를기반으로 IEEE 에의해개발됨

들은비트로데이타가인코드(encode)된피연산자를사용하는함수를사용한다.예를

들어서 POSIX 시스템에서파일시스템접근권한에는 3 가지유형의유저들이있다 :
유저(user) (정확히말하자면오너(owner)), 그룹(group),아더(others) 이다.각각의유

형의사용자들은파일을읽기,쓰기, 실행수있는권한을가질수있다.파일의권한을

변경하기위해선C프로그래머가각각의비트들을수정할수있어야만한다. POSIX는

이를돕기위해몇가지의매크로를정의했다 (표 3.6참조) chmod함수는파일의권한

을설정하기위해사용된다.이함수는두개의인자를가지는데,하나는권한을설정할

파일의이름문자열이고,또하나는어떻게권한을설정할지에대한정수(Integer)값 5)

이다.예를들어서아래의코드는파일의권한을오너는파일을읽고쓰기가능,그룹은

읽기가능,그리고아더는어떠한접근도불가하게설정하는코드이다.

chmod(”foo”, S_IRUSR | S_IWUSR | S_IRGRP );

POSIX stat 함수는 현재 파일의 접근 권한 비트값을 알아내는데 사용할 수 있다.

chmod함수와함께사용하면다른값들을바꾸지않고도현재파일의권한들을변경할

수 있다. 아래는 아더의 쓰기 권한을 지우고, 오너에 읽기 권한을 추가하는 명령이다.

다른권한들은변경되지않았다.

1 struct stat öle_stats ; /* stat() 이사용할구조체 */

2 stat (”foo” , & öle_stats ); /* 파일정보를읽는다

3 öle_stats.st_mode에는권한비트가들어간다. */

4 chmod(”foo”, ( öle_stats . st_mode & S_IWOTH) | S_IRUSR);

3.5 빅, 리틀엔디안표현

1 장에서단일바이트이상의데이터에대한빅, 리틀엔디안표현에대해다루었

었다. 그러나, 저자는많은사람들이아직도이개념에대해잘이해하지못하므로이

단원에서좀더심화되게다루고자한다.

여러분은엔디안이멀티바이트데이타의원소들의개개의바이트(비트가아님)를

메모리에저장하는순서를가리키는말이라는사실을기억할것이다.빅엔디안은직관

적인방법이다.이는최상위바이트를먼저저장하고그순서로쭉저장한다.다시말해,

가장높은자리의비이트들이먼저저장된다.리틀엔디안은바이트를역순으로 (낮은

순서대로)저장한다. x86 계열의프로세서들은리틀엔디안표현을사용한다.

예를들어서더블워드 1234567816 을생각해보자.빅엔디안표현에선바이트들은

12 34 56 78 순으로저장된다.하지만리틀엔디안표현에선바이트들이 78 56 34 12

순으로저장된다.

아마독자여러분은머리속에다음과같은생각이떠오를것이다. 왜 칩디자이너

들이리틀엔디안표현을사용할까?사악한인텔엔지니어들이프로그래머들을혼란

스럽게하기위해그런것이였을까?아마여러분은 CPU가메모리에이러한방식으로

데이터를저장하기위해부수적인작업을해야한다고생각한다. (또한메모리에서도

역순으로읽어들일때도)하지만,그답은 ’아니오’이다.리틀엔디안표현을사용할때

에도 CPU는어떠한부수적인작업을하지않는다.여러분은 CPU가단지비트값에만

작업하는 수 많은 전기 회로들로 구성되어 있다는 것일 명심해야 한다. 그리고 비트

(또한바이트)들의순서는 CPU 에서정렬되어있을필요가없다.

5) 사실, 이정수형은 int 형으로 typedef 에의해 mode_t 형으로정의되어있다

unsigned shortword = 0x1234; /* sizeof(short) == 2라생각 */
unsigned char * p = (unsigned char *) &word;

if ( p[0] == 0x12 )
printf (”Big Endian Machine\n”);

else
printf (” Little Endian Machine\n”);

그림 3.4: 무슨엔디안인지어떻게결정할까?

2 바이트 AX 레지스터를 생각해 보자. 이는 두 개의 바이트 레지스터 AH 와 AL 로

나뉜다. CPU 에선 AH 와 AL 레지스터의 값들을 지정하는 회로 들이 있다. CPU 에선

이 회로들이 특정한 순서로 놓여 있지 않다. 따라서, AH 를 위한 회로들은 AL 을 위한

회로들앞에나뒤에놓여있다고말할수없다. AX레지스터의값을메모리로복사하는

mov명령은 AL의값을복사한다음 AH의값을복사하지만이순서를바꿔서해도CPU
에선아무런상관이없다.

같은논리가바이트의개개의비트들에도적용이된다. CPU의회로에선이개개의

비트들이특정한순서로놓여있지는않다. 그러나, 개개의비트들이 CPU 나메모리

에서 주소 지정이 될 수 없기 때문에 우리는 CPU 내부 적으로 어떠한 순서로 이들이

놓여있는지알방법(알필요도)이없다.

그림 3.4에선CPU가무슨엔디안형식을사용하는지알수있다. p포인터는word

를두개의원소를가지는문자배열로취급한다.따라서 p[0]는 word의메모리에저

장된값의첫번째바이트가된다.이는 CPU가어떤엔디안형식을사용하는지에따라

달라지게된다.

3.5.1 리틀혹은빅엔디안인지언제신경써야하는가?

많은 경우 프로그래밍 시에 CPU 가 무슨 엔디안 형식을 사용하는지는 큰 상관이

없다. 그러나바로두개의서로다른컴퓨터시스템에서이진데이터가송신될때이

를심각하게고려해야한다.이렇게송신되는경우는주로특정한형식의물리데이타

미디어(예를 들어 디스크) 나 네트워크 이다. . 아스키 데이터는 단일 바이트 이므로 UNICODE 와같은멀티바이트

문자세트의출현에따라텍스

트데이터에서조차도무슨엔

디안을 사용하는지 중요하게

되었다. UNICODE 는 두 엔디

안형식을모두지원하며데이

터를 표현하는 엔디안이 무슨

형식이니알아내는메카니즘도

있다

엔디안은큰문제가아니다.

모든 TCP/IP 헤더들은 데이터를 빅 엔디안 형식(네트워크 바이트 순서(network
byte order) 라 부른다) 으로정수들을저장한다. TCP/IP 라이브러리는 C 로하여금엔

디안 관련 문제들을 해결하기 위한 함수들을 지원한다. 예를 들어서 htonl() 함수는

더블워드(혹은 long integer) 를 호스트(host) 의 엔디안 형식에서 네트워크(network)
의 엔디안형식으로변환한다. ntohl() 함수는이와정반대인작업을수행한다. 6) 빅

엔디안시스템에서,두함수는두함수들은입력값을변화시키지않고리턴한다.이를

통해프로그래머들은네트워크프로그램들을엔디안의형식에상관하지않고어떠한

시스템에서도 올바르게 동작할 수 있게 한다. 엔디안 형식과 네트워크 프로그래밍에

대한자세한정보는 W. Richard Steven 의놀라운책인 UNIX Network Programming 을
읽어보면된다.

6) 사실,정수의엔디안형식을변환하는일은단순히바이트들을역순으로재배치하는것이다 :따라서,
빅에서리틀엔디안형식으로바꾸는작업은리틀에서빅엔디안형식으로바꾸는것과동일하다.따
라서이두함수는같은작업을한다.

1 unsigned invert_endian( unsigned x )
2 {

3 unsigned invert ;
4 const unsigned char * xp = (const unsigned char *) &x;
5 unsigned char * ip = (unsigned char *) & invert ;
6

7 ip[0] = xp[3]; /* 개개의바이트를역으로재배치 */
8 ip[1] = xp[2];
9 ip[2] = xp[1];

10 ip[3] = xp[0];
11

12 return invert ; /*뒤집어진바이트를리턴*/
13 }

그림 3.5: _엔디안변환함수

1 int count_bits( unsigned int data )

2 {

3 int cnt = 0;
4

5 while( data != 0 ) {

6 data = data & (data − 1);
7 cnt++;
8 }

9 return cnt ;
10 }

그림 3.6: 비트수세기 : 첫번째방법

그림 3.5 은더블워드의엔디안형식을변환하는 C 함수를보여주고있다. 486 프

로세서는 32 비트레지스터의바이트를역순으로재배치하는명령인 BSWAP를추가

하였다.예를들면

bswap edx ; edx 의 바이트를 역으로 재배치

이명령은16비트레지스터에서는사용될수없다.그러나 XCHG명령은두개의8비트

레지스터로 쪼개어 질 수 있는 16비트 레지스터의 바이트를 역순으로 재배치 할 수

있다.예를들면

xchg ah,al ; ax 의 바이트를 역순으로 재배치

3.6 비트수세기

이전에더블워드에서직관적인방법으로켜진비트들의수를세는프로그램을만

들었던 적이 있었다. 이 섹션에서는 덜 직관적이지만 이 장에서 논의된 비트 연산을

이용하여세는방법에대해이야기하고자한다.



비트수세기 53

1 static unsigned char byte_bit_count[256]; /* 룩업테이블 */
2

3 void initialize_count_bits ()

4 {

5 int cnt , i , data;
6

7 for( i = 0; i < 256; i++ ) {

8 cnt = 0;
9 data = i ;

10 while( data != 0 ) { /* 첫번째방법 */
11 data = data & (data − 1);
12 cnt++;
13 }

14 byte_bit_count[ i ] = cnt ;
15 }

16 }

17

18 int count_bits( unsigned int data )

19 {

20 const unsigned char * byte = ( unsigned char *) & data;
21

22 return byte_bit_count[byte[0]] + byte_bit_count[byte[1]] +
23 byte_bit_count[byte[2]] + byte_bit_count[byte [3]];
24 }

그림 3.7: 두번째방법

3.6.1 첫번째방법

첫번째방법은매우단순하지만명료하다.그림 3.6 이그코드를나타낸다.

이방법은,루프의매단계에서 data의한개의비트는꺼진다.만일데이터의모든

비트들이 꺼진다면(i.e. data 가 0 일 때) 루프가 멈춘다. data 를 0 으로 만들기 위해

필요한작업의회수는 data에들어있는켜진비트들의수와동일하게된다.

6 행에서 data 의 한 개의 비트가 꺼진다. 왜 그럴까? 예를 들어 data 값에서 오

른쪽에서부터최초로 1이나타나는지점을생각하자. 정의에따라, 그 1이나타나기

이전비트들은모두0이된다.그렇다면 data - 1의이진표현은어떻게될까?아까

최초로 1 이나타나는부분부터마지막자리까지는모두이전데이터의보수가되고

나머지부분은그대로일것이다.예를들면
data = xxxxx10000
data - 1 = xxxxx01111

이때, x들은이전의숫자와동일하다.만약 data가 data - 1과 AND연산하였다

면나머지비트들은바뀌지않지만오른쪽에서처음으로나타났던1은 0이되어있을

것이다.

3.6.2 두번째방법

임의의더블워드에서의켜진비트수를세는방법으로룩업테이블(lookup table)
7) 을 이용하는 방법도 있다. 직관적으로 모든 더블워드들의 값 마다 켜진 비트 들의

수를계산해놓으면될것같다.그러나,이방법엔2가지문제점이있다.첫째로,대략

40억개의가능한더블워드의값이존재한다. 이말은, 그룩업테이블의크기는매우

크고이를초기화하는작업도매우오래걸릴것이다. (사실상,배열의크기가 40억이

넘어간다면그냥켜진비트의수를계산하는것보다도이를초기화하는것이시간이

훨씬오래걸린다)

좀더 현실 적인 방법으로는 모든 바이트 값들의 켜진 비트 수를 계산하여 배열 에

저장해놓는것이다.그리고더블워드를4개의바이트를쪼개어서각각의바이트를배

열에서찾아각켜진비트들의합을더한면된다.그림 3.7는이방법을적용한코드를

보여주고있다.

initialize_count_bits함수는 count_bits함수를호출하기전에반드시호출

되어야만한다.왜냐하면이함수는전역 byte_bit_count배열을초기화하기때문이

다. count_bits함수는 data를크기가더블워드인변수로보지않고4개의바이트의

배열로 생각한다. dword 포인터 는 이 4 바이트 배열의 포인터 역할을 한다. 따라서

dword[0]은 data의한개의바이트(하드웨어가빅엔디안이냐, 리틀엔디안이냐의

따라서이바이트가최상위바이트거나최하위바이트가된다)를가리킨다.당연하게

도여러분은아래와같은방법을사용하여

(data >> 24) & 0x000000FF

최상위바이트값을찾을수있고,다른바이트들에게도비슷한방법을적용할수있다.

그러나,이러한방법은위배열을이용한방법보다속도가느리다.

마지막으로 볼 것은 for 루프를 통해 22 와 23 행의 수 들의 합을 손쉽게 계산할

수 있다. 그러나 for 루프는 루프 상단에 인덱스의 초기화를 필요로 하고,루프 마다

조건을검사하고, 인덱스를증가시킬것이다. 하지만위처럼그냥 4 개의수를직접

더하면훨신빠르게된다.사실,똑똑한컴파일러는 for루프를이용한형태를위처럼4

개의합으로변경해줄것이다.이렇게되풀이되는루프를줄여주는것은루프언롤링

(loop unrolling) 이라는컴파일러최적화기술이다.

3.6.3 세번째방법

이는 데이터의 켜진 비트수를 세는 또다른 놀라운 방법이다. 이 방법은 말그대로

데이터의1들과0들을더하게된다.이합은당연히이데이터의1의개수와동일하게

된다.예를들어 data변수에들어있는1바이트데이터의1의개수를센다고하자.첫

번째로알의명령을수행하게된다.

data = (data & 0x55) + ((data >> 1) & 0x55);

위작업은무엇을할까?16진수 0x55는이진수로 01010101로나타낸다.위덧셈의

첫번째항에서, data는 0x55와 AND연산을하게되어홀수번째비트들만남게된다.

두 번째부분 ((data >> 1) & 0x55) 는먼저 짝수번째비트들을 모두 홀수 번째로

쉬프트하고다시같은마스크를이용하여홀수번째비트들만남긴다.따라서,첫번째

7) (역자주)참고로룩업테이블이란컴퓨터의계산속도를향상시키기위하여컴퓨터가미리계산결
과를표로정리해놓은것이다.대표적으로 sine 값을계산하기위한룩업테이블이있다.

1 int count_bits(unsigned int x )

2 {

3 static unsigned int mask[] = { 0x55555555,
4 0x33333333,
5 0x0F0F0F0F,
6 0x00FF00FF,
7 0x0000FFFF };
8 int i ;
9 int shift ; /* 오른쪽으로쉬프트연산할횟수 */

10

11 for( i=0, shift =1; i < 5; i++, shift *= 2 )

12 x = (x & mask[i]) + ( (x >> shift ) & mask[i] );
13 return x ;
14 }

그림 3.8: 세번째방법

부분은 data의홀수번째비트들을, 두번째부분은짝수번째비트들을가지게된다.

이 두 개의 부분이 더해진다면 data 의 홀수 번째 비트들과 짝수 번째 비트들이 서로

더해진다.예를들어 data가 101100112 였다면

data & 010101012 00 01 00 01

+ (data >> 1) & 010101012 or + 01 01 00 01

01 10 00 10
오른쪽에 나타난 덧셈은 실제로 더해지는 비트들을 보여준다. 바이트의 비트들은 사

실상4개의 2비트부분으로나뉘어서4개각각이독립적인덧셈을수행하는것처럼

보인다. 2 비트부분들의합이최대 2이므로오버플로우가일어나다른부분들의덧

셈의결과를엉망으로만드는경우는발생하지않는다.

당연하게도,켜진비트들의총수는아직계산되지않았다.그러나,위와같은방법

을 몇 번 적용하게 되면 켜진 비트들의 총 수를 계산할 수 있게 된다. 두 번째 단계는

다음과같다 :

data = (data & 0x33) + ((data >> 2) & 0x33);

계속적용하면, (data는이제 011000102 이다)

data & 001100112 0010 0010

+ (data >> 2) & 001100112 or + 0001 0000

0011 0010

이제, 2 개의 4비트부분들이독립적으로더해졌다.

다음단계는이 2비트합들을더해서결과를얻는것이다.

data = (data & 0x0F) + ((data >> 4) & 0x0F);

data (001100102 와값이같다)를이용한결과는
data & 000011112 00000010

+ (data >> 4) & 000011112 or + 00000011

00000101

이제 data는 5이고이는정확한결과이다. 그림 3.8 은더블워드의경우어떻게

적용하는지나타내고있다. 이는 for루프를이용하여합을계산한다. 루프를풀어내

면좀더빠르게수행할것이다. 그러나, 루프를통해데이터의크기가다른경우에도

어떻게적용하는지보기쉽다.

서브프로그램

이장에서는서브프로그램(subprogram)을이용하여모듈화된프로그램(modular
program)및C와같은고급언어들과함께작업하는방법에대해서다루어보고자한다.

고급언어에서의서브프로그램으로는함수와프로시져들이있다.

서브프로그램을호출하는코드와서브프로그램그자체는반드시이들사이에데

이터를주고받는방법이같아야한다.이렇게,이들사이에어떠한방식으로데이터를

주고받는방법을정해놓은것을호출규약(calling convention) 이라고부른다.이장은

C 프로그램들과소통(interface)할 수있는어셈블리서브프로그램제작을위해서, C
호출 규약에 대해 많은 부분을 할애할 것이다. 이 호출 규약은 (그 외 많은 호출 규약

들도) 데이터의주소(i.e. 포인터) 를전달해서브프로그램이메모리의데이터에접근

할수있게하는방식으로이루어진다.

4.1 간접주소지정

간접적인주소지정은레지스터를포인터변수들처럼사용할수있게한다.레지스

터가포인터로사용됨을알려주기위해서대괄호([])로묶어주어야한다.예를들어

1 mov ax, [Data] ; 워드의 직접 메모리 주소 지정

2 mov ebx, Data ; ebx = & Data
3 mov ax, [ebx] ; ax = *ebx

AX 가워드를보관하고있기때문에 3행은 EBX 에저장된주소에서시작되는워

드를읽어들이게된다. 만약 AX 대신 AL 을이용한다면오직한개의바이트만읽어

들이게 될 것이다. C 에서의 변수들과는 달리 레지스터들은 형(type) 이 없다는 사실

을명심해야한다. EBX 에무엇을가리키냐는어떠한명령이사용되냐에따라완전히

달라진다.심지어명령에의해서도 EBX가포인터라는사실도정해진다.만약 EBX가

잘못사용되어도어셈블러는아무런오류를내놓지않는다.그러나,프로그램은올바

르게작동하지않는다. 이때문에어셈블리프로그래밍은고급언어프로그래밍보다

훨씬오류가많게된다.

모든32비트범용레지스터 (EAX, EBX, EDX, EDX) 와인덱스 (ESI, EDI) 레지스터

들은모두간접주소지정으로사용될수있다.통상적으로16비트와 8비트는사용될

수없다,

4.2 간단한서브프로그램예제

서브프로그램은프로그램의다른영역에서쓰이는독립적인코드라고볼수있다.

즉, 서브프로그램은 C 언어에서의함수라고생각하면된다. 분기명령을통해서브프

로그램을호출할수있지만리턴하는부분에서문제가생긴다. 만약서브프로그램이

프로그램의각기다른영역에서호출되었다면서브프로그램은호출된각각의위치로

돌아가기위해엄청난수의라벨을사용하게되므로매우복잡해질것이다.반면에아

래의코드는이를간접형태의 JMP명령을사용하여구현한것을볼수있다.이형태의

명령은 레지스터에 저장된 값을 이용하여 어디로 분기할 것인지 알 게 된다. (따라서,

레지스터들은 C 에서의함수포인터(function pointer) 과같이쓰인다)아래는 1장의

첫번째프로그램을서브프로그램을이용한형태로다시작성한것이다.

sub1.asm
1 ; 파일: sub1.asm
2 ; 서브프로그램 예제 프로그램

3 %include "asm_io.inc"
4

5 segment .data
6 prompt1 db "Enter a number: ", 0 ; 널 종료 문자를 잊지 마시오

7 prompt2 db "Enter another number: ", 0

8 outmsg1 db "You entered ", 0

9 outmsg2 db " and ", 0

10 outmsg3 db ", the sum of these is ", 0

11

12 segment .bss
13 input1 resd 1

14 input2 resd 1

15

16 segment .text
17 global _asm_main
18 _asm_main:
19 enter 0,0 ; 셋업 루틴

20 pusha
21

22 mov eax, prompt1 ; prompt 를 출력

23 call print_string
24

25 mov ebx, input1 ; input1 의 주소를 ebx 에 저장

26 mov ecx, ret1 ; ecx 에 리턴 주소를 저장

27 jmp short get_int ; 정수를 읽는다

28 ret1:
29 mov eax, prompt2 ; prompt 출력

30 call print_string
31

32 mov ebx, input2

33 mov ecx, $ + 7 ; ecx = 현재주소 + 7

34 jmp short get_int
35

36 mov eax, [input1] ; eax = input1 의 dword
37 add eax, [input2] ; eax += input2 의 dword
38 mov ebx, eax ; ebx = eax
39

40 mov eax, outmsg1
41 call print_string ; 첫 번째 메세지 출력

42 mov eax, [input1]
43 call print_int ; input1 을 출력

44 mov eax, outmsg2
45 call print_string ; 두 번째 메세지 출력

46 mov eax, [input2]
47 call print_int ; input2 를 출력

48 mov eax, outmsg3
49 call print_string ; 세 번째 메세지 출력

50 mov eax, ebx
51 call print_int ; 합 (ebx) 출력

52 call print_nl ; 개행 문자 출력

53

54 popa
55 mov eax, 0 ; C 로 리턴

56 leave
57 ret
58 ; 서브프로그램 get_int
59 ; 인자:
60 ; ebx - 정수를 저장한 dword 의 주소

61 ; ecx - 리턴할 명령의 주소

62 ; 참고 사항:
63 ; eax 에 저장된 값은 사라진다.

64 get_int:
65 call read_int
66 mov [ebx], eax ; input 을 메모리에 저장

67 jmp ecx ; 호출한 곳으로 리턴
sub1.asm

get_int서브프로그램은단순한레지스터기반호출규약을사용한다. 이는 EBX
레지스터가입력값을저장할DWORD의주소를갖고있고, ECX레지스터가되돌아갈

코드주소값을가지고있다고생각한다. 25 에서 28 행까지 ret1라벨은리턴주소를

계산하기위해사용되었다. 32에서 34행까지 $연산자는리턴주소를계산하기위해

사용되었다. $ 는 현재 행의 주소를 리턴한다. $ + 7 은 36 행의 MOV 명령의 주소를

계산하기위해사용되었다.

위두가지방식의리턴주소계산은매우복잡하다.첫번째방법은매서브프로그램

호출마다라벨을필요로한다.두번째방법은라벨을필요로하진않지만오류가나지

않기위해심사숙고할필요가있다. 예를들어 near 분기대신에 short 분기가사용되

었다면 $에더해진값은 7이아니게된다. 다행이도, 서브프로그램을호출하기위한

더쉬운방법이있다.이방법은스택(stack) 을이용한다.

4.3 스택

대다수의 CPU 는 스택의 사용을 지원한다. 스택은 후입 선출(Last-In First-Out,
LIFO 라고 부르며, 나중에 들어간 데이터가 먼저 나오게 된다) 리스트 이다. 스택은

이와같은규칙으로구성된메모리의일부분이다. PUSH명령을통해스택에데이터를

추가하고 POP명령을통해데이터를빼낼수있다.언제나빼내진데이터는마지막으로

추가된데이터이다. (이때문에스택을후입선출리스트라부른다)

SS세그먼트레지스터는스택을보관한세그먼트를정의한다. (보통이는데이터가

저장된세그먼트와동일하다) ESP레지스터는스택으로부터빼내질데이터의주소를

보관한다.이데이터는흔히스택의최상위(top) 에있다고말한다.데이터는오직더블

워드의형태로만저장된다.따라서,스택에단일바이트를집어넣을수없다.

PUSH명령은 ESP 레지스터의값을 4감소시킨후, 더블워드1) 를 [ESP]에위치한

더블워드에 집어 넣는다. POP 명령은 [ESP] 에 위치한 더블워드를 읽어 들이고, ESP
에 4를더한다. 아래의코드는 ESP 가 1000H이였다고할때명령에따라값이어떻게

변화하는지보여준다.

1 push dword 1 ; 1 이 0FFCh 에 저장됨, ESP = 0FFCh
2 push dword 2 ; 2 가 0FF8h 에 저장됨, ESP = 0FF8h
3 push dword 3 ; 3 이 0FF4h 에 저장됨, ESP = 0FF4h
4 pop eax ; EAX = 3, ESP = 0FF8h
5 pop ebx ; EBX = 2, ESP = 0FFCh
6 pop ecx ; ECX = 1, ESP = 1000h

스택은 임시적으로 데이터를 저장하는데 요긴하게 사용할 수 있다. 이를 통해서

인자(parameter)들과 지역 변수(local variable) 을 전달하고 서브프로그램 호출을 할

수있다.

80x86 은 또한 PUSHA명령을이용하여 EAX, EBX, ECX, EDX, ESI, EDI, EBP 레지

스터의값들을모두스택에집어넣을(푸시) 수있게한다. (이순서로집어넣는것은

아니다) POPA명령을통해이들을다시빼낼(팝)수있게한다.

4.4 CALL과 RET명령

80x86은서브프로그램을빠르고간편하게호출하기위해서스택을이용한2가지

의명령을지원한다.CALL명령은서브프로그램으로의무조건분기를한후,그다음에

실행될명령의주소를스택에푸시(push)한다. RET명령은그주소를팝(pop) 한후그

주소로점프를한다.이명령을이용할때,스택을정확하게관리하여 RET명령에의해

정확한주소값이팝될수있도록해야한다.

따라서이전의프로그램은새로운명령을이용하여아래와같이 25 에서 34 행을

바꾸어쓸수있다.

1) 사실,워드들도스택에들어갈수있지만32비트보호모드에서는오직더블워드들만가지고다루는
것이낫다.

mov ebx, input1
call get_int

mov ebx, input2
call get_int

또한서브프로그램 get_int를아래와같이바꾼다 :

get_int:
call read_int
mov [ebx], eax
ret

CALL 과 RET 에는몇가지장점들이있다.

• 이는단순하다!

• 서브프로그램호출이쉽게이루어지게한다. get_int가 read_int를호출하는

부분을보아라.이호출은스택에또다른주소를집어넣는다. read_int코드의

마지막부분에는 RET가리턴주소를빼내고, get_int의코드로다시분기된다.

그리고 get_int 의 RET 이 이루어진다면 리턴 주소를 빼내어 asm_main 으로

분기된다.이는스택이 LIFO 구조이기에올바르게일어난다.

스택에푸시된모든데이터는나중에반드시다시팝해야한다.예를들어,아래와

같은소스코드를생각해보자 :

1 get_int:
2 call read_int
3 mov [ebx], eax
4 push eax
5 ret ; 리턴 주소 대신 EAX 값을 팝!

이코드는올바르게리턴되지않는다.

4.5 호출규약

서브프로그램이 호출 되었을 때, 호출하는 코드와 서브프로그램 (이를 피호출자

(callee) 은서로의데이터전송방식이같아야한다.고급언어에서는호출규약(calling
conventions) 이라는기본적인데이터전송방법이있다.고급언어코드와어셈블리가

함께작업하기위해서는어셈블리코드들도반드시고급언어에서의규약과일치해야

한다. 이 호출규약은컴파일러마다다를수있으며, 코드가어떠한방식으로컴파일

되느냐에따라서 (e.g 최적화의유무)도다를수있다.하나의공통된점이라하면코드

는 CALL을통해호출하고 RET을통해리턴된다는점이다.

모든 PCC컴파일러는이장의나머지부분에서다루게될호출규약을지원한다.이

규약을통해재진입(reentrant) 서브프로그램을만들수있다. 재진입서브프로그램은

프로그램의어떤부분에서도자유롭게호출될수있다. (심지어서브프로그램그자체

내부에서도)

ESP + 4 인자
ESP 리턴주소

그림 4.1:

ESP + 8 인자
ESP + 4 리턴주소
ESP 서브프로그램데이터

그림 4.2:

1 subprogram_label:
2 push ebp ; 원래의 EBP 값을 스택에 저장

3 mov ebp, esp ; 새로운 EBP = ESP
4 ; subprogram code
5 pop ebp ; 이전의 EBP 값을 불러온다.

6 ret

그림 4.3: 통상적인서브프로그램의형태

4.5.1 스택에인자들을전달하기

서브프로그램으로스택을통해인자들이전달될수있다.이인자들은 CALL명령을

하기전에스택으로푸시되어진다. C에서처럼서브프로그램에서인자의값이바뀌기

위해서는 인자의 값이 아닌, 인자의 주소가 반드시 전달 되어야만 한다. 만약 인자의

크기가더블워드보다작다면,반드시푸시되기전에더블워드로변환되어야만한다.

스택에저장된인자들은서브프로그램에의해빼내어(Pop)지지않는다.그대신에,

이들은스택자체에서접근이가능하다.왜냐하면,

• 이들은 CALL명령전에스택에푸시되어야만하기때문에리턴주소는가장먼저

팝되어야만한다

• 인자들은종종서브프로그램의여러부분에서사용된다.보통,서브프로그램실

행내내레지스터에인자들을저장할수없으므로,메모리에저장되어야한다.따

라서,인자들을스택에보관함으로써서브프로그램의어떠한부분에서도인자에

접근할수있게된다.

스택에하나의인자를저장한서브프로그램을생각하자서브프로그램이호출되었간접 주소지정을 사용시,

80x86 프로세서는 간접

주소지정 표현시 어떠한

레지스터가 사용되느냐에

따라서 각기 다른 세그먼트에

접근한다. ESP (또한 EBP) 는

스택 세그먼트를 사용하며,

EAX, EBX, ECX, EDX 는데이타

세그먼트를 사용한다. 그러나,

보호 모드 프로그램에서는

별로 중요한 문제는 아니다.

왜냐하면 데이터 세그먼트와

스택 세그먼트는 동일하기

때문이다.

을때,스택은 4.1과같이나타난다.간접주소지정을사용하여인자에접근([ESP+4]
2)) 할수있다.

만일, 서브프로그램내에서도데이터를저장하기위해스택을사용한다면, ESP 에

더해저야할값이달라지게된다.예를들어그림 4.2에서더블워드가스택에푸시되었

을때어떻게나타나는지보여준다.이때,인자는 ESP + 4가아닌 ESP + 8에위치하게

2) 간접주소지정을사용시레지스터에상수를더하는것이가능하다. 뿐만아니라더복잡한표현들도
가능하다.이것에대한자세한내용은다음장에서다룬다

ESP + 8 EBP + 8 인자
ESP + 4 EBP + 4 리턴주소
ESP EBP 저장된 EBP

그림 4.4:

1 push dword 1 ; 1 을 인자로 전달

2 call fun
3 add esp, 4 ; 스택에서 인자를 제거

그림 4.5: 서브프로그램호출예제

된다. 따라서, ESP 를이용하여인자에접근시오류를낼확률이크다. 이문제를해결

하기위해 80386 은 EBP 라는또다른레지스터를사용가능하게했다. 이레지스터의

목적은 스택의 데이터를 가리키는 것 이다. C 호출 규약에선 반드시 서브프로그램이

먼저 EBP 의 값을 스택에 저장하고 EBP 가 ESP 와 같게 만들게 해야 한다. 이를 통해

EBP 의값을변경하지않고도스택에데이터가푸시되고팝됨에따라 ESP 가변경될

수있게한다. 서브프로그램의끝부분에선원래의 EBP 값이반드시불러와져야한다

(이때문에우리가서브프로그램처음에 EBP값을저장했다)그림 4.3 은위규약들을

만족하는서브프로그램의일반적인모습을보여준다.

그림 4.3의 2와 3행에서보통의서브프로그램의시작부분을보여주고있다. 5와

6행에서는마지막부분을보여준다.그림 4.4에서시작부분을막통과한후의스택의

모습을 보여준다. 이제, 인자들은 [EBP + 8] 을 통해 서브프로그램에 의해 스택에

데이터가푸시되고팝되는것에무관하게어떠한부분에서도인자에접근할수있게

되었다.

서브프로그램이종료되면,스택에푸시되었던인자들은반드시제거되어야만한다.

C 호출규약은호출자의코드가이를하게명시했다. 모든언어의호출규약이동일한

것이아니다.예를들어파스칼의호출규약은서브프로그램이이를제거하라명시했다.

(이를쉽게하기위한 RET 명령의다른형태가있다) 일부 C 컴파일러들은이규약을

지원하기도한다. pascal키워드를함수의원형과정의부분에서사용하여컴파일러로

하여금파스칼의호출규약을이용하게만들수있다.사실, 마이크로소프트윈도우즈

API C 함수들이 이용하는 stdcall 규약 도 파스칼의 경우와 같다. 이 방법을 이용하

면 어떠한 장점이 있을까? 이는 C 호출 규약보다 조금 더 능률적이다. 그렇다면 왜 C
함수들은이호출규약을이용하지않는것일까?보통 C는함수가가변개수의인자를

가짐을지원한다 (e.g., printf와 scanf가대표적).이러한형식의함수에선스택에서

인자들을지우는명령들의횟수가달라지게된다. 이러한형식의함수들은, 스택으로

부터인자들을제거하는명령은함수의호출마다달라지게된다. C 호출규약에선이

러한연산들을손쉽게할수있는명령들을지원한다. 그러나파스칼이나 stdcall 호출

규약에선이러한연산들은매우복잡하다. 따라서파스칼호출규약은이러한형식의

함수를 지원하지 않는다. 마이크로소프트 윈도우즈의 경우 어떠한 API 함수도 가변

인자를가지지않기때문에파스칼형태의호출규약을이용할수있었다.

그림 4.5는 C호출규약을이용하는서브프로그램을어떻게호출하는지보여준다.

3행은스택포인터의값을조정함으로써인자를스택에서지워버린다. POP명령을통

해이를할수도있으나,필요없는값이레지스터에저장되어야한다.사실,이특별한

경우에 대해선 많은 컴파일러들은 POP ECX 명령을 이용하여 인자를 제거한다. 왜냐

하면 ADD명령이더많은바이트를사용하기때문이다. 그러나 POP또한 ECX 의값을

바꾼다. 다음의또다른예제는위에서설명한 C 호출규약을사용한프로그램들이다.

54 행 (그리고다른행들에서) 에서하나의소스파일에여러개의데이타와텍스트세

그먼트들이정의되었음을볼수있다. 이들은링크과정에서하나의데이터와텍스트

세그먼트들로 합쳐질 것이다. 데이터와 코드를 여러개의 세그먼트로 쪼갬을 통해 서

브프로그램이사용하는데이터가서브프로그램근처에정의되게할수있다.

sub3.asm
1 %include "asm_io.inc"
2

3 segment .data
4 sum dd 0

5

6 segment .bss
7 input resd 1

8

9 ;
10 ; 의사 코드 알고리즘

11 ; i = 1;
12 ; sum = 0;
13 ; while( get_int(i, &input), input != 0 ) {

14 ; sum += input;
15 ; i++;
16 ; }

17 ; print_sum(num);
18 segment .text
19 global _asm_main
20 _asm_main:
21 enter 0,0 ; 셋업 루틴

22 pusha
23

24 mov edx, 1 ; 의사 코드에서의 i 가 edx 이다.

25 while_loop:
26 push edx ; i 를 스택에 저장

27 push dword input ; 입력값의 주소를 스택에 저장 % 정확한가?

28 call get_int
29 add esp, 8 ; i 와 &input 을 스택에서 제거

30

31 mov eax, [input]
32 cmp eax, 0

33 je end_while
34

35 add [sum], eax ; sum += input

36

37 inc edx
38 jmp short while_loop
39

40 end_while:
41 push dword [sum] ; 합을 스택에 푸시

42 call print_sum
43 pop ecx ; 스택에서 [sum] 을 제거

44

45 popa
46 leave
47 ret
48

49 ; 서브프로그램 get_int
50 ; 인자들 (스택에 푸시된 순서로 나타남)

51 ; 입력값의 수 ([ebp + 12] 에 위치)

52 ; 입력값을 저장할 워드의 주소 ([ebp + 8] 에 위치)

53 ; 참고:
54 ; eax 와 ebx 의 값들은 파괴된다.

55 segment .data
56 prompt db ") Enter an integer number (0 to quit): ", 0

57

58 segment .text
59 get_int:
60 push ebp
61 mov ebp, esp
62

63 mov eax, [ebp + 12]

64 call print_int
65

66 mov eax, prompt
67 call print_string
68

69 call read_int
70 mov ebx, [ebp + 8]

71 mov [ebx], eax ; 입력값을 메모리에 저장

72

73 pop ebp
74 ret ; 호출자로 분기

75

76 ; 서브프로그램 print_sum
77 ; 합을 출력한다

78 ; 인자:
79 ; 출력하기 위한 합([ebp+8] 에 위치)

1 subprogram_label:
2 push ebp ; 기존의 EBP 값을 스택에 저장한다.

3 mov ebp, esp ; 새로운 EBP = ESP
4 sub esp, LOCAL_BYTES ; 지역 변수에 의해 = # 바이트 필요

5 ; 서브프로그램 코드

6 mov esp, ebp ; 지역 변수를 해제(deallocate)
7 pop ebp ; 기존의 EBP 값을 불러온다.

8 ret

그림 4.6: 지역변수를이용한일반적인서브프로그램모습

80 ; 참고: eax 의 값이 파괴된다.

81 ;
82 segment .data
83 result db "The sum is ", 0

84

85 segment .text
86 print_sum:
87 push ebp
88 mov ebp, esp
89

90 mov eax, result
91 call print_string
92

93 mov eax, [ebp+8]
94 call print_int
95 call print_nl
96

97 pop ebp
98 ret sub3.asm

4.5.2 스택에서의지역변수

스택은지역변수들의편리한저장창고로사용될수있다.이는 C가보통의변수들

(또는C 용어로 automatic)을 저장하는 곳이다. 변수들을 위해 스택을 사용하는 것은

재진입서브프로그램을만드는데유용하다.재진입서브프로그램은서브프로그램그

자체를포함한어떠한곳에서도호출되어도잘작동하는서브프로그램을말한다.다시

말해, 재진입 서브프로그램은 재귀적(recursively)으로 호출 가능하다. 변수들을 위해

스택을사용하는것을통해메모리도아낄수있다.스택에저장되지않는데이터들은

프로그램의시작부터종료될때까지메모리상에상주하게된다. (C 는이러한형식

의변수들을전역(global) 이나정적(static) 변수라부른다)스택에저장된데이터들의

경우오직서브프로그램이실행될때에만메모리상에상주한다.

지역 변수들은 스택에 저장된 EBP 값 바로 다음부터 저장된다. 이들은 프로그램

서두에서, ESP 로 부터 필요한 만큼의 바이트를 뺌으로써 할당 될 수 있다. 그림 4.6

1 void calc_sum( int n, int * sump )
2 {

3 int i , sum = 0;
4

5 for( i=1; i <= n; i++ )

6 sum += i ;
7 *sump = sum;
8 }

그림 4.7: 합프로그램의 C 언어버전

1 cal_sum:
2 push ebp
3 mov ebp, esp
4 sub esp, 4 ; 지역 변수 sum 의 공간을 할당

5

6 mov dword [ebp - 4], 0 ; sum = 0

7 mov ebx, 1 ; ebx (i) = 1

8 for_loop:
9 cmp ebx, [ebp+8] ; i <= n 인가?

10 jnle end_for
11

12 add [ebp-4], ebx ; sum += i
13 inc ebx
14 jmp short for_loop
15

16 end_for:
17 mov ebx, [ebp+12] ; ebx = sump
18 mov eax, [ebp-4] ; eax = sum
19 mov [ebx], eax ; *sump = sum;
20

21 mov esp, ebp
22 pop ebp
23 ret

그림 4.8: 합프로그램의어셈블리버전

는 새로운 서브프로그램의 구조를 보여준다. EBP 레지스터는 지역 변수에 접근하기

위해사용된다. 그림 4.7 의 C 함수를보자. 그림 4.8 은이와동일한서브프로그램이

어셈블리어로어떻게쓰일수있는지보여준다.

그림 4.9은그림 4.8의프로그램상단부분을실행한후의스택의모습을보여준다.

인자,리턴정보,지역변수를포함한위와같은스택의영역을스택프레임(stack frame)
이라부른다. C 함수를호출할때마다스택에새로운스택프레임을생성하게된다. 비록 ENTER 와 LEAVE 명령들

이 서브프로그램의 시작 부분

과끝부분을간단하게해주지

만잘쓰이지는않는다.왜냐하

면,이들은동일한간단한명령

들의모임보다더느리기때문

이다.이는우리가언제나하나

의명령이여러개의명령들보

다빠르다고확신할수없음을

보여주는하나의예이다.

프로그램의시작부분과끝부분은오직이러한목적을위해만들어진두명령 ENTER
와 LEAVE 명령이 사용되고 있다. ENTER 명령은 시작 부분의 명령을 처리하고 LEAVE
명령은마지막부분의명령을처리한다. ENTER명령은두개의즉시피연산자를가진다.

C호출규약에선두번째피연산자는언제나0이다.첫번째인자는지역변수가필요로

하는바이트의크기를나타낸다. LEAVE명령은피연산자를갖지않는다.그림 4.10은

ESP + 16 EBP + 12 sump
ESP + 12 EBP + 8 n
ESP + 8 EBP + 4 리턴주소
ESP + 4 EBP 저장된 EBP
ESP EBP - 4 sum

그림 4.9:

1 subprogram_label:
2 enter LOCAL_BYTES, 0 ; 지역 변수에 의해 필요한 = # 바이트

3 ; subprogram code
4 leave
5 ret

그림4.10: ENTER와 LEAVE명령,그리고지역변수를사용한일반적인서브프로그램의
모습

이러한명령들이어떻게쓰이는지보여주고있다.이때,골격프로그램(그림 1.7)또한

ENTER와 LEAVE를사용하고있음을주목해라.

4.6 다중모듈프로그램

다중모듈프로그램(multi-module program)은한개보다많은수의목적파일(object
ĕle)들로 구성된 프로그램을 말한다. 여태까지 소개한 프로그램들은 모두 다중 모듈

프로그램이였다. 이들은 C 드라이버목적파일과 어셈블리목적 파일 (C 라이브러리

목적파일도)들로이루어져있다.링커는이러한여러개의목적파일들을하나의실행

가능한(executable)프로그램으로통합해주는것을상기해라.링커는각모듈(i.e.목적

파일)들에서의라벨들을각각의정의에맞게대응시켜주어야만한다.모듈 B에서정

의된라벨을모듈A에서사용하기위해선, extern지시어가사용되어만한다. extern
지시어옆에는반점으로구분은라벨들의목록들이나열된다.이지시어는어셈블러에

이라벨들이모듈이외부(external)에서정의되었음을알려준다. 따라서, 이러한라벨

들은 다른 모듈에서 정의되었음에도 불구하고 사용될 수 있다. asm_io.inc 파일은

read_int, 루틴등등을모두외부로선언한다.

어셈블리에서는기본적으로라벨들은외부에서접근이불가능하도록되어있다.만

약라벨이다른모듈에서접근이가능하다면이라벨은정의된모듈에서반드시전역으

로선언되어야한다. global지시어가이일을한다.아래그림 1.7의뼈대프로그램의

13행을보면, _asm_main라벨이전역으로정의되어있음을볼수있다.이정의없이는

링커 오류가 발생할 것이다. 왜냐하면 C 코드는 내부(internal)로 선언된 _asm_main
라벨을참조할수없기때문이다.

다음 코드는 이전의 예제를 두 개의 모듈을 이용하여 사용한 것이다. 이 두 개의

서브프로그램 (get_int 와 print_sum) 은 _asm_main루틴과는달리 2 개의 개개의

소스파일에작성되어있다.

main4.asm
1 %include "asm_io.inc"
2

3 segment .data
4 sum dd 0

5

6 segment .bss
7 input resd 1

8

9 segment .text
10 global _asm_main
11 extern get_int, print_sum
12 _asm_main:
13 enter 0,0 ; 셋업 루틴

14 pusha
15

16 mov edx, 1 ; edx 는 의사코드에서 'i' 와 같다.

17 while_loop:
18 push edx ; i 를 스택에 저장한다.

19 push dword input ; 입력값의 주소를 스택에 푸시

20 call get_int
21 add esp, 8 ; i 와 &input 을 스택에서 제거

22

23 mov eax, [input]
24 cmp eax, 0

25 je end_while
26

27 add [sum], eax ; sum += input
28

29 inc edx
30 jmp short while_loop
31

32 end_while:
33 push dword [sum] ; 합을 스택에 푸시

34 call print_sum
35 pop ecx ; [sum] 을 스택에서 제거

36

37 popa
38 leave
39 ret main4.asm

sub4.asm
1 %include "asm_io.inc"
2

3 segment .data
4 prompt db ") Enter an integer number (0 to quit): ", 0

5

6 segment .text
7 global get_int, print_sum
8 get_int:
9 enter 0,0

10

11 mov eax, [ebp + 12]

12 call print_int
13

14 mov eax, prompt
15 call print_string
16

17 call read_int
18 mov ebx, [ebp + 8]

19 mov [ebx], eax ; 입력값을 메모리에 저장

20

21 leave
22 ret ; 호출자로 분기

23

24 segment .data
25 result db "The sum is ", 0

26

27 segment .text
28 print_sum:
29 enter 0,0
30

31 mov eax, result
32 call print_string
33

34 mov eax, [ebp+8]
35 call print_int
36 call print_nl
37

38 leave
39 ret sub4.asm

이전의코드는오직전역코드라벨들만가지고있었다. 그러나, 전역데이터라벨

들은똑같은방법으로작동된다.

4.7 C와소통하기

오늘날,매우적은수의프로그램들만완전히어셈블리어로작성된다.컴파일러는

고급레벨코드를효율적인기계어코드로변환하는일을매우잘한다.고급언어로코

드를작성하는것이쉽기에,이는훨씬대중적이다.게다가,고급언어는어셈블리어에

비해훨씬이식성이높다!

1 segment .data
2 x dd 0

3 format db "x = %d\n", 0

4

5 segment .text
6 ...
7 push dword [x] ; x 의 값을 푸시

8 push dword format ; 형식 문자열의 주소를 푸시

9 call _printf ; _ 에 주목!

10 add esp, 8 ; 스택에서 인자들을 제거

그림 4.11: printf로호출

만일어셈블리어가직접적으로사용된다면오직코드의작은부분들에서만사용된

다.이는두가지방법으로가능하다 : C에서어셈블리서브루틴을호출하던지,인라인

(inline) 어셈블리를이용하는것이다.인라인어셈블리는프로그래머로하여금 C코드

에어셈블리문장을직접적으로넣을수있게해주는것이다. 이는매우편리하다. 그

러나,인라인어셈블리에도몇가지단점이있다.어셈블리코드는반드시컴파일러가

사용하는형식으로작성되어야만한다. 그러나, 현재 NASM 의형식을지원하는컴파

일러가없다.다른컴파일러들은다른형식을사용한다.볼랜드와마이크로소프트사의

경우MASM형식을이용한다.DJGPP와리눅스의 gcc는 GAS 3) 형식을이용한다. PC
에선어셈블리서브루틴을호출하는방법이훨씬표준화되어있다.

어셈블리루틴들은다음과같은이유로 C 코드와함게사용된다 :

• C 로하기에어렵거나매우힘든컴퓨터하드웨어의직접적인접근

• 매우빠르게작동되어야만루틴, 컴파일러가할수있는것보다도빠르게프로

그래머가직접손으로최적화를해야되는부분

마지막 이유는 예전과는 달리 거의 사실이 아니게 되었다. 컴파일러 기술은 해를

거듭하여진보하였고,컴파일러들은매우효율적인코드(특히,컴파일러최적화가진

행된다면)를 만들어 낼 수 있다. 어셈블리 루틴의 단점은 : 이식성이 낮고, 가독성이

떨어진다.

대부분의 C 호출규약은이미이야기하였다. 그러나, 몇개의설명이필요한부분

들이남아있다.

4.7.1 레지스터저장하기

먼저, C 는 서브루틴이 다음의 레지스터의 값을 보관하다고 생각한다 : EBX, ESI,
EDI, EBP, CS, DS, SS, ES.이말은,서브루틴이내부적으로이들을바꿀수없다는것이 register키워드는컴파일러

에게 C 변수를 메모리 공간에

저장하지 않고 레지스터에 저

장하라고명령하는것이다.현

대의 컴파일러들은 이를 특별

히 명시하지 않고도 자동적으

로해준다.

아니다. 그대신에, 이이것말은이레지스터들의값을바꾸더라도서브루틴이리턴할

경우이전의값들을복원할수있어야만한다는것이다. EBX, ESI, EDI의값을은반드시

변경되면안되는데왜냐하면 C는이레지스터들을레지스터변수(register variable) 로
사용하기때문이다.보통스택은위레지스터들의원래값들을저장하는데사용된다.

3) GAS 는 모든 GNU 컴파일러가 사용하는 어셈블러 이다. 이는 AT&T 문법을 이용하는데, AT&T 의
문법은문법이대략비슷한 MASM, TASM, NASM 들과는달리매우다르다.

EBP + 12 x의값
EBP + 8 형식문자열의주소
EBP + 4 주소를리턴
EBP EBP 를저장

그림 4.12: printf내부의스택모습

4.7.2 함수들의라벨

대부분의 C컴파일러들은함수나,전역/정적변수들의이름앞에 _한개를붙인다.

예를 들어서 f 라는 이름의 함수는 _f 라는 이름의 라벨로 대응된다. 따라서, 이것이

어셈블리 루틴이 되기 위해선 반드시 f 가 아닌 _f 로 라벨이 붙어야만 한다. 리눅스

gcc 컴파일러는 어떠한 문자도 붙이지 않는다. 리눅스 ELF 실행 가능 파일들의 경우,

C 함수 f 에 대해 그냥 f 라는 이름의 라벨을 사용한다. 그러나 DJGPP 의 gcc 는 _ 를

붙인다.어셈블리골격프로그램(그림 1.7)을보면메인루틴의라벨은 _asm_main로
되어있음을주목해라.

4.7.3 인자전달하기

C호출규약의경우,스택에푸시된함수들의인자들은함수호출시역순으로나타

나게된다.

예를들어다음과같은 C문장을생각해보자 :printf("x = %d\n",x);그림 4.11

는이것이어떻게컴파일될지보여준다. (동일한 NASM형식으로나타남)그림 4.12

은 printf의처음부분을실행한후의스택의모습을보여준다. printf함수는 C 라

이브러리의 함수중하나로, 가변개수의인자들을가질수있다. C 호출규약의 경우

이러한형식의함수들을작성할수있다.C 에서 어셈블리를 이용하여

임의의개수의인자를받는과

정을 만들어 줄 필요는 없다.

stdarg.h헤더파일은위과정

을손쉽게할수있는메크로를

제공해준다.다른 C언어책을

참조하면된다.

형식문자열의주소가마지막에푸시되므로,이것의스택에서의위치는언제나얼

마나많은수의인자들이푸시되느냐에상관없이 EBP+8이된다. printf코드는형식

문자열을보아서알마나많은수의인자들이전해졌는지알아내어,스택을찾아볼수

있다.

당연하게도,printf("x = %d\n")와같이실수를한다면printf코드는 [EBP+12]
에위치한더블워드값을출력할것이다.그러나,이값은 x의값이아니다!

4.7.4 지역변수의주소계산하기

data나 bss세그먼트에정의된라벨의주소를찾는것은간단하다.기본적으로링

커가이러한작업을한다.그러나,직관적으로스택에저장된지역변수(혹은인자)의

주소를계산하는것은어렵다.그러나,우리가서브루틴들을호출시에이작업을자주

하게된다.아래와같이변수(x라하자)의주소를함수(이를 foo라한다)에전달하는

경우를살펴보자.만일 x가 EBP− 8의스택에위치해있다면,우리는아래와같이그냥

사용할수없다.

mov eax, ebp - 8

왜냐하면 MOV명령을통해 EAX로저장되는값이계산되어야만하기때문이다. (이

명령의피연산자는반드시상수여야한다.) 그러나, 위와같은연산을하는명령이있

다.이는 LEA (Load Effective Address 의약자)이다. 아래의코드는 EAX 로 x의주소를

계산하여집어넣는다.

lea eax, [ebp - 8]

이제 EAX 는 x 의 주소를 가지고, foo 를 호출시에 스택에 푸시될 수 있다. 종종

위명령을 [EBP−8] 에있는데이터를읽어들이는것으로착각하는데,이것은사실이

아니다. LEA명령은절대로메모리를읽어들이지않는다. 이는오직다른명령이읽어

들일 주소값을 계산하고, 이를 첫 번째 레지스터 피연산자에 저장할 뿐 이다. 이것이

메모리를읽어들이지않기때문에메모리크기를지정(e.g dword) 할필요가없다.

4.7.5 리턴값

void C함수가아닌것들은모두값을반환한다. C호출규약은이를어떻게해야할

지정해놓았다.리턴값은레지스터를통해서전달된다.모든정수형 (char, int, enum,
etc.)리턴값들은 EAX레지스터에저장되어리턴된다.만일리턴값이32비트보다작을

경우 32 비트로확장되어 EAX 에저장된다. 64 비트값들은 EDX:EAX 레지스터쌍에

저장된다.포인터값또한 EAX에저장된다.부동소수점값들의경우,수치부프로세서

의 ST0 레지스터에저장된다. (이레지스터에대해선,부동소수점장에서다루겠다.)

4.7.6 다른호출규약

위에설명한것들은모두 80x86 C 컴파일러에의해지원되는표준 C 호출규약이

였다.보통컴파일러들은다른형식의호출규약들도지원한다.어셈블리언어와다른

언어를같이이용한다면,컴파일러가무슨호출규약을이용하는지아는것이매우중

요하다.많은경우기본적으로는표준호출규약을이용한다.그러나,모든컴파일러가

그러는 것은 아니다. 4) 여러 종류의 호출 규약을 사용할 수 있는 컴파일러들의 경우,

커맨드라인에서스위치를이용하여기본값으로무슨호출규약을이용할지설정할

수있다.또한, C의문법을확장하여개개의함수를다른호출규약을이용하여컴파일

할수도있다.그러나,이러한확장은표준화되어있지않으며,컴파일러마다다를수

있다.

GCC컴파일러는여러종류의호출규약을지원한다.함수의호출규약은__attribute__
.확장을이용하여개별로지정할수있다.예를들어서,표준호출규약을이용하는 f란
이름의1개의 int인자를가지는 void함수를선언하려면,함수의원형에대해아래와

같이이용하면된다.

void f ( int ) __attribute__((cdecl ));

GCC 는 또한 표준 호출(standard call) 규약 을 지원한다. stdcall 로 정의하려면

cdecl을 stdcall로바꾸면된다. stdcall과 cdecl의차이점은 stdcall의경우

서브루틴이스택으로부터인자를제거해야된다는점이다. (이는 Pascal 호출규약과

유사하다)따라서, stdcall호출규약은오직고정된수의인자를가지는함수들에게

서만사용될수있다. (i.e 이는 printf나 scanf와는다르다)

4) Watcom C 컴파일러는표준호출규약을기본값으로이용하지않는예이다. 자세한내용은 Watcom
을위한예제코드를참조해라.

GCC 는또한 regparm라는 attribute 를지원하여컴파일러로하여금스택을이용

하지 않고 레지스터를 통해 최대 3 개의 정수 인자를 전달하게 한다. 이러한 형식의

최적화는많은컴파일러들이이용하고있다.

볼랜드와 마이크로소프트는 호출 규약을 선언할 때 동일한 문법을 사용한다. 이

들은C에 __cdecl와 __stdcall키워드를추가하였다.이키워드들은함수의원형의

이름바로앞에써주면된다.예를들어위의함수 f는볼랜드나마이크로소프트사의

경우아래와같이정의된다.

void __cdecl f ( int );

각각의호출규약에는장점과단점이있다. cdecl의가장큰장점으로는이는매우

단순하고다루기쉽다. 이는어떠한형식의 C 함수나 C 컴파일러에서든지사용될수

있다. 다른형식의호출규약의경우서브루틴의이식성을떨어뜨린다. 하지만, 이 규

약의가장큰단점은바로다른호출규약들에비해느릴수있고,더많은메모리를차

지한다는것이다. (왜냐하면함수가호출될때마다,코드에저장된인자들을제거하는

명령을수행해야되기때문이다)

stdcall호출규약의장점은 cdecl보다메모리를적게소모한다는점이다. CALL
명령이후에스택을정리할필요가없다.다만,이규약의가장큰단점으로는이규약이

가변개수의인자들을가지는함수들에게사용될수없다는점이다.

레지스터를이용하는호출규약의장점으로는인자를전달하는속도가매우빠르

다는점이다. 그러나, 이러한규약들은좀더복잡하다. 인자들중일부는레지스터에,

나머지는스택에보관되어야하기때문이다.

4.7.7 예제

다음예제는어떻게어셈블리루틴이 C 프로그램에서작동할수있는지보여준다.

(이 프로그램은 그림 1.7 과 같은 어셈블리 뼈대 프로그램이나, driver.c 모듈과 같은

것을사용하지않는다.)

main5.c

1 #include <stdio .h>

2 /* 어셈블리루틴의원형 */

3 void calc_sum( int , int * ) __attribute__((cdecl ));
4

5 int main( void )

6 {

7 int n, sum;

8

9 printf (”Sum integers up to : ” );

10 scanf(”%d”, &n);

11 calc_sum(n, &sum);

12 printf (”Sum is %d\n”, sum);

13 return 0;

14 }

main5.c

sub5.asm
1 ; 서브루틴 _calc_sum
2 ; 1 부터 n 까지 정수들의 합을 구한다.

3 ; 인자:
4 ; n - 어디까지 합할 지 저장 ([ebp + 8] 에 위치)

5 ; sump - sum 을 저장할 공간에 대한 int 포인터([ebp + 12] 에 위치)

6 ; 의사 C 코드:
7 ; void calc_sum( int n, int * sump )

8 ; {

9 ; int i, sum = 0;
10 ; for( i=1; i <= n; i++ )

11 ; sum += i;
12 ; *sump = sum;
13 ; }

14

15 segment .text
16 global _calc_sum
17 ;
18 ; 지역변수:
19 ; [ebp-4] 에 위치한 합

20 _calc_sum:
21 enter 4,0 ; sum 을 위한 스택 공간을 할당

22 push ebx ; 매우 중요하다!

23

24 mov dword [ebp-4],0 ; sum = 0

25 dump_stack 1, 2, 4 ; ebp-8 에서 ebp+16 까지 스택을 출력

26 mov ecx, 1 ; ecx 는 의사코드에서의 i
27 for_loop:
28 cmp ecx, [ebp+8] ; cmp i 와 n
29 jnle end_for ; if i <= n 가 아니면, 종료

30

31 add [ebp-4], ecx ; sum += i
32 inc ecx
33 jmp short for_loop
34

35 end_for:
36 mov ebx, [ebp+12] ; ebx = sump
37 mov eax, [ebp-4] ; eax = sum
38 mov [ebx], eax
39

40 pop ebx ; restore ebx
41 leave

Sum integers up to: 10

Stack Dump # 1

EBP = BFFFFB70 ESP = BFFFFB68
+16 BFFFFB80 080499EC
+12 BFFFFB7C BFFFFB80
+8 BFFFFB78 0000000A
+4 BFFFFB74 08048501

+0 BFFFFB70 BFFFFB88
-4 BFFFFB6C 00000000

-8 BFFFFB68 4010648C
Sum is 55

그림 4.13: sub5 프로그램의실행예제

42 ret sub5.asm

왜 sub5.asm의 22 행이중요할까? 왜냐하면 C 호출규약은 EBX 의값이바뀌면

안되기때문이다.만일, 이처리를해주지않는다면프로그램이제대로동작하지않을

가능성이매우크다.

25행은 dump_stack매크로가어떻게작동하는지보여주고있다.첫번째인자는

단지정수라벨,두번째와세번째인자는출력할 EBP의범위를설정하는것임을기억

하자.그림 4.13은프로그램을실행한모습을보여준다.이덤프에서,합을저장한곳의

주소가 BFFFFB80 (EBP + 12 에위치) 이고, 어디까지합할지기억한값은 0000000A
(EBP + 8에위치), 루틴의리턴주소는08048501(EBP + 4에위치), 저장된 EBP의값

은 BFFFFB88(EBP에위치), 지역변수의값은0 (EBP - 4에위치), 마지막으로저장된

EBX 의값은 4010648C(EBP - 8 에위치)임을알수있다.

calc_sum함수는합을저장하는것을포인터인자로받아들이는대신에합을리턴

값으로출력하게수정할수있다.합이정수값이기때문에그합은 EAX에저장되어야

한다. main5.c의 11행은아래와같이바뀐다 :

sum = calc_sum(n);

또한, calc_sum의원형또한변경되어야한다.아래는수정된어셈블리코드이다 :

sub6.asm
1 ; 서브루틴 _calc_sum
2 ; 1 부터 n 까지 정수들의 합을 구한다.

3 ; 인자:
4 ; n - 어디까지 합할 지 저장 ([ebp + 8] 에 위치)

5 ; 리턴값:
6 ; 합한 값

7 ; 의사 C 코드:
8 ; int calc_sum( int n )

9 ; {

10 ; int i, sum = 0;
11 ; for( i=1; i <= n; i++ )

12 ; sum += i;



C 와소통하기 77

1 segment .data
2 format db "%d", 0

3

4 segment .text
5 ...
6 lea eax, [ebp-16]
7 push eax
8 push dword format
9 call _scanf

10 add esp, 8

11 ...

그림 4.14: 어셈블리에서 scanf호출하기

13 ; return sum;
14 ; }

15 segment .text
16 global _calc_sum
17 ;
18 ; local variable:
19 ; sum at [ebp-4]
20 _calc_sum:
21 enter 4,0 ; 합을 저장할 공간을 스택에 할당

22

23 mov dword [ebp-4],0 ; sum = 0

24 mov ecx, 1 ; ecx 는 의사코드에서의 i
25 for_loop:
26 cmp ecx, [ebp+8] ; cmp i 와 n
27 jnle end_for ; if i <= n 가 아니면, 종료

28

29 add [ebp-4], ecx ; sum += i
30 inc ecx
31 jmp short for_loop
32

33 end_for:
34 mov eax, [ebp-4] ; eax = sum
35

36 leave
37 ret sub6.asm

4.7.8 어셈블리에서 C함수를호출하기

C 와 어셈블리를 같이 이용하는데 있어서 가장 큰 장점은 바로 어셈블리 코드를

통해서 C라이브러리와유저가작성한함수들에접근이가능하다는점이다.예를들어,

만약 scanf함수를호출하여키보드로부터정수를입력받고싶다고하자.그림 4.14

는 위 작업을 하는 코드를 보여준다. scanf 는 C 호출 규약을 지킨다는 점을 잊지 말

아야한다. 이말은즉슨, 이는 EBX, ESI, EDI 레지스터의값을저장하지만, EAX, ECX,
EDX레지스터의값은변경될수있다는점이다.사실, EAX는 scanf호출의리턴값을

저장하기때문에값이반드시바뀌게된다. C 코드를함께이용하는어셈블리코드들

의예를보고싶다면 asm_io.obj를만들기이용하였던 asm_io.asm의코드를한번

봐보면좋다.

4.8 재진입및재귀서브프로그램

재진입서브프로그램(reetrant) 는다음의조건들을반드시만족해야한다.

• 이는 어떠한 코드 명령도 수정하면 안된다. 고급 언어에선 조금 어렵겠지만, 어

셈블리에서는자기자신의코드를수정하지않는일은어렵지않다.예를들어 :

mov word [cs:$+7], 5 ; 5 를 7 바이트 앞에 있는 워드에 복사

add ax, 2 ; 이전의 문장은 2 에서 5 로 바뀐다.

위코드는실제모드에서작동한다.그러나보호모드에서는코드세그먼트가오직

읽기전용(Read-Only)이다.따라서,위첫번째행이실행된다면보호모드시스템

에선프로그램이중단될것이다.이러한형식의프로그래밍은여러이유에서좋

지않다.이는혼란스러울뿐더러,유지,보수가어렵고,코드공유(code sharing)
를할수없다. (아래참고)

• 이는전역데이터를수정하면안된다. (예를들어 data나 bss세그먼트에들어

있는데이터)모든변수들은스택에보관된다.

재진입코드를쓰면몇가지좋은점들이있다.

• 재진입프로그램은재귀적으로호출가능하다.

• 재진입프로그램은다수의프로세스들에의해공유가능하다.많은수의멀티태스

킹운영체제에선, 하나의프로그램에서작동되는여러개의인스턴스(instance)
가있을때,오직코드의한부분만이메모리에상주한다.공유되는라이브러리와

DLL (동적 링크 라이브러리(Dynamic Link Libraries)) 또한 이 아이디어를 이용

한다.

• 재진입프로그램은다중쓰레드(multi-threaded)에서더잘작동된다. 5) programs.
Windows 9x/NT그리고대부분의UNIX꼴의운영체제(Solaris, Linux, etc.)가다

중쓰레드프로그램을지원한다.

4.8.1 재귀서브프로그램

이러한형식의서브프로그램들은자기자신을호출한다.재귀호출은직접(direct)
혹은간접(indirect)호출이될수있다.직접재귀호출은 foo라는서브프로그램이 foo
내부에서스스로호출했을때를일컫는다. 간접재귀호출은서브프로그램이자기자

5) 다중쓰레드프로그램은실행시여러개의쓰레드를가진다.이말은,프로그램자체가멀티태스킹을
한다는뜻이다.

1 ; finds n!
2 segment .text
3 global _fact
4 _fact:
5 enter 0,0
6

7 mov eax, [ebp+8] ; eax = n
8 cmp eax, 1

9 jbe term_cond ; 만일 n <= 1 이면 종료

10 dec eax
11 push eax
12 call _fact ; eax = fact(n-1)
13 pop ecx ; eax 에 답 저장

14 mul dword [ebp+8] ; edx:eax = eax * [ebp+8]
15 jmp short end_fact
16 term_cond:
17 mov eax, 1

18 end_fact:
19 leave
20 ret

그림 4.15: 재귀팩토리얼함수

.

n=3 프레임

n=2 프레임

n=1 프레임

n(3)
리턴주소

저장된 EBP
n(2)

리턴주소

저장된 EBP
n(1)

리턴주소

저장된 EBP
그림 4.16: 팩토리얼함수의스택프레임

신으로부터직접적으로호출되지않았지만다른서브프로그램이호출하기에호출된

것을의미한다.예를들면 foo는 bar을호출할수있고, bar은 foo를호출할수있는

경우이다.

재귀 서브프로그램들은 반드시 종료 조건(termination condition) 이 있어야 한다.

만약위조건이참이라면더이상의재귀호출은만들어지지않는다.만약재귀루틴이

종료조건이없거나,조건이절대로참이될수없다면이는스스로재귀를무한히반복

할것이다. (마치무한루프처럼)

그림 4.15 은 재귀적으로 팩토리얼을 계산하는 함수를 보여준다. 이는 C 를 통해

아래와같이호출될수있다 :

x = fact (3); /* önd 3! */

1 void f ( int x )

2 {

3 int i ;
4 for( i=0; i < x ; i++ ) {

5 printf (”%d\n”, i );
6 f ( i );
7 }

8 }

그림 4.17: 또다른예제 (C 버전)

그림 4.16 은마지막으로호출될때의스택의모습을보여준다.

그림 4.17 와 4.18 은각각 C 와어셈블리를이용한좀더복잡한예제를보여준다.

f(3)의결과는무엇일까? ENTER명령이새로운 i를각재귀호출마다스택에생성한

다는점을주목해라. 따라서, 각각의 f의재귀인스턴스는독립적인변수 i를가지게

된다.이는 i를데이터세그먼트에더블워드로정의하는것과다르다.

4.8.2 C변수저장형식에대해서

C 는변수를저장하기위한방법으로몇가지형식들을지원한다.

전역(global) 이러한변수들은모든함수의외부에서정의되며,고정된메모리위치에

저장되어 있다 (data 나 bss 세그먼트에) 또한, 프로그램 시작 부터 끝날 때 까

지계속존재한다. 기본적으로이들은프로그램의어떠한함수에서든지접근이

가능하다. 그러나, 이들이 static으로선언되어있으면, 오직같은모듈에서의

함수들만이이들에접근가능하다 (i.e.어셈블리형식으로말하자면,라벨이외부

(external) 이아닌내부(internal) 이다).

정적(static) 이들은함수의지역(local) 변수로, static으로선언된다. (불행이도, C
언어는 static을두가지용도로사용한다!)이러한변수들또한고정된메모리

위치에상주한나, (data나 bss) 오직이들이정의되어있는함수에서만직접적

으로접근이가능하다.

자동(automatic) 이는C변수가함수내에서정의될때기본적으로지정되는형식이다.

이러한변수들은함수가호출될때스택에할당되고,함수가리턴될때스택에서

사라진다.따라서이들은고정된메모리위치를갖고있지않는다.

레지스터(register) 이키워드는컴파일러로하여금이변수의데이터를위해레지스

터를 사용하도록 한다. 이는 단순히 요청 이다. 즉, 컴파일러는 이를 반드시 지

켜야 할 필요가 없다. 이 변수의 주소가 프로그램에서 사용된다면 이 요청은 무

시된다 (왜냐하면레지스터들은주소가없기때문이다). 또한, 오직단순한정수

형식들만이레지스터의값이될수있다.구조형은될수없다 ;왜냐하면이들은

레지스터크기에맞지않기때문이다! C 컴파일러는종종보통의자동변수들을

프로그래머에알리지않고레지스터변수로바꾸어주기도한다.

휘발성(volatile) 이키워드는컴파일러에게이변수의값이어무때나바뀔수있음을

알려준다. 이말은컴파일러가이변수가언제수정될지에대한정보를알수없

다는뜻이다.종종컴파일러는변수의값을레지스터에잠시보관해놓고,변수가

1 %define i ebp-4
2 %define x ebp+8 ; 유용한 매크로

3 segment .data
4 format db "%d", 10, 0 ; 10 = '\n'
5 segment .text
6 global _f
7 extern _printf
8 _f:
9 enter 4,0 ; i 를 위한 스택 공간 할당

10

11 mov dword [i], 0 ; i = 0

12 lp:
13 mov eax, [i] ; i < x 인가?

14 cmp eax, [x]
15 jnl quit
16

17 push eax ; printf 호출

18 push format
19 call _printf
20 add esp, 8

21

22 push dword [i] ; f 호출

23 call _f
24 pop eax
25

26 inc dword [i] ; i++
27 jmp short lp
28 quit:
29 leave
30 ret

그림 4.18: 또다른예제 (어셈블리버전)

사용되는 코드에서 이를 대신 사용한다. 그러나, volatile 형식의 변수들에겐

이러한형태의최적화를할수없다.가장흔한 volatile예제로는다중쓰레드프로

그램에서두개의쓰레드에의해값이변경될수있는변수이다.아래의코드를

고려하자

1 x = 10;

2 y = 20;

3 z = x ;

x는다른쓰레드에의해변경될수있고, 1과 3행에서다른쓰레드가 x의값을

변경하여 z가0이안될수있다.그러나, x가휘발성으로선언되지않으면컴파

일러는 x가변경되지않았다고생각하고 z를 10으로바꾸어준다.

volatile은컴파일러가변수를위해레지스터를사용하는것을막을수있다.

배열

5.1 소개

배열(array)은메모리상의연속된데이터들의목록이라볼수있다.배열의개개의

원소들은같은형이며,메모리상에서같은크기의바이트를사용해야한다.따라서이

러한특징때문에배열에서의번째수(index)를통해원소에효율적인접근이가능하다.

특정한원소의주소는다음세가지사실을통해계산될수있다.

• 배열의첫번째원소의주소

• 원소의바이트크기

• 원소의번째수

배열의첫번째원소의인덱스를 0으로생각하면매우편리하다 (C 에서처럼). 첫

번째원소의위치를다른값으로도정의할수있지만이는계산을불편하게한다.

5.1.1 배열정의하기

데이타(data)와 bss 세그먼트에서배열정의하기

값들이초기화된배열을데이타(data)세그먼트에정의하기위해선 db, dw, 등등

의지시어를사용하면된다. NASM 은또한 TIMES라는유용한지시어를제공하는데

이를통해같은문장을손으로여러번쓰는번거러움없이도반복할수있다.

그림 5.1 은이러한것들의몇가지예를보여주고있다.

bss세그먼트에값들이초기화되지않은배열을정의하기위해선, resb, resw, 등
등의지시어를이용하면된다.이러한지시어들은얼마만큼의메모리를지정할지알

려주는피연산자가있다.그림 5.1 는또한이러한형식의정의의예를보여준다.

배열을스택에서의지역변수로정의하기

스택에직접적으로지역배열변수를정의할방법은없다. 이전처럼우리는모든

지역변수들에의해필요한메모리바이트수(배열포함)을계산하여 ESP로부터빼야

한다. (직접적으로빼거나, ENTER명령을이용할수있다)예를들어함수가, 1개의 char
변수, 2개의더블워드정수,그리고1개의 50개의원소를가지는워드배열을필요로

하다면, 이는총 1 + 2 × 4 + 50 × 2 = 109 바이트를필요로하게된다. 그러나, ESP

1 segment .data
2 ; 1,2,... 10 까지 10 개의 더블워드들로 구성된 배열을 정의

3 a1 dd 1, 2, 3, 4, 5, 6, 7, 8, 9, 10

4 ; 10 개의 0 으로 워드 들로 구성된 배열 정의

5 a2 dw 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

6 ; TIMES 를 이용하여 위와 같은 작업 한다.

7 a3 times 10 dw 0

8 ; 200 개의 0 과 100 개의 1 로 구성된 바이트 배열 정의

9 a4 times 200 db 0

10 times 100 db 1

11

12 segment .bss
13 ; 10 개의 원소를 가지는 초기화 안된 배열 정의

14 a5 resd 10

15 ; 100 개의 원소를 가지는 초기화 안된 배열 정의

16 a6 resw 100

그림 5.1: 배열정의하기

EBP - 1 char
사용안됨

EBP - 8 더블워드 1
EBP - 12 더블워드 2 워드

배열

워드
배열 EBP - 100

EBP - 104 더블워드 1
EBP - 108 더블워드 2
EBP - 109 char

EBP - 112 사용안됨

그림 5.2: 스택에서의배열

에서빼지는더블워드값은반드시4의배수 (위경우112)여야한다.왜냐하면 ESP를

항상더블워드경계(Double word boundary)에놓이게해야하기때문이다.우리는109

바이트에배열을두가지방법으로배열할수있다.그림 5.2는그방법들을보여준다.

첫번째배열방법은더블워드가더블워드경계에오도록사용되지않은부분을배열해

메모리접근을향상시켰다.

5.1.2 배열의원소에접근하기

C에서와는달리어셈블리에서는 []연산자가없다.배열의원소에접근하기위해서

는원소의주소가계산되어야만한다.예를들어아래와같은두개의배열을생각하자.

array1 db 5, 4, 3, 2, 1 ; 바이트 배열

array2 dw 5, 4, 3, 2, 1 ; 워드 배열

아래는위배열들을사용하는몇가지예들이다.

1 mov ebx, array1 ; ebx = array1 의 주소

2 mov dx, 0 ; dx 에 합이 저장된다.

3 mov ah, 0 ; ?
4 mov ecx, 5

5 lp:
6 mov al, [ebx] ; al = *ebx
7 add dx, ax ; dx += ax (al 가 아님!)

8 inc ebx ; bx++
9 loop lp

그림 5.3: 배열의원소들의합구하기 (버전 1)

1 mov ebx, array1 ; ebx = array1 의 주소

2 mov dx, 0 ; dx 에 합이 저장

3 mov ecx, 5

4 lp:
5 add dl, [ebx] ; dl += *ebx
6 jnc next ; 캐리가 없다면 next 로 분기

7 inc dh ; dh 1 증가

8 next:
9 inc ebx ; bx++

10 loop lp

그림 5.4: 배열의원소들의합구하기(버전 2)

1 mov al, [array1] ; al = array1[0]
2 mov al, [array1 + 1] ; al = array1[1]
3 mov [array1 + 3], al ; array1[3] = al
4 mov ax, [array2] ; ax = array2[0]
5 mov ax, [array2 + 2] ; ax = array2[1] (array2[2] 가 아님!)

6 mov [array2 + 6], ax ; array2[3] = ax
7 mov ax, [array2 + 1] ; ax = ??

5 행에서워드배열의두번째원소 (원소 1) 에접근하였다. 세번째원소 (원소 2) 가

아니다. 왜냐하면, 워드는 2바이트를차지하기때문에워드배열의다음원소를지정

하기위해선1이아닌2바이트앞으로가야만한다. 7행은첫번째원소의1바이트와

두번째원소의 1바이트를읽어들일것이다. C 에선,컴파일러는포인터의형을보고

얼마만큼의바이트를앞으로가야할지계산을하기때문에프로그래머가직접계산

할필요가없다.그러나어셈블리에선이모든것을프로그래머가고려해야한다.

그림 5.3은 array1의원소들을모두더하는코드를보여준다. 7행에서AX는DX
에더해진다.왜AL이아니냐면,먼저 ADD명령의두개의피연산자의크기는동일해야

한다.두번째로는합이1바이트에들어가기엔너무커지는경우가많다.그러나DX를

이용하여최대65,535까지의합을저장할수있다.그러나, AH또한더해지기때문에

AH 를 3행에서 0으로세트하였다. 1)

1) AH를0으로세팅하는것은AL이부호없는수라고암묵적으로생각하는것이다.만일부호가있다면
6과 7행에 CBW 명령을넣어야한다.

1 mov ebx, array1 ; ebx = array1 의 주소

2 mov dx, 0 ; dx 에 합이 저장

3 mov ecx, 5

4 lp:
5 add dl, [ebx] ; dl += *ebx
6 adc dh, 0 ; dh += 캐리 플래그 + 0

7 inc ebx ; bx++
8 loop lp

그림 5.5: 배열의원소들의합구하기 (버전 3)

그림 5.4 와 5.5 은 합을 구하는 또다른 두 가지 방법들을 보여준다. 이탈릭체로

표시된행은그림 5.3 의 6과 7행을대체한것이다.

5.1.3 좀더향상된간접주소지정

놀랍지않게도, 배열에선간접주소지정이종종이용된다. 가장널리쓰이는형태

로는 :

[ base reg + factor*index reg + constant]

이는

베이스레지스터(base reg) 는 EAX, EBX, ECX, EDX, EBP, ESP, ESI EDI중하나이다.

인수(factor) 는 1, 2, 4 or 8 중하나이다. (1 이면생략된다)

인덱스레지스터(index reg) 는EAX, EBX, ECX, EDX, EBP, ESI, EDI중하나이다. (ESP
가없음을유의해라)

상수(constant) 는32비트상수이다.이상수는라벨혹은라벨식(label expression)이
될수있다.

5.1.4 예제

아래는 함수에 배열을 인자로 전달하는 예제 이다. 이는 array1c.c 프로그램을

(아래에나열됨)드라이버로사용한다. driver.c가아니다.

array1.asm
1 %define ARRAY_SIZE 100

2 %define NEW_LINE 10

3

4 segment .data
5 FirstMsg db "First 10 elements of array", 0

6 Prompt db "Enter index of element to display: ", 0

7 SecondMsg db "Element %d is %d", NEW_LINE, 0

8 ThirdMsg db "Elements 20 through 29 of array", 0

9 InputFormat db "%d", 0

10

11 segment .bss
12 array resd ARRAY_SIZE
13

14 segment .text
15 extern _puts, _printf, _scanf, _dump_line
16 global _asm_main
17 _asm_main:
18 enter 4,0 ; EBP - 4 에 위치한 지역 더블워드 변수

19 push ebx
20 push esi
21

22 ; 배열을 100, 99, 98, 97, ... 로 정의

23

24 mov ecx, ARRAY_SIZE
25 mov ebx, array
26 init_loop:
27 mov [ebx], ecx
28 add ebx, 4

29 loop init_loop
30

31 push dword FirstMsg ; FirstMsg 를 출력

32 call _puts
33 pop ecx
34

35 push dword 10

36 push dword array
37 call _print_array ; 배열의 첫 10 개 원소를 출력

38 add esp, 8

39

40 ; prompt user for element index
41 Prompt_loop:
42 push dword Prompt
43 call _printf
44 pop ecx
45

46 lea eax, [ebp-4] ; eax = 지역 더블워드의 주소

47 push eax
48 push dword InputFormat
49 call _scanf
50 add esp, 8

51 cmp eax, 1 ; eax = scanf 의 리턴값

52 je InputOK
53

54 call _dump_line ; 나머지 행을 덤프 한 후 재시작 한다.

55 jmp Prompt_loop ; 입력값이 올바르지 않다면 분기

56

57 InputOK:
58 mov esi, [ebp-4]
59 push dword [array + 4*esi]
60 push esi
61 push dword SecondMsg ; 원소의 값을 출력

62 call _printf
63 add esp, 12

64

65 push dword ThirdMsg ; 원소 20-29 의 값을 출력

66 call _puts
67 pop ecx
68

69 push dword 10

70 push dword array + 20*4 ; array[20] 의 주소

71 call _print_array
72 add esp, 8

73

74 pop esi
75 pop ebx
76 mov eax, 0 ; C 로 리턴

77 leave
78 ret
79

80 ;
81 ; _print_array 루틴

82 ; C 호출 가능한 루틴으로, 더블워드 배열의 원소들을 부호 있는 정수로

83 ; 출력한다.

84 ; C 원형:
85 ; void print_array( const int * a, int n);
86 ; 인자:
87 ; a - 출력할 배열에 대한 포인터(스택의 ebp+8 에 위치)

88 ; n - 출력할 정수의 개수 (스택의 ebp+12 에 위치)

89

90 segment .data
91 OutputFormat db "%-5d %5d", NEW_LINE, 0

92

93 segment .text
94 global _print_array
95 _print_array:
96 enter 0,0
97 push esi
98 push ebx

99

100 xor esi, esi ; esi = 0

101 mov ecx, [ebp+12] ; ecx = n
102 mov ebx, [ebp+8] ; ebx = 배열의 주소

103 print_loop:
104 push ecx ; printf 는 ecx 의 값을 바꾼다!

105

106 push dword [ebx + 4*esi] ; array[esi] 를 푸시

107 push esi
108 push dword OutputFormat
109 call _printf
110 add esp, 12 ; 인자를 제거 (ecx 만 남긴다!)

111

112 inc esi
113 pop ecx
114 loop print_loop
115

116 pop ebx
117 pop esi
118 leave
119 ret array1.asm

array1c.c

1 #include <stdio .h>

2

3 int asm_main( void );

4 void dump_line( void );

5

6 int main()

7 {

8 int ret_status ;

9 ret_status = asm_main();

10 return ret_status ;

11 }

12

13 /*

14 * 함수 dump_line

15 * 입력버퍼로부터현재행에남은모든문자(char)를덤프한다.

16 */

17 void dump_line()

18 {

19 int ch;

20

21 while( (ch = getchar()) != EOF && ch != ’ \n ’ )

22 /* null body*/ ;

23 }

array1c.c

다시보는 LEA 명령어

LEA명령은주소계산뿐만이아니라다른목적으로도사용될수있다.대표적으로

빠른계산을위해서이다.아래와같은경우를보자.

lea ebx, [4*eax + eax]

이는효과적으로 5× EAX를 EBX에저장한다. LEA를통해서 MUL을이용하는것보다

훨씬빠르고쉽게사용할수있다.그러나,우리는대괄호속에들어가있는식이반드시

올바른간접주소지정형태여야함을명심해야한다.따라서,위명령을통해6을곱할

순없다.

5.1.5 다차원배열

다차원배열은우리가이전에이야기한평범한1차원배열과크게다를바가없다.

사실,다차원배열도메모리상에서는 1차원배열과다를바가없다.

2차원배열

가장 단순한 다차원 배열은 2 차원 배열이다. 2 차원 배열은 원소들을 평면 위에

격자형태로그린다. 각각의원소들은두개의수를이용하여구별된다. 첫 번째수는

원소가위치한행을나타내고,두번째수는원소가위치한열을나타낸다.

아래와같이 3행 2열의배열을생각하자

int a [3][2];

C컴파일러는6 (= 2 × 3) 개의정수를위한공간을할당하고,각원소들을아래와

같이나타낸다.

위치 0 1 2 3 4 5

원소 a[0][0] a[0][1] a[1][0] a[1][1] a[2][0] a[2][1]

위표가시사하는바는 a[0][0]이라나타낸원소는6개의원소를가지는1차원배열

의가장처음의원소이다. a[0][1]은그옆에위치하고나머지도주르륵배열된다. 2

차원배열의각행은메모리상에연속된형태로나타난다.한행의마지막원소다음에

는그다음행의첫번째원소가이어서나타나게된다. 이러한형식을배열의행기준

(rowwise) 표현이라하며,이는 C/C++컴파일러가배열을표현하는방법이기도하다.

컴파일러는행기준표현에서 a[i][j]가어디에나타날지어떻게결정할까?이는

i와 j로이루어진간단한공식을계산하면결정할수있다.위의경우 2i+ j 를계산하

면된다.왜이것을계산해야되는지이해하는것은간단하다.왜냐하면각행은2개의

원소로 구성되어 있기 때문이다. i 행의 첫 번째 원소는 2i 위치에 된다. 또한 j 번째

열에 위치하게 되므로 결과적으로 원소의 위치는 j 를 2i 에 더하면 알 수 있게 된다.

1 mov eax, [ebp - 44] ; ebp - 44 는 i 의 위치

2 sal eax, 1 ; i 에 2 를 곱한다.

3 add eax, [ebp - 48] ; j 를 더한다.

4 mov eax, [ebp + 4*eax - 40] ; ebp - 40 는 a[0][0] 의 주소 이다.

5 mov [ebp - 52], eax ; 결과를 x 에 저장 (ebp - 52 에 위치)

그림 5.6: x = a[ i ][ j ] 의어셈블리표현

이를통해 N개의열을가지는배열의경우에이공식이어떻게일반화될수있는지를

유추할수있다. N × i + j. 이때이공식은행의개수에상관이없음을주목하자.

예를들어 gcc 가아래의코드를어떻게컴파일하는지보자. (위에서정의한 a배열

을이용하여):

x = a[ i ][ j ];

그림 5.6 는이것이어셈블리로어떻게나타났는지보여준다. 따라서컴파일러는

코드를아래와같이바꾼다.

x = *(&a[0][0] + 2*i + j );

그리고사실,프로그래머가위와같이써도같은결과가나타난다.

행 기준 표현법에는 딱히 놀랄 만한 점은 없다. 열 기준(columnwise) 표현도 같은

방식으로작동한다.

위치 0 1 2 3 4 5

원소 a[0][0] a[1][0] a[2][0] a[0][1] a[1][1] a[2][1]

열기준표현법에선각열이연속적으로저장된다.원소 [i][j]는 i+3j 에저장된

다.다른언어들 (FORTRAN등)은이열기준표현법을이용한다.이러한사실을아는

것은다른언어들과함께작업할때중요하다.

3이상의차원의배열

2보다큰크기의차원을갖는배열들에대해서도같은논리가적용된다.예를들어

3차원배열을생각해보면 :

int b[4][3][2];

이배열은아마네개의 [3][2]배열이연속적으로메모리에저장된형태로나타나

게될것이다.아래테이블은배열이어떻게나타는지보여준다.

위치 0 1 2 3 4 5

원소 b[0][0][0] b[0][0][1] b[0][1][0] b[0][1][1] b[0][2][0] b[0][2][1]

위치 6 7 8 9 10 11

원소 b[1][0][0] b[1][0][1] b[1][1][0] b[1][1][1] b[1][2][0] b[1][2][1]

b[i][j][k]의위치를계산하는공식은 6i +2j + k 가된다. 6은 [3][2]배열의크기

로의해결정된다. 보통 a[L][M][N]으로정의된배열의원소 a[i][j][k]의위치는

M × N × i + N × j + k 이된다.이때,첫번째차원 (L) 이공식에나타나지않음을

주목해라.

더 높은 차원의 배열들의 경우에도 마찬가지로 위와 같은 방법을 적용할 수 있다.

예를들어 n 차원배열이 D1 부터 Dn 까지의차원을가진다고할때, i1 부터 in 을나

타난원소의위치는다음과같은공식으로계산된다 :

D2 × D3 · · · × Dn × i1 + D3 × D4 · · · × Dn × i2 + · · · + Dn × in−1 + in

또는간단히표현하여,

n∑
j=1

(
n∏

k=j+1

Dk

)
ij

이는 여러분들이 저자가 물리

를전공했다고이야기할수있

는이유이다.

첫번째차원 D1 은공식에서나타나지않는다.

열기준표현에서는일반화된공식은아래와같이나타난다.

i1 + D1 × i2 + · · ·+ D1 ×D2 × · · · ×Dn−2 × in−1 + D1 ×D2 × · · · ×Dn−1 × in

또는간단히표현하여

n∑
j=1

(
j−1∏
k=1

Dk

)
ij

이경우마지막차원인 Dn 은공식에서나타나지않는다.

다차원배열을 C에서인자로전달하기

다차원 배열의 행 기준 표현은 C 프로그래밍에서 직접적인 영향이 있다. 일 차원

배열의 경우 원소가 메모리 상에 어디에 있는지 계산 할 때 배열의 크기는 필요하지

않다. 그러나 다차원 배열들의 경우 이는 사실이 아니다. 이러한 배열들의 원소들에

접근하기 위해선 컴파일러는 첫 번째 차원을 제외한 나머지 모든 차원들을 알아야만

한다.이는다차원배열을인자로가지는함수의원형을만들때고려해야할부분이다.

아래의코드는컴파일되지않는다.

void f ( int a[ ][ ] ); /* 차원에대한정보가없다 */

그러나아래의코드는컴파일된다.

void f ( int a[ ][2] );

또한2개의열을가지는배열은이함수에인자로전달될수있다.일차원의크기는

필요하지않는다. 2)

아래와같은함수의원형과혼돈하면안된다.

void f ( int * a[ ] );

이는정수포인터들에대한일차원배열을정의한다 (이는이차원배열처럼행동

하는배열의배열들을만드는데사용할수있다.).

2) 물론필요하다면크기를지정할수는있지만,컴파일러에의해무시된다.

LODSB AL = [DS:ESI]
ESI = ESI ± 1

STOSB [ES:EDI] = AL
EDI = EDI ± 1

LODSW AX = [DS:ESI]
ESI = ESI ± 2

STOSW [ES:EDI] = AX
EDI = EDI ± 2

LODSD EAX = [DS:ESI]
ESI = ESI ± 4

STOSD [ES:EDI] = EAX
EDI = EDI ± 4

그림 5.7: 문자열읽기쓰기명령

1 segment .data
2 array1 dd 1, 2, 3, 4, 5, 6, 7, 8, 9, 10

3

4 segment .bss
5 array2 resd 10

6

7 segment .text
8 cld ; 이 부분을 잊지 말것!

9 mov esi, array1
10 mov edi, array2
11 mov ecx, 10

12 lp:
13 lodsd
14 stosd
15 loop lp

그림 5.8: 결과를불러오고저장한다.

좀더높은차원의배열들의경우나머지차원들이반드시필요하나첫번째차원만

예외적으로써줄필요가없다.예를들어, 4차원배열인자는아래와같다.

void f ( int a[ ][4][3][2] );

5.2 배열/문자열명령

80x86 계열의 프로세서들은 배열들에게만 적용되는 몇 가지 명령들을 지원한다.

이러한명령들은문자열명령(string instruction) 이라부른다.이들은인덱스레지스터

(ESI, EDI) 를사용하여작업을실행한후,자동적으로하나또는인덱스레지스터모두

의 값을 1 증가 시키거나 감소시킨다. 플래그 레지스터의 방향 플래그(Direction Flag,
DF) 는 이인덱스레지스터의값이증가되야할지감소되야할지결정한다. 이방향

플래그의값을수정하는명령은두개가있다.

CLD 는방향플래그를 0으로한다.이상태에선인덱스레지스터들이증가된다.

STD 는방향플래그를 1로한다.이상태에선인덱스레지스터들이감소된다.

80x86 프로그래밍시방향플래그를올바른상태에세팅해놓지않는경우가매우

자주발생한다.이는코드가대부분의경우에작동되나 (방향플래그가올바른상태에

있을때)가끔씩프로그램이올바르게작동되지않는경우가발생하여버그를찾아내기

힘들게한다.

MOVSB byte [ES:EDI] = byte [DS:ESI]
ESI = ESI ± 1

EDI = EDI ± 1

MOVSW word [ES:EDI] = word [DS:ESI]
ESI = ESI ± 2

EDI = EDI ± 2

MOVSD dword [ES:EDI] = dword [DS:ESI]
ESI = ESI ± 4

EDI = EDI ± 4

그림 5.9: 메모리이동문자열명령

1 segment .bss
2 array resd 10

3

4 segment .text
5 cld ; 이를 잊지 마세요!

6 mov edi, array
7 mov ecx, 10

8 xor eax, eax
9 rep stosd

그림 5.10: 배열의모든원소를 0으로하는예제

5.2.1 메모리에쓰고읽기

단순한문자열명령은메모리에쓰거나읽거나, 아니면둘다수행한다. 이는바이

트나워드,더블워드를메모리에쓰고읽을수있다.그림 5.7은이러한명령을수행하

는짧은의사코드가나와있다. 이 코드에몇가지주목할부분들이있다. 먼저 ESI 는
읽기에 사용되고, EDI 는 쓰기에 사용된다. 이는 SI 가 소스 인덱스(Source Index), DI
가 데스티네이션인덱스(Destination Index) 의약자로사용되어있음을상기하면편하

다. 또한, 레지스터가 가지는 데이터는 고정되어 있다. (이는 AL, AX, EAX 중 하나)

마지막으로저장명령은어떠한세그먼트에쓰기를할지에대해 DS 가아닌 ES 에의

해결정된다는사실이다.보호모드프로그래밍에서이는큰문제가아니다.왜냐하면

오직 한 개의 데이타 세그먼트가 있고, ES 는 반드시 이를 참조하기 위해 자동적으로

초기화되어야하기때문이다. (DS와같이)그러나,실제모드프로그래밍에선올바른

세그먼트셀렉터값 3) 으로 ES를초기화하는것이매우중요하다.그림 5.8는이러한

명령을사용하여배열을다른것에복사하는예를보여준다.

LODSx명령와 STOSx명령을같이 (그림 5.8 의 13과 14행에서와같이)이용하는

경우는매우흔하다. 사실상, 이는한번의 MOVSx문자열명령을통해실행될수있다.

그림 5.9은이러한명령들이실행하는작업들을설명한다.그림 5.8의 13과 14행은

같은작업을하는단일 MOVSD명령으로바뀔수도있다.유일한차이점은 EAX레지스

터가루프전체에서사용되느냐사용되지않느냐이다.

3) 또다른골치아픈문제로는하나의 MOV명령을이용하여DS레지스터의값을 ES레지스터에직접적으
로대입할수없기때문이다.그대신에 DS의값은범용레지스터 (AX등)에대입된후,그값을다시
ES 에대입해야만한다.따라서두번의 MOV 연산이필요하다.

CMPSB 바이트 [DS:ESI] 와 바이트 [ES:EDI] 를 비교

ESI = ESI ± 1

EDI = EDI ± 1

CMPSW 워드 [DS:ESI] 와 워드 [ES:EDI] 를 비교

ESI = ESI ± 2

EDI = EDI ± 2

CMPSD 더블워드 [DS:ESI] 와 더블워드 [ES:EDI] 를 비교

ESI = ESI ± 4

EDI = EDI ± 4

SCASB AL 과 [ES:EDI] 를 비교

EDI ± 1

SCASW AX 와 [ES:EDI] 를 비교

EDI ± 2

SCASD AX 와 [ES:EDI] 를 비교

EDI ± 4

그림 5.11: 문자열비교명령들

5.2.2 REP 명령접두어

80x86계열은특별한명령접두어 4) REP를이용하여,문자열명령위에사용될수

있다. 이 접두어는 CPU 로 하여금 특정한 횟수 동안 아래의 문자열 명령을 반복하게

한다. ECX 레지스터는반복횟수를세는데사용된다 (LOOP명령에서처럼). REP접두

어를이용하여그림 5.8 ( 12에서 15행)의루프를아래와같이하나의문장으로바꿀

수있다.

rep movsd

그림 5.10 는배열의원소들을모두 0으로바꾸는또다른예제를보여준다.

5.2.3 비교문자열명령

그림 5.11 은메모리와다른메모리혹은레지스터와비교하는몇가지새로운문

자열 명령들을 보여준다. 이들은 배열을 비교하거나 검색하는데 유용하게 사용할 수

있다.이명령들은플래그레지스터를 CMP명령과같이세트할수있다. CMPSx 명령들

은대응되는메모리와비교를수행하고 SCASx 는특정한값을메모리에서검색한다.

그림 5.12는더블워드배열에서12를검색하는짧은코드르보여준다. 10행에서

의 SCASD명령은찾고자하는값이찾아졌을때에도 EDI에4를더한다.따라서,따라서

배열에서 12가보관된주소를구하고자한다면 16행에서수행한것처럼 EDI 에서 4

를빼야한다.

5.2.4 REPx 명령접두어

REP- 처럼문자열비교명령들과함께사용될수있는명령접두어들은몇가지더

있다.그림 5.13는두개의새로운접두어와이들의역할을나타냈다. REPE와 REPZ는
단순히동의어관계이다. (REPNE와 REPNZ의관계도마찬가지) 만일반복된문자열

4) 명령 접두어는명령이아니다, 이는문자열명령앞에붙는 특별한 바이트로, 명령의작업을 바꾸어
주는 역할을 한다. 다른 접두어들의 경우 메모리 접근에 대한 세그먼트 기본 설정들을 변경하는데
사용할수있다.

1 segment .bss
2 array resd 100

3

4 segment .text
5 cld
6 mov edi, array ; 배열의 시작에 대한 포인터

7 mov ecx, 100 ; 원소의 수

8 mov eax, 12 ; 검색할 수

9 lp:
10 scasd
11 je found
12 loop lp
13 ; 찾지 못할 경우 수행할 코드

14 jmp onward
15 found:
16 sub edi, 4 ; edi 는 이제 배열의 12 를 가리킨다.

17 ; code to perform if found
18 onward:

그림 5.12: 검사프로그램

REPE, REPZ Z플래그가세트되어있을동안이나,최대 ECX만큼시행할때까지
명령을반복한다.

REPNE, REPNZ Z 플래그가 클리어 되어 있을 동안이나, 최대 ECX 만큼 시행할 때
까지명령을반복한다.

그림 5.13: REPx명령접두어

비교 명령이 비교 결과 때문에 멈췄을 경우에도 인덱스 레지스터 혹은 레지스터들은

증가되고, ECX 는감소된다.그러나플래그레지스터는반복이종료되었을때의상태

를유지하고있다.따라서, Z플래그를이용하여반복된비교명령이비교결과때문에, 왜 반복되는 비교 문장 뒤에

ECX 의 값이 0 인지 확인하는

문장을넣지않았을까?

혹은 ECX 가 0이되서종료되었는지구분할수있게된다.

그림 5.14 는두메모리블록의값이같은지확인하는예제코드이다. 7 행에서의

JE는이전의명령에서결과를확인하는데사용된다.만일반복된비교명령이두개의

서로 다른 바이트들을 찾았기 때문에 비교 명령이 중단되었으면 Z 플래그는 계속 클

리어되어있을것이고분기는일어나지않는다.그러나 ECX 가 0이되었기에비교가

중단되었다면 Z 플래그는세트되어있으므로, equal라벨로분기하게될것이다.

5.2.5 예제

이단원은문자열관련작업들을수행하는몇가지함수들을다룬다. 이들은모두

어셈블리로작성되어있으며이함수들은대표적인 C 라이브러리함수들과동일한작

업을한다.

memory.asm
1 global _asm_copy, _asm_find, _asm_strlen, _asm_strcpy
2

3 segment .text

1 segment .text
2 cld
3 mov esi, block1 ; 첫 번째 블록의 주소

4 mov edi, block2 ; 두 번째 블록의 주소

5 mov ecx, size ; 바이트로 나타낸 블록의 크기

6 repe cmpsb ; Z 플래그가 세트 될 동안 반복

7 je equal ; Z 가 세트 된다면 두 블록은 같다.

8 ; 두 블록이 같지 않다면 수행할 문장

9 jmp onward
10 equal:
11 ; 두 블록이 같다면 수행할 문장

12 onward:

그림 5.14: 메모리블록을비교하기

4 ; _asm_copy 함수

5 ; 메모리 블록을 복사한다.

6 ; C원형

7 ; void asm_copy( void * dest, const void * src, unsigned sz);
8 ; 인자:
9 ; dest - 복사 될 것이 저장될 버퍼를 가리키는 포인터

10 ; src - 복사 될 버퍼를 가리키는 포인터

11 ; sz - 복사 할 바이트 수

12

13 ; 다음으로 몇 가지 유용한 심볼들을 정의하였다.

14

15 %define dest [ebp+8]
16 %define src [ebp+12]
17 %define sz [ebp+16]
18 _asm_copy:
19 enter 0, 0

20 push esi
21 push edi
22

23 mov esi, src ; esi = 복사 될 버퍼의 주소

24 mov edi, dest ; edi = 복사 된 것이 저장될 버퍼의 주소

25 mov ecx, sz ; ecx = 복사 할 바이트 수

26

27 cld ; 방향 플래그를 초기화

28 rep movsb ; movsb 를 ECX 만큼 반복

29

30 pop edi
31 pop esi
32 leave
33 ret

34

35

36 ; _asm_find 함수

37 ; 메모리에서 특정한 바이트를 검색한다.

38 ; void * asm_find( const void * src, char target, unsigned sz);
39 ; 인자:
40 ; src - 검색할 버퍼를 가리키는 포인터

41 ; target - 검색할 바이트 값

42 ; sz - 버퍼의 바이트 크기

43 ; 리턴값:
44 ; 만일 target 을 찾았다면 버퍼에서의 첫 번째 target 을 가리키는 포인터가

45 ; 리턴된다.

46 ; 그렇지 않으면 NULL 이 리턴된다.

47 ; 참고: target 은 바이트 값이지만 스택에 더블워드의 형태로 푸시된다.

48 ; 따라서 바이트 값은 하위 8 비트에 저장된다.

49 ;
50 %define src [ebp+8]
51 %define target [ebp+12]
52 %define sz [ebp+16]
53

54 _asm_find:
55 enter 0,0
56 push edi
57

58 mov eax, target ; al 은 검색할 바이트 값을 보관

59 mov edi, src
60 mov ecx, sz
61 cld
62

63 repne scasb ; ECX == 0 혹은 [ES:EDI] == AL 때 까지 검색

64

65 je found_it ; 제로 플래그가 세트되었다면 검색에 성공

66 mov eax, 0 ; 찾지 못했다면 NULL 포인터를 리턴

67 jmp short quit
68 found_it:
69 mov eax, edi
70 dec eax ; 찾았다면 (DI - 1) 을 리턴

71 quit:
72 pop edi
73 leave
74 ret
75

76

77 ; _asm_strlen 함수

78 ; 문자열의 크기를 리턴

79 ; unsigned asm_strlen( const char * );
80 ; 인자:
81 ; src - 문자열을 가리키는 포인터

82 ; 리턴값:
83 ; 문자열 에서의 char 의 수 (마지막 0 은 세지 않는다) (EAX 로 리턴)

84

85 %define src [ebp + 8]

86 _asm_strlen:
87 enter 0,0
88 push edi
89

90 mov edi, src ; edi = 문자열을 가리키는 포인터

91 mov ecx, 0FFFFFFFFh ; ECX 값으로 가능한 가장 큰 값

92 xor al,al ; al = 0

93 cld
94

95 repnz scasb ; 종료 0 을 찾는다.

96

97 ;
98 ; repnz 은 한 단계 더 실행되기 때문에, 그 길이는 FFFFFFFE - ECX,
99 ; 가 아닌 FFFFFFFF - ECX 이다.

100 ;
101 mov eax,0FFFFFFFEh
102 sub eax, ecx ; 길이 = 0FFFFFFFEh - ecx
103

104 pop edi
105 leave
106 ret
107

108 ; _asm_strcpy 함수

109 ; 문자열을 복사한다.

110 ; void asm_strcpy( char * dest, const char * src);
111 ; 인자:
112 ; dest - 복사 된 결과를 저장할 문자열을 가리키는 포인터

113 ; src - 복사 할 문자열을 가리키는 포인터

114 ;
115 %define dest [ebp + 8]

116 %define src [ebp + 12]

117 _asm_strcpy:
118 enter 0,0
119 push esi
120 push edi
121

122 mov edi, dest
123 mov esi, src
124 cld
125 cpy_loop:
126 lodsb ; AL 을 불러오고 & si 증가

127 stosb ; AL 을 저장하고 & di 증가

128 or al, al ; 조건 플래그를 세트 한다.

129 jnz cpy_loop ; 만일 종료 0 을 지나지 않닸다면, 계속 진행

130

131 pop edi
132 pop esi
133 leave
134 ret memory.asm

memex.c

1 #include <stdio .h>

2

3 #deöne STR_SIZE 30

4 /* prototypes */

5

6 void asm_copy( void *, const void *, unsigned ) __attribute__((cdecl ));
7 void * asm_önd( const void *,

8 char target , unsigned ) __attribute__((cdecl ));
9 unsigned asm_strlen( const char * ) __attribute__((cdecl ));

10 void asm_strcpy( char *, const char * ) __attribute__((cdecl ));
11

12 int main()

13 {

14 char st1[STR_SIZE] = ” test string ” ;

15 char st2[STR_SIZE];

16 char * st ;

17 char ch;

18

19 asm_copy(st2, st1 , STR_SIZE); /* 문자열의30문자를복사 */

20 printf (”%s\n”, st2 );

21

22 printf (”Enter a char : ” ); /* 문자열의바이트를찾는다. */

23 scanf(”%c%*[^\n]”, &ch);

24 st = asm_önd(st2, ch, STR_SIZE);

25 if ( st )

26 printf (”Found it : %s\n”, st );

27 else
28 printf (”Not found\n”);

29

30 st1[0] = 0;

31 printf (”Enter string : ” );

32 scanf(”%s”, st1 );

33 printf (”len = %u\n”, asm_strlen(st1 ));

34

35 asm_strcpy( st2 , st1 ); /* 문자열의의미있는데이터를복사 */

36 printf (”%s\n”, st2 );

37

38 return 0;

39 }

memex.c

부동소수점

6.1 부동소수점표현

6.1.1 정수가아닌이진수

우리가첫번째장에서기수법에대해이야기하였을때, 오직정수들에대해서만

이야기 하였다. 그러나, 명백히도 십진법과 같이 다른 진법 체계에서도 정수가 아닌

수를표현할수있다.십진법에선소수점다음으로나타나는숫자들은10에음수승을

취하게된다.

0.123 = 1 × 10−1 + 2 × 10−2 + 3 × 10−3

물론,이진수들에대해서도동일하게적용된다.

0.1012 = 1 × 2−1 + 0 × 2−2 + 1 × 2−3 = 0.625

1 장에서 정수 이진수를 십진수로 바꾸는데 사용했던 방법을 동일하게적용하여

아래와같이정수가아닌이진수를십진수로바꿀수있다.

110.0112 = 4 + 2 + 0.25 + 0.125 = 6.375

십진수를이진수로바꾸는것도크게어렵진않다.보통,십진수를두개의부분으로

나눈다 : 정수 부분과 소수 부분. 정수 부분은 1 장에서 다루었던 방법으로 이진수로

바꾼다.소수부분은아래의설명한방법을통해이진수로변경될수있다.

예를 들어서 각각의 비트를 a, b, c, . . . 로나타낸 이진수를 생각하자. 이는 아래와

같이나타난다 :

0.abcdef . . .

위수에 2를곱한수는아래와같이이진법으로나타난다.

a.bcdef . . .

첫번째비트가이제 1의자리에있음을유의해라. a 를 0 으로바꾼다면

0.bcdef . . .

0.5625 × 2 = 1.125
0.125 × 2 = 0.25
0.25 × 2 = 0.5
0.5 × 2 = 1.0

ĕrst bit = 1
second bit = 0

third bit = 0
fourth bit = 1

그림 6.1: 0.5625 를이진수로바꾸기

0.85 × 2 = 1.7
0.7 × 2 = 1.4
0.4 × 2 = 0.8
0.8 × 2 = 1.6
0.6 × 2 = 1.2
0.2 × 2 = 0.4
0.4 × 2 = 0.8
0.8 × 2 = 1.6

그림 6.2: 0.85 를이진수로바꾸기

그리고 2를다시곱한다면

b.cdef . . .

이제, 두 번째 비트 (b) 가 1 의 자리에 오게 된다. 이 방법을 여러번 반복하여 필요한

비트수만큼까지값을찾을수있다. 그림 6.1 는 0.5625 를 이진수로변환하는예를

보여주고있다.이방법은소수부분의값이 0이될때중단을한다.

또다른예제에선 23.85 를이진수로바꾸었다. 정수부분은이진수로바꾸기가매

우 쉽다 (23 = 101112). 하지만 소수 부분 (0.85)은 어떨까? 그림 6.2 는 소수 부분

을이진수로바꾸는처음몇부분의과정을보여주고있다. 사실, 이수를완전히이진

수로 바꾸기 위해선 무한번의 연산이 필요하다. 이 말은 0.85 는 순환 이진 소수라는

것이다. 1) 위계산에서나타나는숫자들은특정한규칙들이있다.이를잘살펴본다면

0.85 = 0.1101102 임을알수있다.따라서, 23.85 = 10111.1101102 이다.

따라서,위계산결과를통해알수있는점은23.85는유한개의이진비트를사용하

여수를정확히나타낼수없다라는점이다. (이는 1
3
이십진법으로유한개의자리수를

가지는 소수로 정확하게 나타낼 수 없는 점과 같다) 이 장에서는 C 에서의 float 과

double변수들이이진수로어떻게저장되는지살펴볼것이다. 따라서 23.85 와같은

값들은이변수에정확하게저장될수없다.오직 23.85 의근사값만이변수에저장될

1) 이는전혀놀라운일이아닌데, 어떤수는특정한진법을통해표현한다면무한소수가될수있지만
다른진법에서는유한소수가될수있다.예를들어서 1

3
을생각해보면십진법으로표현시3이무한히

반복되지만 3진법으로표현한다면 0.13 과같이깔끔하게나타날수있다.

31 30 23 22 0
s e f

s 부호비트 - 0 = 양수, 1 = 음수
e 편향지수(8비트) =실제지수값 + 7F (십진수로127).지수의값이00과 FF

일때는특별한의미를가진다.
f 가수 - 1. 다음의처음 23비트

그림 6.3: IEEE 단일정밀도

것이다.

하드웨어를단순화하기위해부동소수점수들은일정한형식으로저장된다.이형

식은과학적기수법표현 (다만이진수에서는10의멱수를사용하지않고2의멱수를

사용한다)을사용한다.예를들어23.85, 혹은 10111.11011001100110 . . .2 은다음과

같이저장된다.

1.011111011001100110 . . . × 2100

(지수 (100) 도 이진수로 나타난다.). 정규화 된 부동 소수점 수는 아래의 같은 꼴을

가진다.

1.ssssssssssssssss × 2eeeeeee

이때 1.sssssssssssss 은가수(signiĕcand) 라하고 eeeeeeee 은지수(exponent) 라한

다.

6.1.2 IEEE부동소수점표현

IEEE (전기전자기술자협회, Institute of Electrical and Electronic Engineers) 는부

동소수점수들을저장하기위한구체적인이진형식을만든국제적인조직이다.이형

식은오늘날만들어진대부분 (모든컴퓨터는아니다!) 의컴퓨터에서사용된다. 많은

경우이형식은컴퓨터하드웨어자체에서지원된다.예를들어인텔의수치보조처리기

(numeric coprocessor) 이다. (이는 Pentium 부터 모든 인텔 CPU 들에게 추가되었다)

IEEE는정밀도가다른두개의형식을지원한다.이는단일정밀도(single precision)와
2배정밀도(double precision) 이다.단일정밀도는 C언어의 float변수들이사용하며

2배정밀도는 double변수들이사용한다.

인털의 수치 보조처리기는 또한 더 높은 정밀도를 가지는 확장 정밀도(extended
precision)를 사용한다. 사실, 보조처리기에있는모든데이터들은이정밀도로표현된

다. 보조처리기는이를메모리에저장할때에단일또는 2배정밀도를가지는수들로

자동으로 변환한다. 2) 확장 정밀도는 IEEE 의 Ęoat 이나 double에 비해 다른 형식을

사용하며여기서는이야기하지않도록하겠다.

e = 0 , f = 0 이면 0을나타낸다. (이는정규화될수없다) +0 와
-0 가있음을유의해라.

e = 0 , f 6= 0 는비정규화된수임을나타낸다. 이것은다음단원
에서다룬다.

e = FF , f = 0 는 (∞) 을 나타낸다. 음의 무한과 양의 무한이 있을
수있다.

e = FF , f 6= 0 NaN(Not A Number) 라부르는정의되지않은값을
나타낸다.

표 6.1: f 와 e 의특별한값들

IEEE단일정밀도

단일정밀도부동소수점형식은수를32비트로표현한다.이는대체로십진수로7

자리정도정확하다.부동소수점수들은정수들에비해훨씬복접한형식으로저장된

다.그림 6.3은 IEEE단일정밀도수의기본적인형식을보여준다.이형식에선약간의

이상한 점이 있다. 부동 소수점 수들은 음수를 위해 2 의 보수 표현법을 사용하지 않

는다. 이들은부호있는크기표현(signed magnitude) 을 사용한다. 31 째비트는위에

나와있듯이수의부호를정한다.

이진지수는그값그대로저장되지않는다.그대신7F를더해서23 30번째비트에

더해진다.따라서이편향된지수(biased exponent) 는언제나음수가아니다.

소수 부분은 언제나 정규화 된 가수 부분이라 생각한다 (1.sssssssss 형식으로).

언제나첫번째비트가 1이므로, 1은저장되지않는다. 이를통해비트하나를추가적

으로저장할수있게되어정확성을향상시킬수있다.이아이디어는숨겨진1표현법

(hidden one representation) 이라한다.

그렇다면 23.85 는어떻게저장될까?먼저이수는양수이므로부호비트는0이다.언제나 명심해 둘 것은 프로

그램이 41 BE CC CD 를 어떻

게 해석하느냐에 따라 그 값

이 달라질 수 있다는 사실이

다! 예를 들어 단일 정밀도 부

동 소수점 수라고 생각한다면

이값은23.850000381을나

타나게 된다. 하지만 더블워

드 정수라고 생각한다면 이는

1,103,023,309 를 의미한다.

CPU는어는것이이수에대한

정확한표현인지알수없다.

지수값은4이므로,편향된지수값은 7F+ 4 = 8316 가된다.마지막으로가수부분은

01111101100110011001100 이다. (맨처음의 1은숨겨져있다는사실을명심해라)

이를 모두 합치면 (참고로 이해를 돕기 위해서 부호 비트와 가수 부분은 밑줄이 처져

있고,비트들은 4비트니블들로나뉘어나타냈다)

0 100 0001 1 011 1110 1100 1100 1100 11002 = 41BECCCC16

이는23.85를정확하게표현한것은아니다 (왜냐하면이진법에서순환소수로나타나

기 때문). 만일 위 수를 십진수로 바꾼다면 그 값은 대략 23.849998474 가 될 것이다.

이값은 23.85 와매우가깝지만정확히같지는않다.사실, C 에선 23.85 를위와같이

나타내지않는다. C에선,정확한이진표현에서단일정밀도형식으로변환시,마지막

으로잘라낸비트가 1이므로마지막비트는 1로반올림된다.따라서 23.85 는 41 BE
CC CD로나타나게된다.이는십진수로바꾸면23.850000381이므로좀더23.85 를

가깝게표현할수있다.

-23.85는어떻게저장될까?단순히부호비트만바꾸면된다.따라서C1 BECCCD
가된다. 2의보수표현법을취하지않는다!

e 와 f 의 특정한 조합은 IEEE Ęoat 에서 특별한 의미를 가진다. 표 6.1 는 이러한

2) 일부컴파일러 (예를들어 Borland) 의 long double 형은이확장정밀도를이용한다.그러나대부분
의경우 double 과 long double 형모두 2배정밀도를사용한다. (이는 ANSI C 에서허용된다)

63 62 52 51 0
s e f

그림 6.4: IEEE 2배정밀도

특별한값들을보여준다.무한대는오버플로우(over Ęow)나 0 으로의나눗셈에의해

발생된다.정의되지않은값(NaN) 은올바르지못한연산,예를들면음수의제곱근을

구한다던지,두개의무한대를더할때등등발생된다.

정규화된단일정밀도수들은 1.0× 2−126 (≈ 1.1755× 10−35) 에서 1.11111 . . .×
2127 (≈ 3.4028 × 1035) 까지의값을가질수있다.

비정규화된수

비정규화 된 수(Denormalized number)들은 너무 값이 작아서 정규화 할 수 없는

수들을 나타내기 위해 사용된다. (i.e. 1.0 × 2−126 미만) 예를 들어 1.0012 × 2−129

(≈ 1.6530 × 10−39) 라는수를생각하자. 이수를정규화하기에는지수가너무작다.

그러나, 이수는비정규화된꼴 : 0.010012 × 2−127 로나타낼수있다.이수를저장하

기 위해 편향된 지수는 0 으로 맞춰진다. (표 6.1 참조) 그리고 2−127 와 곱해진소수

부분은맨앞자리의비트를포함한완전한수로나타나게된다.따라서, 1.001× 2−129

는 :
0 000 0000 0 001 0010 0000 0000 0000 0000

IEEE 2배정밀도

IEEE 2배정밀도표현은 64 비트를사용하며, 십진법으로대략 15 자리정도정확

하게나타낼수있다. 그림 6.4 에서나타나듯이기본적인형식은단일정밀도와매우

유사하다.다만편향된지수와가수부분이각각 11비트와 52비트로늘어났다.

편향된지수부분의크기가더커졌으므로두가지변화를야기하게된다.먼저편향

된지수값을구할때진짜지수에 3FF(1023) 을더한다 (단일정밀도의 7F 가아니다).

두 번째로 진짜 지수가 가질 수 있는 값의 범위가 넓어 졌다. 2배 정밀도가 표현할 수

있는수의범위는대략 10−308 to 10308 정도된다.

또한가수부분이더커졌으므로유효숫자의자리수가더늘어나게되었다.

예를들어 23.85 를다시생각해보자.편향된지수는 16진수로 4 + 3FF = 403가
될것이다.따라서 2배정밀도표현으로는 :

0 100 0000 0011 0111 1101 1001 1001 1001 1001 1001 1001 1001 1001 1001 1001 1010

혹은 40 37 D9 99 99 99 99 9A 로나타나게된다. 이수를십진수로변환한다면그값

은 23.8500000000000014 가되어 (무려 12개의 0이있다!) 23.85 를훨씬정밀하게

표현할수있게된다.

2배 정밀도도 단일 정밀도 표현과 같이 특별한 값들이 있다. 3) 비정규화 된 수도

또한매우비슷하다.유일한차이점은비정규화된수는2−127 가아닌 2−1023 가된다.

3) 유일한차이점은무한대와 NaN 의경우인데,편향된지수값이 7FF 가아니라 FF 가된다.

6.2 부동소수점수들의산술연산

컴퓨터에서의부동소수점수들의산술연산은수들이연속된수학에서와는다르다.

수학에서는모든수들이정확한값을가졌다고생각된다.위에서다루었듯이컴퓨터가

다루는 많은 수들은 유한개의 비트를 통해 정확히 나타낼 수 없다. 모든 연산들은 제

한된정밀도를가진다. 이 단원에서의예제들은단순화를위해 8 비트의유효숫자를

가진다고생각한다.

6.2.1 덧셈

두 개의 부동 소수점 수들을 더하기 위해선 지수가 반드시 같아야 한다. 만일, 이

지수들이 같지 않다면 수의 가수를 쉬프트 하여 큰 지수에 맞추어야 한다. 예를 들어

10.375 + 6.34375 = 16.71875 를이진법에서수행한다고하자.

1.0100110 × 23

+ 1.1001011 × 22

이두수들은지수값이같지않으므로가수부분을쉬프트하여두수의지수가같게한

뒤더한다.

1.0100110 × 23

+ 0.1100110 × 23

10.0001100 × 23

이 때 1.1001011 × 22 에 쉬프트 연산을 취하면서 마지막 비트를 없애고 반올림을

하여 0.1100110 × 23 가 되었음을 유의하자. 덧셈의 결과는 10.0001100 × 23 (혹은

1.00001100 × 24) 로 10000.1102 혹은 16.75 와 같이같다. 이는정확한계산결과인

16.71875와같지않다!이는단지덧셈과정에서반올림에의해오차가나타난근사값

에불과하다.

컴퓨터 (혹은계산기)에서의부동소수점연산은언제나근사임을명심해야한다.

수학에서의 법칙들은 컴퓨터에서의 부동 소수점들에 적용될 수 없다. 수학은 언제나

컴퓨터가절대로만들어낼수없는무한대의정밀도를가정한다.예를들어서수학에선

(a + b) − b = a 라말하지만컴퓨터에서는이식이성립하지않을수있다!

6.2.2 뺄셈

뺄셈은 덧셈과 매우 유사하며, 같은 문제를 가지고 있다. 예를 들어서 16.75 −
15.9375 = 0.8125 를생각해보자.

1.0000110 × 24

− 1.1111111 × 23

1.1111111 × 23 를쉬프트하면 1.0000000 × 24 (반올림됨)

1.0000110 × 24

− 1.0000000 × 24

0.0000110 × 24

0.0000110 × 24 = 0.112 = 0.75 이므로원답과일치하지않는다.

6.2.3 곱셈과나눗셈

곱셈의경우유효숫자부분은곱해지고,지수부분은더해진다.예를들어 10.375×
2.5 = 25.9375 을생각해보면

1.0100110 × 23

× 1.0100000 × 21

10100110
+ 10100110

1.10011111000000 × 24

당연하게도실제결과는 8비트로반올림되어

1.1010000 × 24 = 11010.0002 = 26

나눗셈은좀더복잡하지만반올림으로인하여같은문제를가진다.

6.2.4 프로그래밍을위한조언

이단원의핵심적인내용은부동소수점연산이정확하지않다는것이다.프로그래

머는언제나이를고려해야한다. 프로그래머들이가장많이범하는실수로는연산이

정확하다고 가정 하에 부동 소수점 수들을 비교하는 것이다. 예를 들어 f (x) 라는 이

름의함수를고려하자.이함수는복잡한연산을수행한다.우리의프로그램의목적은

이함수의근을찾는것이다. 4) 여러분은아마 x 가근인지확인하기위해아래의식을

이용할것이다.

if ( f (x) == 0.0 )

그러나만일 f (x)가 1 × 10−30 을리턴하면?이는진짜근의훌륭한근사값이다.그러

나,위식은성립하지않는다.아마도 f (x)에서의반올림으로인한오차로인해어떠한

IEEE 부동소수점형식으로표현된 x값도정확히 0을리턴하지않을것이다.

위보다훨씬낳은방법은아래와같다.

if ( fabs( f (x)) < EPS )

EPS는매우작은양수값인매크로로정의되어있다. (예를들면 1× 10−10) 이는 f (x)

가0에매우가까울때성립이된다.통상적으로두개의부동소수점값 (x) 을다른 (y)

값에비교할때아래와같이사용한다 :

if ( fabs(x − y)/fabs(y) < EPS )

4) 함수의근이라하면 f(x) = 0 을만족하는값 x 를말한다.

6.3 수치부프로세서

6.3.1 하드웨어

가장초기의인털프로세서들은부동소수점연산을지원하는하드웨어가없었다.

이말은부동소수점연산을할수없었다는것이아니다. 이는단지부동소수점수들

의 연산을 위해 많은 수의 비-부동 소수점 명령들을 수행해야 했다는 뜻이다. 이러한

초기의시스템에선인텔은부수적인수치부프로세서(math coprocessor)를제공하였다.

이 수치 부프로세서는 부동 소수점 수들의 연산을 위한 소프트웨어 상의 프로시저의

비해훨씬빠른속도로명령을수행하였다 (초기의프로세서의경우대략10배이상!).

8086/8088을 위한 부프로세서는 8087 이라 불리었다. 80286 에선 80287이, 80386

에선80387이부프로세서였다. 80486DX프로세서에선80486에수치부프로세서가

통합되었다. 5) 펜티엄이후 80x86 세대의모든프로세서들에게는수치부프로세서가

통합되었다. 그러나, 독립적으로 떨어져 있었을 때 처럼 프로그램 된다. 심지어 부프

로세서가 없던 초기의 시스템들은 수학 부프로세서를 에뮬레이트 할 수 있는 소프트

웨어가있었다.이에뮬레이터는프로그램이부프로세서명령을실행하면자동적으로

활성화되어소프트웨어프로시저를통해부프로세서가내놓았을같은값을내놓았다.

(비록매우느렸지만)

수치부프로세서는8개의부동소수점레지스터가있다.각각의레지스터는80비

트의데이터를보관한다.부동소수점수들은이레지스터들에언제나80비트확장정

밀도로저장된다.이레지스터들은각각 ST0, ST1, ST2, . . . ST7로이름붙여졌다.부동

소수점레지스터들은주 CPU에서의정수레지스터들과는달리다른용도로사용된다.

부동 소수점 레지스터들은 스택(stack)으로 구성되어 있다. 스택이 후입 선출 (LIFO)
리스트임을상기하자. ST0는언제나스택의최상단의값을가리킨다. 모든새로운수

들은스택의최상단에들어간다.기존의수들은새로운수를위한공간을만들기위해

스택한칸내려간다.

수치부프로세서에는상태레지스터(status register)가있다.이는몇가지의플래그

를가진다.비교연산에서는오직4개의플래그들이사용된다 :C0, C1, C2, C3.이것들의

사용은나중에다룰것이다.

6.3.2 명령

보통의 CPU 명령들과부프로세서의명령들을쉽게구별하기위해모든부프로세

서의연상기호들은 F로시작한다.

데이터를불러오고저장하기

부프로세서의레지스터스택의최상위데이터에값을저장하는몇가지연산들이

있다.

5) 그러나 80486SX 에는통합된부프로세서가없었다. 이를위해독립적인 80487SX 칩이있었다.



수치부프로세서 109

FLD source 는 메모리에서 부동 소수점 값을 불러와 스택 최상단에 저장한다.

source 는단일혹은 2배정밀도값, 아니면부프로세서의레지스터

여야한다.

FILD source 는정수를메모리로부터읽어들어,부동소수점형식으로변환한뒤

그결과를스택최상단에저장한다. source 는워드, 더블워드, 쿼드

워드중하나여야한다.

FLD1 1 을스택최상단에저장한다.

FLDZ 0 을스택최상단에저장한다.
스택으로부터의데이터를메모리에저장하는명령들도몇가지있다.이러한명령

들의일부는스택에서수를팝(pop) (i.e. 제거하다)한후에그값을저장한다.
FST dest 스택의 최상단(ST0) 을 메모리에 저장한다. dest 은 단일 혹은 2배

정밀도수이거나부프로세서레지스터가될수있다.

FSTP dest 은스택의최상단값을 FST와같이메모리에저장한다.그러나수가

저장된 다음에는 그 값은 스택에서 팝 된다. dest 은 단일 혹은 2배

정밀도수이거나부프로세서레지스터가될수있다.

FIST dest 는스택의최상단의값을정수로변환한뒤메모리에저장한다. dest
는 워드 혹은 더블워드가 되어야만 한다. 스택 그 자체로는 바뀌지

않는다.부동소수점수가정수로어떻게변환되느냐넨부프로세서

의제어워드(control word)에몇개의비트값에따라달라진다.이는

특별한 (비-부동소수점)워드레지스터로부프로세서가어떻게작

동할지제어한다.기본적으로제어워드는초기화되어있는데이때

에는수에가장가까운정수로변환되어진다. 그러나, FSTCW (저장

제어워드 Store Control Word) 와 FLDCW (불러오기제어워드, Load
Control Word) 명령들을통해어떻게변환할지를바꿀수있다.

FISTP dest FIST와두가지빼고동일하다.하나는,스택의최상단이팝된다는

것이고,다른하나는 dest 가쿼드워드도될수있다는것이다.
스택자체에서데이터를이동,혹은제거할수있는명령이두개가있다.

FXCH STn 스택에서의 ST0와 STn의값을바꾼다. (이때, n은1부터7까지의

값을가질수있다.)

FFREE STn 레지스터가비었거나미사용중임을표시하여스택에서의레지스터

의값을비운다.

덧셈과뺄셈

각각의 덧셈 명령들은 ST0 과 다른 피연산자의 합을 계산한다. 그 결과는 언제나

부프로세서레지스터에저장된다.
FADD src ST0 += src. src 는어떤부프로세서레지스터이거나메모

리상의단일혹은 2배정밀도수여도상관없다.

FADD dest, ST0 dest += ST0. dest는임의의부프로세서레지스터이다.

FADDP dest또는

FADDP dest, STO
dest += ST0 하고팝한다. dest 는임의의부프로세서레

지스터이다.

FIADD src ST0 += (float) src.정수를 ST0에더한다. src는메모리

상의워드혹은더블워드값이다.

1 segment .bss
2 array resq SIZE
3 sum resq 1

4

5 segment .text
6 mov ecx, SIZE
7 mov esi, array
8 fldz ; ST0 = 0

9 lp:
10 fadd qword [esi] ; ST0 += *(esi)
11 add esi, 8 ; 다음 double 로 이동

12 loop lp
13 fstp qword sum ; sum 에 결과를 저장

그림 6.5: 배열의원소합구하는예제

두개의피연산자에대해덧셈명령은한가지이지만, 뺄셈명령은 2 가지의서로

다른뺄셈명령이존재할수있다. 왜냐하면피연산자의순서가뺄셈에서는중요하기

때문이다. (i.e. a + b = b + a, 이지만 a − b 6= b − a 은 아니다!) 각 명령에 대해

뺄셈을역으로수행하는명령이있다.이역명령들은끝에언제나 R이나 RP가붙는다.

그림 6.5 는 배열의 double 형 원소들을 모두 더하는 코드를 보여주고 있다. 10 에서

13 행 까지 우리는 메모리 피연산의 크기를 확실히 지정해야 한다. 그렇지 않는 다면

어셈블러는메모리피연산자가 Ęoat(더블워드)인지 double(쿼드워드)인자알수없기

때문이다.
FSUB src ST0 -= src. src는임의의부프로세서레지스터거나메모

리상의단일혹은 2배정밀도수이면된다.

FSUBR src ST0 = src - ST0. src는임의의부프로세서레지스터거나

메모리상의단일혹은 2배정밀도수이면된다.

FSUB dest, ST0 dest -= ST0. dest 는임의의부프로세서레지스터이다.

FSUBR dest, ST0 dest = ST0 - dest.dest 는임의의부프로세서레지스터

이다.

FSUBP dest혹은

FSUBP dest, STO
dest -= ST0한후팝한다. dest 는임의의부프로세서레

지스터이다.

FSUBRP dest or
FSUBRP dest, STO

dest = ST0 - dest한후팝한다. dest 는임의의부프로

세서레지스터이다.

FISUB src ST0 -= (float) src.ST0에서정수를뺀다. src는반드시

메모리상의워드혹은더블워드여야한다.

FISUBR src ST0 = (float) src - ST0.ST0에서정수를뺀다. src 는
반드시메모리상의워드혹은더블워드여야한다.

곱셈과나눗셈

곰셉명령은덧셈명령과완벽히동일하다.

FMUL src ST0 *= src. src는임의의부프로세서레지스터거나메모

리상의단일혹은 2배정밀도수이면된다.

FMUL dest, ST0 dest *= ST0.dest 는임의의부프로세서레지스터이다.

FMULP dest혹은

FMULP dest, STO
dest *= ST0한후팝한다. dest 는임의의부프로세서레

지스터이다.

FIMUL src ST0 *= (float) src. ST0에정수를곱한다. src는메모리

상의워드혹은더블워드여야한다.
놀랍지 않게도 나눗셈 명령은 뺄셈 명령과 거의 동일하다. 다만 0 으로 나눈다면

무한대가된다.

FDIV src ST0 /= src. src는임의의부프로세서레지스터거나메모

리상의단일혹은 2배정밀도수이면된다.

FDIVR src ST0 = src / ST0. src는임의의부프로세서레지스터거나

메모리상의단일혹은 2배정밀도수이면된다.

FDIV dest, ST0 dest /= ST0. dest 는임의의부프로세서레지스터이다.

FDIVR dest, ST0 dest = ST0 / dest. dest 는임의의부프로세서레지스터

이다.

FDIVP dest혹은

FDIVP dest, STO
dest /= ST0한후팝한다. dest 는임의의부프로세서레

지스터이다.

FDIVRP dest혹은

FDIVRP dest, STO
dest = ST0 / dest한후팝한다. dest 는임의의부프로

세서레지스터이다.

FIDIV src ST0 /= (float) src. ST0를정수로나눈다. src는메모리

상의워드혹은더블워드여야한다.

FIDIVR src ST0 = (float) src / ST0. 정수를 ST0로나눈다. src 는
메모리상의워드혹은더블워드여야한다.

비교

부프로세서는부동소수점수들에대한비교명령도수행한다. FCOM계열의명령들

이이작업을한다.

FCOM src ST0와 src를비교. src는임의의부프로세서레지스터거나메모리

상의단일혹은 2배정밀도수이면된다.

FCOMP src ST0와 src를비교한후팝한다.src는임의의부프로세서레지스터

거나메모리상의단일혹은 2배정밀도수이면된다.

FCOMPP ST0와 ST1를비교한후팝을두번한다.

FICOM src ST0와 (float) src를비교한다. src 는메모리상의워드혹은더

블워드정수이다.

FICOMP src ST0와 (float)src를비교한후팝한다. src 는메모리상의워드

혹은더블워드정수이다.

FTST ST0와 0을비교한다.
이명령들은부프로세서상태레지스터의C0, C1, C2, C3 비트들을바꾼다.불행이도

CPU가직접이러한비트에접근하는것은불가능하다.조건분기명령은부프로세서

의상태레지스터가아닌플래그레지스터를사용한다.그러나상태워드의비트들을

이에대응되는플래그레지스터의비트에옮기는것은아래의몇가지새로운명령을

1 ; if ( x > y )

2 ;
3 fld qword [x] ; ST0 = x
4 fcomp qword [y] ; STO 와 y 를 비교

5 fstsw ax ; 플래그 레지스터에 C 비트를 이동

6 sahf
7 jna else_part ; x>y 가 아니면 else_part 로 분기

8 then_part:
9 ; then 을 위한 부분

10 jmp end_if
11 else_part:
12 ; else 를 위한 부분

13 end_if:

그림 6.6: 비교예제

이용한다면쉽다

FSTSW dest 부프로세서의상태워드를메모리의워드나 AX레지스터에저장한

다.

SAHF AH 레지스터를플래그레지스터에저장한다.

LAHF 플래그레지스터의비트들과함게 AH 를불러온다.
그림 6.6는코드예제를보여준다. 5행과6행은부프로세서의상태워드의 C0, C1,

C2, C3 비트들을플래그레지스터에대입한다. 전달한비트들은두개의부호가 없는

정수의비교결과와동일한다.이는우리가 7행에서 JNA를사용한이유이다.

펜티엄프로(그리고그이후의프로세서들 (펜티엄 II, III)) 는두개의새로운비교

명령을지원하는데이는 CPU 의플래그레지스터의값을직접적으로변경할수있다.
FCOMI src ST0 와 src 를 비교한다. src 는 반드시 부프로세서 레지스터 여야

한다.

FCOMIP src ST0와 src를비교한후팝한다.src는반드시부프로세서레지스터

여야한다.
그림 6.7 은두개의 double 값중큰것을찾는 FCOMIP를사용한예제서브루틴을보

여주고있다.이명령들을정수비교함수들과혼동하지말라 (FICOM과 FICOMP).

잡다한명령들

이단원은부프로세서가제공하는여러잡다한명령들에대해다루어보도록한다.
FCHS ST0 = - ST0, ST0의부호를바꾼다.

FABS ST0 = |ST0| ,ST0의절대값을취한다.

FSQRT ST0 =
√
STO, ST0의제곱근을구한다.

FSCALE ST0 = ST0× 2bST1c ,ST0에2의멱수를빠르게곱한다. ST1는스택

에저제거되지않는다.그림 6.8 는이명령을어떻게사용하는지에

대한예제를보여준다.

1 global _dmax
2

3 segment .text
4 ; function _dmax
5 ; 두 개의 double 인자 들 중 큰 것을 리턴한다.

6 ; C 원형

7 ; double dmax( double d1, double d2 )

8 ; 인자들:
9 ; d1 - 첫 번째 double

10 ; d2 - 두 번째 double
11 ; 리턴값:
12 ; d1 과 d2 중 큰 것 (ST0 에 대입하여)

13 %define d1 ebp+8
14 %define d2 ebp+16
15 _dmax:
16 enter 0, 0

17

18 fld qword [d2]
19 fld qword [d1] ; ST0 = d1, ST1 = d2
20 fcomip st1 ; ST0 = d2
21 jna short d2_bigger
22 fcomp st0 ; d2 를 팝 한다.

23 fld qword [d1] ; ST0 = d1
24 jmp short exit
25 d2_bigger: ; d2 가 더 크면 아무것도 안함

26 exit:
27 leave
28 ret

그림 6.7: FCOMIP예제

1 segment .data
2 x dq 2.75 ; double 형식으로 변환

3 five dw 5

4

5 segment .text
6 fild dword [five] ; ST0 = 5

7 fld qword [x] ; ST0 = 2.75, ST1 = 5

8 fscale ; ST0 = 2.75 * 32, ST1 = 5

그림 6.8: FSCALE예제

6.3.3 예제

6.3.4 2차방정식의근의공식

우리의첫번째예제는 2차방정식의근의공식을어셈블리로나타낸것이다.

ax2 + bx + c = 0

근의공식을통해두개의근을구할수있다. x: x1 와 x2.

x1, x2 =
−b ±

√
b2 − 4ac

2a

제곱근 안의 식 (b2 − 4ac) 은 판별식(discriminant) 라 부른다. 이 값은 근이 어떠한

성질을띄는지판별하는데유용하게쓰인다.

1. 실수인중근을가진다. b2 − 4ac = 0

2. 두개의실수근을가진다. b2 − 4ac > 0

3. 두개의복소근을가진다. b2 − 4ac < 0

아래어셈블리서브루틴을사용하는작은 C 프로그램을나타냈다.

quadt.c

1 #include <stdio .h>

2

3 int quadratic( double, double, double, double *, double *);

4

5 int main()

6 {

7 double a,b,c , root1 , root2 ;

8

9 printf (”Enter a , b, c : ” );

10 scanf(”%lf %lf %lf” , &a, &b, &c);

11 if (quadratic( a , b, c , &root1, &root2) )

12 printf (”roots : %.10g %.10g\n”, root1, root2);

13 else
14 printf (”No real roots\n” );

15 return 0;

16 }

quadt.c

여기어셈블리서브루틴이있다 :

quad.asm
1 ; 함수 quadratic
2 ; 2차 방정식의 근을 찾는다.

3 ; a*x^2 + b*x + c = 0

4 ; C 원형:
5 ; int quadratic( double a, double b, double c,
6 ; double * root1, double *root2 )

7 ; 인자:
8 ; a, b, c - 2차 방정식의 각 계수들 (위를 참조)

9 ; root1 - 첫 번째 근을 저장할 double 을 가리키는 포인터

10 ; root2 - 두 번째 근을 저장할 double 을 가리키는 포인터

11 ; 리턴값:
12 ; 실근을 찾으면 1 을 리턴, 아니면 0 리턴

13

14 %define a qword [ebp+8]
15 %define b qword [ebp+16]
16 %define c qword [ebp+24]
17 %define root1 dword [ebp+32]
18 %define root2 dword [ebp+36]
19 %define disc qword [ebp-8]
20 %define one_over_2a qword [ebp-16]
21

22 segment .data
23 MinusFour dw -4
24

25 segment .text
26 global _quadratic
27 _quadratic:
28 push ebp
29 mov ebp, esp
30 sub esp, 16 ; 2 개의 더블을 할당한다 (disc 와 one_over_2a)
31 push ebx ; 원래의 ebx 값을 저장

32

33 fild word [MinusFour]; 스택 -4
34 fld a ; 스택: a, -4
35 fld c ; 스택: c, a, -4
36 fmulp st1 ; 스택: a*c, -4
37 fmulp st1 ; 스택: -4*a*c
38 fld b
39 fld b ; 스택: b, b, -4*a*c
40 fmulp st1 ; 스택: b*b, -4*a*c
41 faddp st1 ; 스택: b*b - 4*a*c
42 ftst ; 0 과 비교

43 fstsw ax
44 sahf
45 jb no_real_solutions ; 만일 disc < 0, 이면 해가 없다.

46 fsqrt ; 스택: sqrt(b*b - 4*a*c)

47 fstp disc ; 저장 후 스택을 팝 한다.

48 fld1 ; 스택: 1.0

49 fld a ; 스택: a, 1.0

50 fscale ; 스택: a * 2^(1.0) = 2*a, 1

51 fdivp st1 ; 스택: 1/(2*a)
52 fst one_over_2a ; 스택: 1/(2*a)
53 fld b ; 스택: b, 1/(2*a)
54 fld disc ; 스택: disc, b, 1/(2*a)
55 fsubrp st1 ; 스택: disc - b, 1/(2*a)
56 fmulp st1 ; 스택: (-b + disc)/(2*a)
57 mov ebx, root1
58 fstp qword [ebx] ; *root1 에 저장

59 fld b ; 스택: b
60 fld disc ; 스택: disc, b
61 fchs ; 스택: -disc, b
62 fsubrp st1 ; 스택: -disc - b
63 fmul one_over_2a ; 스택: (-b - disc)/(2*a)
64 mov ebx, root2
65 fstp qword [ebx] ; *root2 에 저장

66 mov eax, 1 ; 리턴값은 1

67 jmp short quit
68

69 no_real_solutions:
70 mov eax, 0 ; 리턴값은 0

71

72 quit:
73 pop ebx
74 mov esp, ebp
75 pop ebp
76 ret quad.asm

6.3.5 파일로부터배열을읽기

이번 예제에서는 어셈블리가 파일로 부터 double 을 읽어들인다. 아래는 짧은 C
테스트프로그램이다.

readt.c

1 /*

2 * 이프로그램은32비트 read_doubles() 어셈블리프로시저를테스트한다.

3 * 이는 stdin으로부터 double을읽어들인다. (파일로부터읽기위해

4 * 리다이렉션(redirection)을사용한다.)

5 */

6 #include <stdio .h>

7 extern int read_doubles( FILE *, double *, int );

8 #deöneMAX 100

9

10 int main()

11 {

12 int i , n;

13 double a[MAX];

14

15 n = read_doubles(stdin, a , MAX);

16

17 for( i=0; i < n; i++ )

18 printf (”%3d %g\n”, i , a[ i ]);

19 return 0;

20 }

readt.c

아래는어셈블리루틴이다.

read.asm
1 segment .data
2 format db "%lf", 0 ; format for fscanf()
3

4 segment .text
5 global _read_doubles
6 extern _fscanf
7

8 %define SIZEOF_DOUBLE 8

9 %define FP dword [ebp + 8]

10 %define ARRAYP dword [ebp + 12]

11 %define ARRAY_SIZE dword [ebp + 16]

12 %define TEMP_DOUBLE [ebp - 8]

13

14 ;
15 ; _read_doubles 함수

16 ; C 원형:
17 ; int read_doubles( FILE * fp, double * arrayp, int array_size );
18 ; 이 함수는 텍스트 파일로 부터 double 을 읽어 들여서 EOF 가 나올 때 까지나

19 ; 배열이 꽉 찰 때 까지 배열에 저장한다.

20 ; 인자:
21 ; fp - 읽어 들일 FILE 포인터 (읽기 가능해야 한다)

22 ; arrayp - 읽어 들인 값을 저장할 double 배열을 가리키는 포인터

23 ; array_size - 배열의 원소의 개수

24 ; 리턴:
25 ; 배열에 저장된 double 의 개수 (EAX 에 저장됨)

26

27 _read_doubles:
28 push ebp
29 mov ebp,esp
30 sub esp, SIZEOF_DOUBLE ; 스택에 하나의 double 을 정의

31

32 push esi ; esi 저장

33 mov esi, ARRAYP ; esi = ARRAYP
34 xor edx, edx ; edx = 배열 원소 위치 (처음이 0)

35

36 while_loop:
37 cmp edx, ARRAY_SIZE ; edx < ARRAY_SIZE 인가?

38 jnl short quit ; 아니면 루프 루프 종료

39 ;
40 ; TEMP_DOUBLE 로 double 값을 읽어 들이기 위해 fscanf() 를 호출

41 ; fscanf() 는 edx 값을 바꿀 수 있으므로 저장해 놓는다.

42 ;
43 push edx ; edx 저장

44 lea eax, TEMP_DOUBLE
45 push eax ; &TEMP_DOUBLE 푸시

46 push dword format ; &format 푸시

47 push FP ; 파일 포인터 푸시

48 call _fscanf
49 add esp, 12

50 pop edx ; edx 복원

51 cmp eax, 1 ; fscanf 가 1 을 리턴했는가?

52 jne short quit ; 아니면 루프 종료

53 ;
54 ; TEMP_DOUBLE 를 ARRAYP[edx] 에 대입

55 ; (8 바이트의 double 은 두 개의 4 바이트 레지스터를 통해 대입된다.)

56 ;
57 mov eax, [ebp - 8]

58 mov [esi + 8*edx], eax ; 하위 4 바이트를 복사

59 mov eax, [ebp - 4]

60 mov [esi + 8*edx + 4], eax ; 상위 4 바이트를 복사

61

62 inc edx
63 jmp while_loop
64

65 quit:
66 pop esi ; esi 를 복원한다.

67

68 mov eax, edx ; 리턴값을 eax 에 저장

69

70 mov esp, ebp
71 pop ebp
72 ret read.asm

6.3.6 소수찾기

마지막예제는소수를찾는프로그램이다.이미앞에서한번다루어보았지만이번

것은좀더속도가향상되었다.이번프로그램은앞에서찾은소수를배열에저장해놓

고특정한수가소수인지를판별하기위해특정한수이전의모든짝수로나누어보는

것이아니라배열에저장해놓은소수들만으로나누어본다.

또다른 차이점은 특정한 수가 소수 인지를 판별하기 위해 그 수의 제곱근 이하의

소수들만으로 나누어 본다는 것이다. 이 때문에 부프로세서의 제어 워드를 변경하여

제곱근을 정수로 저장할 때 반올림 하기 보단 내림을 해야한다. 이는 제어 워드의 10

과 11 번째 비트를 변경하면 된다. 이 비트들은 RC (반올림 제어, Rounding Control)
비트로 불린다. 만일 두 비트 모두 0 이라면 (기본값) 부프로세서는 정수로 변환시에

반올림을한다.두비트가모두 1이라면부프로세서는정수로변환시내림을한다.유

의해야할점은기존의제어워드를저장하고리턴하기전에복원해야한다는것이다.

아래는 C 드라이버프로그램이다

fprime.c

1 #include <stdio .h>

2 #include < stdlib .h>

3 /*

4 * önd_primes함수

5 * 지정된개수의소수들을찾는다

6 * 인자들 :

7 * a −배열을보관할배열

8 * n −찾을소수의개수

9 */

10 extern void önd_primes( int * a , unsigned n );

11

12 int main()

13 {

14 int status ;

15 unsigned i;

16 unsignedmax;

17 int * a ;

18

19 printf (”How many primes do you wish to önd? ” );

20 scanf(”%u”, &max);

21

22 a = calloc ( sizeof( int ), max);

23

24 if ( a ) {

25

26 önd_primes(a ,max);

27

28 /*찾은마지막20개의소수를출력 */

29 for( i= ( max > 20 ) ? max− 20 : 0; i < max; i++ )

30 printf (”%3d %d\n”, i+1, a[ i ]);

31

32 free (a );

33 status = 0;

34 }

35 else {

36 fprintf ( stderr , ”Can not create array of %u ints\n” , max);

37 status = 1;

38 }

39

40 return status ;

41 }

fprime.c

아래는어셈블리루틴이다.

prime2.asm
1 segment .text
2 global _find_primes
3 ;
4 ; find_primes 함수

5 ; 지정한 수 만큼의 소수를 찾는다

6 ; 인자:
7 ; array - 소수를 보관할 배열

8 ; n_find - 찾을 소수의 개수

9 ; C 원형:
10 ;extern void find_primes( int * array, unsigned n_find )

11 ;
12 %define array ebp + 8

13 %define n_find ebp + 12

14 %define n ebp - 4 ; 현재까지 찾은 소수의 개수

15 %define isqrt ebp - 8 ; guess 의 제곱근의 내림한 값

16 %define orig_cntl_wd ebp - 10 ; 원래 제어 워드 값

17 %define new_cntl_wd ebp - 12 ; 새로운 제워 워드 값

18

19 _find_primes:
20 enter 12,0 ; 지역 변수를 위한 공간을 할당

21

22 push ebx ; 가능한 레지스터 변수들을 저장

23 push esi
24

25 fstcw word [orig_cntl_wd] ; 현재의 제어 워드를 저장

26 mov ax, [orig_cntl_wd]
27 or ax, 0C00h ; 반올림 비트를 11 로 한다. (내림)

28 mov [new_cntl_wd], ax
29 fldcw word [new_cntl_wd]
30

31 mov esi, [array] ; esi 는 배열을 가리킨다.

32 mov dword [esi], 2 ; array[0] = 2

33 mov dword [esi + 4], 3 ; array[1] = 3

34 mov ebx, 5 ; ebx = guess = 5

35 mov dword [n], 2 ; n = 2

36 ;
37 ; 이 바깥 루프는 각 루프 마다 새로운 소수를 하나씩 찾는다. 이 새로운 소수는

38 ; 배열의 끝부분에 추가된다. 기존의 소수 찾는 프로그램들과는 달리 이 함수는

39 ; 소수 판별을 위해 그 수 미만의 모든 홀수들로 나누어 보지 않는다.

40 ; 오직 이미 찾은 소수들 만으로 나누어 본다. (이 때문에 이 소수들을 배열에

41 ; 저장한 것이다.)

42 ;
43 while_limit:
44 mov eax, [n]
45 cmp eax, [n_find] ; while ( n < n_find )

46 jnb short quit_limit
47

48 mov ecx, 1 ; ecx 는 원소의 위치를 가리키는데 사용

49 push ebx ; guess 를 스택에 저장

50 fild dword [esp] ;guess 를 부프로세서 스택에 불러온다

51 pop ebx ; 스택에서 guess 를 뺀다.

52 fsqrt ; sqrt(guess) 값을 구한다.

53 fistp dword [isqrt] ; isqrt = floor(sqrt(quess))
54 ;
55 ; 이 내부의 루프는 guess (ebx)를 이전에 찾은 소수들로 나누어서

56 ; guess 의 소수 인수를 찾을 때 까지 혹은

57 ; 소수 인수가 floor(sqrt(guess)) 보다 클 때 까지 나눈다.

58 ;(floor 은 내림 함수)

59 while_factor:
60 mov eax, dword [esi + 4*ecx] ; eax = array[ecx]
61 cmp eax, [isqrt] ; while ( isqrt < array[ecx]
62 jnbe short quit_factor_prime
63 mov eax, ebx
64 xor edx, edx
65 div dword [esi + 4*ecx]

66 or edx, edx ; && guess % array[ecx] != 0 )

67 jz short quit_factor_not_prime
68 inc ecx ; 다음 소수로 시도한다

69 jmp short while_factor
70

71 ;
72 ; 새 소수를 찾음!

73 ;
74 quit_factor_prime:
75 mov eax, [n]
76 mov dword [esi + 4*eax], ebx ; guess 를 배열의 끝 부분에 추가

77 inc eax
78 mov [n], eax ; inc n
79

80 quit_factor_not_prime:
81 add ebx, 2 ; 다음 홀수를 시도해 본다

82 jmp short while_limit
83

84 quit_limit:
85

86 fldcw word [orig_cntl_wd] ; 제어 워드를 복원

87 pop esi ; 레지스터 변수들을 복원

88 pop ebx
89

90 leave
91 ret prime2.asm

구조체와 C++

7.1 구조체

7.1.1 서론

C 에서사용되는구조체는연관된데이터를하나의변수에보관하는것이라말할

수있다.이는몇가지장점이있다.

1. 구조체에정의된데이터가연관되어있음을보임으로써코드를명료하게할수

있다.

2. 이를통해함수로의데이터전달을단순하게할수있다. 여러개의변수를독립

적으로전달하는대신에이를이용해하나의단위만전달하면된다.

3. 이는코드의지역성(locality)1) 을향상시킨다.

어셈블리의관점에서볼때구조체는원소들의크기가제각각인배열로볼수있다.

실제배열의원소들의크기는언제나같은형이자,같은크기여야한다.이를통해실제

배열에서는 배열의 시작 주소와 원소의 크기, 원소의 번째 수 만 알면 원소의 주소를

계산할수있게된다.

그러나구조체의원소들은동일한크기를가질필요가없다. (그리고대부분동일

한크기가아니다)이때문에구조체의각각의원소들은뚜렷이지정되어있어야하고

수치적인위치대신에태그(tag) (또는이름)를가진다.

어셈블리에서구조체의원소에접근하는것은배열의원소의접근하는것과비슷하

다.원소에접근하기위해서는먼저구조체의시작주소를알아야하고,구조체의시작

부분으로부터의각원소의상대오프셋(relative offset)을알아야한다.그러나,배열의

경우 이 오프셋이 원소의 번째 수 만 알고도 계산될 수 있었지만, 구조체의 원소들은

컴파일러에의해오프셋이지정된다.

예를들어아래의구조체를보자.

struct S {

short int x ; /* 2바이트정수 */

int y ; /* 4바이트정수 */

double z ; /* 8바이트 øoat */

};

1) 운영체제에관련한책의가상메모리관리단원을살펴보라.

오프셋 원소
0 x
2

y
6

z

그림 7.1: 구조체 S

오프셋 원소
0 x
2 사용되지않음
4

y
8

z

그림 7.2: 구조체 S

그림 7.1은 S형의변수가컴퓨터메모리상에어떻게나타나는지보여준다. ANSI
C 표준에따르면메모리상에배열된구조체의원소들의순서는 struct정의에서나

타난 순서와 동일해야 한다고 하였다. 또한 첫 번째 원소는 구조체의 최상단에 위치

해야 한다고 하였다. (i.e 오프셋이 0 ) 이는 또한 stddef.h 헤더 파일에 offsetof()
라는매우유용한매크로를정의하였는데이매크로는구조체의특정한원소의오프셋

값을 계산한 후 리턴한다. 이 매크로는 두 개의 인자를 가지는데 첫 번째는 구조체의

이름이고 두 번째는 오프셋 값을 계산할 원소의 이름이다. 따라서 그림 7.1 에서의

offsetof(S,y)의결과는 2가된다.

7.1.2 메모리배열하기

gcc 컴파일러를 사용시, offsetof 매크로를 이용하여 y 의 오프셋을 구한다면 2

가아닌 4를리턴하게된다.왜냐하면 gcc (그리고다수의컴파일러들이)는변수들을주소가 4로나누어진다면주

소는더블워드경계에놓여있

다는점을상기해라

기본적으로 더블워드 경계에 놓기 때문이다. 32 비트 보호 모드에서 CPU 는 메모리

상의 데이타가 더블워드 경계에 놓여 있을 때 더 빨리 읽어들일 수 있다. 그림 7.2 는

S 구조체가 gcc 를 사용했을 때 메모리 상에 어떻게 나타날지 보여준다. 컴파일러는

구조체에사용되지않는 2바이트를끼워넣어서 y (그리고 z도) 를더블워드경계에

놓이게하였다.이는 offsetof를사용하여오프셋을계산하는것이정의된구조체의

형태만을보고값을계산하는것보다훨씬좋은것인지를알수있다.

당연하게도 구조체가 오직 어셈블리에서만 사용된다면 프로그래머는 스스로 오

프셋을지정할수있다. 그러나어셈블리와 C 를함께사용한다면어셈블리코드와 C
코드의구조체의원소들의오프셋이반드시일치해야한다.한가지문제가되는점은

C컴파일러마다원소에대해다른오프셋값을줄수있다는것이다.예를들어위에서

보았듯이 gcc컴파일러는 S구조체를그림 7.2와같은모습으로만들지만,볼랜드사의

컴파일러는구조체를그림 7.1과같은모습으로만든다. C컴파일러들은데이터를정

struct S {

short int x ; /* 2바이트정수 */
int y ; /* 4바이트정수 */
double z ; /* 8바이트 øoat */

} __attribute__((packed));

그림 7.3: 압축된 gcc 에서의구조체사용

렬하기위해여러가지방법을사용한다.그러나 ANSI C표준은그어떤방법도구체화

하지않았으므로컴파일러들이저마다다른방법들을사용하게되었다.

gcc 컴파일러는데이터의정렬방법을구체화하기위해유연하고도복잡한방법을

지원한다. 이 컴파일러에선 특별한 문법을 사용함으로써 어떠한 형이라도 데이터를

정렬하는방법을지정할수있게하였다.예를들어아래의행을보면

typedef short int unaligned_int __attribute__((aligned (1)));

이는 unaligned_int라는새로운형을정의하는데이는바이트경계에맞추어져있다.

(__attribute__다음의괄호들이꼭필요하다) aligned인자1은2의멱수들로바뀔

수있으며이는다른정렬조건을나타낸다. (2 는워드정렬, 4는더블워드정렬등등)

만일구조체의 y원소가 unaligned_int형이라면 gcc는 y를오프셋2에놓을것이다.

그러나, z 는 오프셋 8 에 놓이게 되는데 왜냐하면 double 형들은 기본값이 더블워드

정렬이기때문이다.물론 z를오프셋 6에놓이게할수있도록변경할수있다.

gcc 컴파일러는 또한 구조체를 압축(pack) 하는 것을 지원한다. 이는 컴파일러로

하여금구조체생성시최소한의공간만을사용하게한다.그림 7.3는 S가이방법으로

어떻게다시쓰일수있는지보여준다.이러한형태의 S는최소한의바이트를사용하여,

이경우 14바이트를사용한다.

마이크로소프트와 볼랜드 사의 컴파일러는 #pragma 지시어를 통해 위와 동일한

작업을할수있다.

#pragma pack(1)

이지시어는컴파일러가바이트경계에맞추어구조체의원소들을압축하게한다. (i.e
추가적인사용없이) 1 은 2, 4, 8, 16 등으로변경될수있는데이는각각워드, 더블워

드, 쿼드워드, 패러그래프 경계에 맞추라는 뜻이다. 이 지시어는 다른 지시어의 의해

변경되기전까지계속효력을발휘한다.이때문에이와같은지시어가만이사용되는

헤더파일에서는문제가발생하기도한다.만일이헤더파일이다른구조체를포함한

헤더파일보다먼저포함(include) 되었다면이구조체는기본값과다른방식으로메모

리상에배열될수있다.이와같은오류는찾기가매우힘들다.프로그램의각기다른

모듈들이구조체의원소들을각기다른장소에배열하게된다.

이와같은문제를피하는방법이있다.마이크로소프트와볼랜드사는현재의정렬

방법을저장한후나중에복원할수있게한다.그림 7.4 는이를나타낸다.

7.1.3 비트필드

비트필드는구조체의멤버들로하여금특정한크기의비트만사용할수있게한다.

비트의크기가꼭 8의배수가되야할필요는없다. 비트필드멤버는 unsigned int나
int 멤버에콜론(:)와비트크기를적음으로써정의된다.그림 7.5는이예를보여준다.

이는 32비트변수를다음과같은부분으로나뉘어정의한다.

#pragma pack(push) /* 이전정렬상태를저장한다*/
#pragma pack(1) /* 바이트기준정렬로정의 */

struct S {

short int x ; /* 2바이트정수 */
int y ; /* 4바이트정수 */
double z ; /* 8바이트 øoat */

};

#pragma pack(pop) /* 이전의정렬상태를복원한다*/

그림 7.4: 마이크로소프트나볼랜드사의압축된구조체

struct S {

unsigned f1 : 3; /*3비트필드 */
unsigned f2 : 10; /*10비트필드*/
unsigned f3 : 11; /* 11비트필드 */
unsigned f4 : 8; /* 8비트필드 */

};

그림 7.5: 비트필드예제

8비트 11비트 10비트 3비트

f4 f3 f2 f1

첫번째비트필드는더블워드의최하위비트들을가리킨다. 2)

그러나우리가이비트들이실제로메모리상에어떻게나타나는지본다면이형식은

그리간단한것이아니다.이문제는비트필드가바이트경계에걸쳐있을때나타난다.

왜냐하면 리틀 엔디안 형식에서 바이트는 메모리에 거꾸로 저장되기 때문이다. 예를

들어 S구조체의비트필드는메모리에아래와같이나타난다.

5비트 3비트 3비트 5비트 8비트 8비트

f2l f1 f3l f2m f3m f4

f2l 라벨은 f2 비트필드의하위 5비트를의미한다 (i.e LSB) . f2m 라벨은 f2비트필드의

상위 5비트를가리킨다. 수직선은바이트경계를표시한다. 만일모든바이트들을거

꾸로배치한다면 f2 와 f3 의조각들이올바른위치에합쳐질것이다.

메모리에서의배열형태는데이터가프로그램바깥으로전송되지않는한중요하지

않다. (사실이러한경우는비트필드를다룰때잘발생한다)하드웨어장치인터페이스

를만들때,홀수의비트를가지는비트필드를사용하는경우가잦은데이는비트필드를

사용하면매우유용하기때문이다.

그예로는 SCSI 3) 를들수있다. SCSI장치에직접적으로읽기명령을내리는방법은

그림 7.6 에나타난형식대로 6바이트의메세지를장치에전달하면된다. 이를비트

2) 사실 ANSI/ISO C 표준은 컴파일러로 하여금 비트들이 어떻게 나열되는지는 정확히 정하지 않았다.
그러나대부분의C컴파일러들(gcc,마이크로소프트,볼랜드사의컴파일러)들은이와같은모습으로
배열한다.

3) 소형 컴퓨터 시스템 인터페이스 (Small Computer Systems Interface) 의 약어로 하드 디스크 등등의
장치에대한규격이다.

바이트 \비트 7 6 5 4 3 2 1 0
0 연산부호 (08h)
1 논리단위 # 논리블록주소의최상위비트
2 논리블록주소의중간
3 논리블록주소의최하위바이트
4 전달크기
5 제어

그림 7.6: SCSI 읽기명령형식

1 #deöneMS_OR_BORLAND (deöned(__BORLANDC__) \
2 || deöned(_MSC_VER))
3

4 #if MS_OR_BORLAND
5 # pragma pack(push)
6 # pragma pack(1)
7 #endif
8

9 struct SCSI_read_cmd {
10 unsigned opcode : 8;
11 unsigned lba_msb : 5;
12 unsigned logical_unit : 3;
13 unsigned lba_mid : 8; /* 중간비트들*/
14 unsigned lba_lsb : 8;
15 unsigned transfer_length : 8;
16 unsigned control : 8;
17 }

18 #if deöned(__GNUC__)
19 __attribute__((packed))
20 #endif
21 ;
22

23 #if MS_OR_BORLAND
24 # pragma pack(pop)
25 #endif

그림 7.7: SCSI 읽기명령형식구조체

8비트 8 bits 8 비트 8 bits 3 비트 5 bits 8 비트
제어 전달크기 lba_lsb lba_mid 논리단위 lba_msb 연산부호

그림 7.8: SCSI_read_cmd필드의모습

필들르이용하여표현하는데어려움은논리블록주소(logical block address)에있는데,

왜냐하면이것이 3개의다른바이트에걸쳐있기때문이다. 그림 7.6 에는빅엔디안

형식으로 저장된 데이터를 볼 수 있다. 그림 7.7 에는 모든 컴파일러에 대해서 사용

가능하게한정의를보여준다.첫 2행은매크로를정의하는데이는마이크로소프트나

볼랜드사의컴파일러로컴파일될때참이된다.아마혼동되는부분은 11에서 14행

까지일것이다.먼저우리는 lba_mid와 lba_lsb 필드가하나의16비트필드로정의되

지 않고 따로 정의됬는지 의문을 가질 것이다. 이는 데이타가 빅 엔디안 순서로 되어

1 struct SCSI_read_cmd {
2 unsigned char opcode;
3 unsigned char lba_msb : 5;
4 unsigned char logical_unit : 3;
5 unsigned char lba_mid; /*중간비트들 */
6 unsigned char lba_lsb;
7 unsigned char transfer_length ;
8 unsigned char control ;
9 }

10 #if deöned(__GNUC__)
11 __attribute__((packed))
12 #endif
13 ;

그림 7.9: 다른형태의 SCSI 읽기명령형식구조체

있기 때문이다. 16 비트 필드는 컴파일러에 의해 리틀 엔디안 순서로 저장될 것이다.

두 번째로 lba_msb 와 logical_unit 는 거꾸로 된 것처럼 보인다. 그러나 이 경우엔

적용되지 않는다. 이들은 이 순서대로 집어너져야 한다. 그림 7.8 은 48 비트필드가

메모리에어떻게 표현되는지보여준다. (바이트경계는 두개의선으로 표현되었다.)

이것이 메모리에 리틀 엔디안 순서로 저장된다면 비트들은 원하는 형식으로 배열될

것이다. (그림 7.6).

일을더복잡하게하는것은, SCSI_read_cmd 은마이크로소프트 C 에서완벽하게

작동하지않는다는점이다. sizeof(SCSI_read_cmd)라는식을계산하였을때,마이크

로소프트C에서는6이아닌8을리턴한다.왜냐하면마이크로소프트컴파일러는비트

를메모리나타낼때비트필드의형을기준으로나타내기때문이다.모든비트필드들이

unsigned로정의되어있으므로컴파일러는구조체맨마지막에2바이트를덧붙여서

더블워드의배수로만들어준다.이와같은현상은모든필드를 unsigned short로정

의해줌으로써해결할수있다.그렇게한다면마이크로소프트는6바이트가 2바이트

워드의정수배이므로부수적인바이트를추가할필요가없어지게된다. 4) 다른컴파

일러는이와같이바꾸어도정확하게동작한다. 그림 7.9 는 3 개 컴파일러모두에서

잘 작동하는 정의를 보여준다. unsigned char 를 이용하여 오직 두 개만 비트필드를

이용했다.

이전에 우리가 이야기 하였던 것이 혼란스러워도 걱정할 필요가 없다. 왜냐하면

정말로 혼란스러운 내용이니까 말이다. 저자는 덜 혼란스럽기 위해서 비트필드를 사

용하지않고비트를각각다루기위해서비트연산을사용한다.

7.1.4 어셈블리에서구조체사용하기

위에서 이야기 하였듯이 어셈블리에서 구조체를 사용하는 것은 배열을 사용하는

것과비슷한다.간단한예로구조체 S의원소 y를0으로만드는루틴을어떻게만들지

생각해보자.이루틴의원형은아래과같다.

void zero_y( S * s_p );

어셈블리루틴은

4) 여러가지다른형의비트필드를사용하는것은매우혼란스러운 .

1 #include <stdio .h>
2

3 void f ( int x )

4 {

5 printf (”%d\n”, x );
6 }

7

8 void f ( double x )
9 {

10 printf (”%g\n”, x );
11 }

그림 7.10: 두개의 f()함수

1 %define y_offset 4

2 _zero_y:
3 enter 0,0
4 mov eax, [ebp + 8] ; 스택에서 s_p(구조체 포인터) 를 불러온다

5 mov dword [eax + y_offset], 0

6 leave
7 ret

C는함수에게구조체를값으로전달하는것을허용한다.그러나이는좋지않은생

각이다.값으로전달을할경우,구조체의모든데이터들이스택에복사된후루틴에게

전달될것이다. 단순히구조체를가리키는포인터를전달하는것이더효과적이라할

수있다.

C는또한구조체타입이함수의리턴값으로도사용할수있게한다.하지만명백히

도, 구조체는 EAX레지스터에저장되어리턴될수없다. 이상황을여러컴파일러들이

각기다른방법으로해결한다.가장많이쓰이는방법은컴파일러가내부적으로함수를

구조체를가리키는포인터를인자로갖게변경하는것이다.이포인터는루틴이호출된

곳밖에서정의된구조체에반환값을대입하는데사용된다.

대부분의어셈블러 (NASM포함)은여러분의어셈블리코드에서구조체를정의하

는것을지원하는자체적인방법이있다.자세한내용은이에관련한문서들을참조하

여라.

7.2 어셈블리와 C++

C++ 프로그래밍 언어는 C 언어의 확장이라 볼 수 있다. 어셈블리와 C 가 서로 작

업하는방법대부분이 C++에서도동일하게적용된다.그러나,몇가지규칙들은수정

되야한다. 또한, C++ 의몇가지확장된부분들은어셈블리로이해하기가편하다. 이

단원은독자여러분이 C++ 의기본적인이해가되어있다고생각한다.

7.2.1 오버로딩과이름맹글링(NameMangling)

C++ 은 같은 이름으로 서로 다른 함수 (그리고 클래스 멤버 함수)를 정의하는 것

을 허용한다. 2 개 이상의 함수가 동일한 이름을 사용한다면, 이 함수들을 오버로드

(overloaded) 되었다라고한다.만일이두함수들이 C에서같은이름으로정의되었다

면C링커는오류를출력할것이다.왜냐하면링킹시오브젝트파일에서동일한심볼에

두 개의 정의를 찾을 것이기 때문이다. 예를 들어 그림 7.10 의 코드를 고려해 보자.

어셈블리에서두개의 _f라이름붙여진라벨은명백히오류를출력한다.

C++은C와동일한링킹과정을거치지만,함수에이름붙여진라벨을이름맹글링

(name mangling)이라는과정을통해오류를피해갈수있다. 사실 C 도이름맹글링을

사용한다. C 에서함수에대한라벨을정의하기위해서함수에이름에 _ 를붙인다.그

러나 C는그림 7.10의두함수의이름을동일한라벨로변경해오류를출력할것이다.

C++ 는 이름 맹글링시 조금 더 복잡한 방법을 사용하여 이 두 함수에 대한 서로 다른

라벨을 만들어 준다. 예를 들어 그림 7.10 의 첫 번째 함수는 DJGPP 에 의해 _f__Fi
라는라벨을가지고두번째함수는 _f__Fd라는라벨을부여받게된다.이를통해링크

오류를피해나갈수있다.

불행히도 C++에서라벨을어떻게만드는지에대한기준이없기때문에각기다른

컴파일러들이 자신들 만의 방법으로 이름 맹글링을 한다. 예를 들어 볼랜드 C++ 은

그림 7.10의두함수에대해 @f$qi와 @f$qd라는라벨을부여하게된다.그러나이러

한규칙들은완전히무작위적이지는않다. 맹글링된이름은함수의특징을나타낸다.

여기서이특징은함수의인자들의배열순서와형에의해정해진다.예를들어 int인
자만을가지는함수는맹글링된이름끝부분에 i가붙게된다. (이는볼랜드와DJGPP
모두동일)또한 double인자를가지는함수는끝에 d 가붙게된다.만일아래와같은

원형을가지는함수 f가있다면,

void f ( int x , int y , double z);

DJGPP 는이함수의이름을 _f__Fiid로맹글링하고볼랜드는 @f$qiid로한다.

함수의리턴형식은함수의특징에포함되지않으며,맹글링된이름에나타나지도

않는다.이를통해우리는C++의오버로딩규칙에대해알수있다.오직고유의특징을

가지는 함수많이 오버로드 될 수 있다. 만일 C++ 에서 동일한 이름과 특징을 가지는

함수가정의된다면,이는동일한맹글링된이름을가지게되므로링커애러를발생하게

된다.기본적으로 C++ 함수들은오버로드되지않아도이름이맹글링된다.왜냐하면

컴파일러는컴파일시에이함수가오버로드됐는지안됐는지알수있는방법이없기

때문이다.사실함수의이름을맹글링하는방법과비슷한방법으로전역변수의이름을

맹글링한다.따라서,만일어떤파일에특정한형을가지는전역변수를정의하였는데

다른 파일에서 다른 형으로 사용한다면 링커 오류가 발생하게 된다. 이러한 C++ 의

특징은형이안전한링킹(typesafe linking) 이라불린다. 이를통해일정하지않은원형

에 대한 오류까지도 잡아낼 수 있다. 이는 특정한 모듈에서의 함수의 정의가 다른 모

듈에서의원형과일치하지않을때발생한다. C 에서는이를디버그하기매우어렵다.

왜냐하면 C는이오류를잡아내지못하기때문이다.프로그램은오류없이컴파일되고

링크되지만 함수를 호출시 함수가 예상했던 값들과 다른 값들을 스택에 넣어 프로그

램이동작을하지못한다.하지만 C++에서는이는링커오류를발생해미연에이러한

일을방지할수있다.

C++컴파일러가함수호출을분석(parsing)할때, 함수에전달되는인자들의형을

살펴 보아서 일치하는 함수가 존재하는지 살펴본다. 5) 만일 일치하는 함수를 찾으면

컴파일러는이름맹글링규칙을사용하여올바른함수에대한 CALL명령을수행한다.

5) 이 형은 정확히 일치할 필요가 없다. 컴파일러는인자들을 형변환 것들도 고려해 보기 때문이다. 이
과정에대한규칙은이책의범위를훨씬뛰어넘는다.자세한내용은 C++ 책들을참조해라.

각기다른컴파일러가각기다른맹글링규칙을사용하므로서로다른컴파일러에

서 컴파일된 C++ 코드는 링크가 되지 않을 수 있다. 이 사실은 우리가 미리 컴파일된

C++ 라이브러리를사용할때중요하게작용할수있다.만일우리가 C++ 코드와함께

사용될 수 있는 함수를 어셈블리에서 작성한다고 한다면 C++ 코드를 컴파일하는데

사용되는 컴파일러를 보고 이름 맹글링 규칙을 적용시켜야 한다. (또는 아래에 나온

기술을이용한다)

아마 눈치가 빠른 학생들은 그림 7.10 의 코드가 예상했던 대로 작동하는 것인지

물어볼 것이다. C++ 는 모든 함수의 이름을 맹글링하기에 printf 함수도 맹글링 되

므로컴파일러는라벨 _printf에올바른 CALL을할수없다는것이다. 이는올바른

생각이다. printf의원형이단순히파일상단에놓여있다면이와같은일이발생할

것이다.그원형은 :

int printf ( const char *, ...);

DJGPP는이를 _printf__FPCce로맹글링할것이다. (F는함수(function)를, P는포

인터(pointer), C는상수(const), c는 char, e는생략(ellipsis) 을각각뜻한다)이는보통

의 C 라이브러리의 printf함수를호출하지않을것이다! 당연하게도 C++ 코드에서

C 함수를호출하는방법이있다. 이는매우유용한데왜냐하면엄청난수의유용한 C
코드들이있기때문이다.게다가 C코드를호출하는것을할수있다면 C++에서또한

보통의 C 맹글링규약을이용한어셈블리코드도호출할수있게된다.

C++ 에선 extern키워드를이용하여특정한함수와전역변수에 C 맹글링규칙을

적용할수있다. C++용어로이렇게선언된함수와변수는 C연계(C linkage)를사용한

다고말한다.예를들어, printf를 C 연계시키려면아래와같이원형을쓰면된다.

extern ”C” int printf ( const char *, ... );

이명령은컴파일러로하여금이함수에대해 C++이름맹글링을적용하지않으며,그

대신에 C 규칙에따라맹글링을하게한다. 그러나, 위와같이정의된 printf는오버

로드 되지 않는다. 이와 같은 방법을 통해 C++ 와 어셈블리가 손쉽게 소통할 수 있다.

함수를 C 연계를사용하게정의한뒤, C 호출규약을사용하는것이다.

편의를 위해, C++ 은 또한 여러개의 함수들과 전역 변수를 한꺼번에 연계시킬 수

있게해준다.이는중괄호를사용하면된다.

extern ”C” {

/* C연계전역변수들과함수원형들*/

}

만일, 여러분이 C/C++ 컴파일러에서 모두 작동되는 ANSI C 헤더를 보게 된다면,

헤더파일상단에다음과같은코드가있음을알수있다.

#ifdef __cplusplus

extern ”C” {

#endif

그리고하단에중괄호로닫는부분이있음을알수있다.C++컴파일러들은__cplusplus
를 매크로 (_ 가 두 개) 를 정의한다. 위 코드는 헤더파일이 C++ 에서 컴파일 될 경우

헤더파일전체를 extern "C"로감싸고, C에서컴파일된다면어떠한짓도하지않는

다. (이는 C 컴파일러를 extern "C"를문법상의오류로처리하기때문에필요하다).

1 void f ( int & x ) // &는레퍼런스인자를의미한다

2 { x++; }
3

4 int main()
5 {

6 int y = 5;
7 f (y ); // y에대한레퍼런스가전달되었다. &가없음을유의!

8 printf (”%d\n”, y ); // 6을출력

9 return 0;
10 }

그림 7.11: 레퍼런스(Reference) 예제

위와 같은 방법으로 C 와 C++ 모두에서 돌아가는 어셈블리 루틴을 위한 헤더파일도

작성할수있다.

7.2.2 레퍼런스

레퍼런스(Reference) 는 C++ 의또다른새기능이다.이는함수의인자들을포인터

를직접적으로사용하지않고도전달될수있게한다. 예를들어그림 7.11 의코드를

고려해보자. 사실, 레퍼런스인자들은정말로단순하다. 이들은정말포인터같다. 컴

파일러는 단순히 이를 프로그래머에게 숨기는 것이다. (이는 파스칼 컴파일러가 var
인자들을 포인터로 실행하는 것처럼) 컴파일러가 함수를 호출하는 7 행에서의 어셈

블리코드를생성할때, 이는 y의주소를전달한다. 만일 f를어셈블리로작성한다면

그것이함수의원형이였던것처럼 6)

void f ( int * xp);

레퍼런스는단순히편의를위해서만들어졌는데,특히오버로딩연산자에편리하

다.이는 C++의또다른기능으로구조체나클래스형에대한평범한연산자에의미를

부여하는것이다. 예를들어더하기 (+) 기호는보통문자열객체를합칠때사용한다.

따라서, 만일 a 와 b 가 문자열이라면 a + b 는 문자열 a 와 b 가 합친 결과가 리턴

될 것이다. C++ 은 사실 이 작업을 하기 위해 함수를 호출한다. (사실, 이러한 식들은

operator +(a,b)와같은함수를이용해나타낼수있다)편의를위해,문자열객체의

값을전달하기보단주소를전달하는것이좋을것이다.레퍼런스를사용하지않는다면

이는 operator +(&a, &b)로가능하지만,연산자사용시 &a + &b와같이나타내야

하는불편함이있다.그러나레퍼런스를이용하면이를단순히 a + b만으로가능하고,

자연스러워보인다.

7.2.3 인라인함수

인라인함수(Inline functions) 는 C++의또다른기능이다. 7) 인라인함수들은오류

가잦은,인자를갖는전처리기기반매크로들을대체하기위하여만들어졌다. C에서

숫자를제곱하는메크로는아래와같이생겼다.

#deöne SQR(x) ((x)*(x))

6) 물론, 7.2.1 에서이야기했던것처럼이름맹글링을피하기위해 C 연계로작성해야한다.
7) C 컴파일러들은종종이기능을 ANSI C 의확장형태로지원한다

1 inline int inline_f ( int x )

2 { return x*x ; }

3

4 int f ( int x )

5 { return x*x ; }

6

7 int main()
8 {

9 int y , x = 5;
10 y = f (x );
11 y = inline_f (x );
12 return 0;
13 }

그림 7.12: 인라인예제

전처리기는 C 를 이해하지 못하고, 단순한 치환만을 하기 때문에 대부분의 경우에

올바른 값을 얻기 위해서 소괄호로 묶어야 한다. 그러나, 이와 같이 함에도 불구하고

SQR(x++)와같은식에서는올바른값을얻을수없다.

매크로들은단순한함수의호출에서복잡한함수호출과정을만드는것을생략하기

위해 만들어진다. 서브프로그램에 관한 장에서 다루었듯이, 함수를 호출하는 과정은

몇 개의 단계를 거친다. 매우 단순한 함수를 호출 할 때에는 함수를 호출하는 과정이

함수를실행하는과정보다더오래걸릴수도있다!인라인함수는보통의함수를작성

하는것과매우비슷하지만,코드에서이함수를 CALL하는명령은이루어지지않는다.
그대신에,인라인함수의호출에관한부분은함수가수행하는코드로대체된다. C++
에선함수의정의앞에 inline키워드를붙임으로써이함수를인라인으로선언할수

있다. 예를들어그림 7.12 에서선언된함수를살펴보자. 10 행에서의 f함수의호출

은보통의함수호출을수행한다. (어셈블리에서 x가 ebp-8에, y가 ebp-4에있다고

생각하자)

1 push dword [ebp-8]
2 call _f
3 pop ecx
4 mov [ebp-4], eax

그러나 11행에서나타난 inline_f함수의호출은아래와같다.

1 mov eax, [ebp-8]
2 imul eax, eax
3 mov [ebp-4], eax

위경우인라인시두가지장점이있다.먼저,인라인함수는더빠르다.어떠한인자

도스택에푸시되지않으며,어떠한스택프레임도생성되거나파괴되지않고,어떠한

분기도이루어지지않는다. 두번째로인라인함수는코드를적게사용한다. 물론, 위

경우이것이사실이였지만모든경우에서사실인것은아니다.

1 class Simple {

2 public :
3 Simple (); // 디폴트생성자

4 ~Simple(); // 소멸자

5 int get_data() const; // 멤버함수들

6 void set_data( int );
7 private :
8 int data; // 멤버데이타

9 };
10

11 Simple :: Simple()
12 { data = 0; }
13

14 Simple::~Simple()
15 { /*아무것도안함*/ }

16

17 int Simple :: get_data() const
18 { return data; }

19

20 void Simple :: set_data( int x )

21 { data = x ; }

그림 7.13: 단순한 C++ 클래스

인라인함수의가장큰단점은인라인코드는링크가되지않기때문에인라인함수

의코드는이를사용하는모든파일에서있어야한다.이전의예제의어셈블리코드는

이를보여준다.인라인이아닌함수를호출하는데필요한정보는오직인자,리턴값의

형, 호출규약과함수에대한라벨의이름이다. 이 모든정보는함수의원형에서얻을

수있다.그러나,인라인함수는함수의모든코드에대한정보를필요로하다.이말은

인라인 함수의 어떠한 부분이라도 바뀐다면 이 함수를 사용하는 모든 소스 파일들은

반드시다시컴파일되야만한다.인라인이아닌함수들의경우함수의원형이바뀌지

않는한,이함수를사용하는파일들은굳이다시컴파일이될필요가없다.이러한이유

때문에인라인함수에대한코드는보통헤더파일에위치한다.이는C에서실행가능한

코드는절대로헤더파일에위치하면안된다라는규칙과상충된다.

7.2.4 클래스

C++ 클래스는 객체(object)의 타입을 설명한다. C++ 에서 객체는 데이타 멤버와

함수 맴버8), 혹은 일반적으로 메서드(method) 를 포함할 수 있다. 다시말해, 이는 데

이타와 함수가 연관이 되어 있는 구조체 이다. 그림 7.13에 정의된 단순한 클래스를

살펴보자. Simple타입의변수는단순히 C 에서의하나의 int를멤버로가지는구조

체 와 같다. 이 함수들은 구조체에 해당하는 메모리에 저장되지 않는다. 그러나 멤버사실 C++은 this키워드를이

용하여맴버함수내부에서객

체를 가리키는 포인터에 접근

한다.

함수들은다른함수들과다르다. 이 들은숨겨진인자를주고받는다. 그 인자는멤버

함수가작업하는객체를가리키는포인터이다.

예를들어그림 7.13의 Simple클래스의 set_data메서드를보자.이것이 C에서

작성된다면그림 7.14과같이함수의코드에서작업하는객체에대한포인터를인자로

8) 보통맴버함수(member function) 이라부른다.

void set_data( Simple * object , int x )

{

object−>data = x;
}

그림 7.14: C 버전의 Simple::set_data()

1 _set_data__6Simplei: ; 맹글링 된 이름

2 push ebp
3 mov ebp, esp
4

5 mov eax, [ebp + 8] ; eax = 객체를 가리키는 포인터 (this)
6 mov edx, [ebp + 12] ; edx = 정수 인자

7 mov [eax], edx ; 데이타는 오프셋 0 에 위치

8

9 leave
10 ret

그림 7.15: Simple::set_data( int ) 가컴파일된모습

받아들이는 함수가 될 것이다. DJGPP 컴파일러 (그리고 gcc 와 볼랜드 컴파일러들도

동일)에 -S스위치를사용하면컴파일러가코드와동일한어셈블리코드를만들어낸

다. DJGPP 와 gcc 가생성한어셈블리파일에는 .s확장자가붙는데, 불행이도 NASM
과 MASM 문법과다른 AT&T 어셈블리를사용한다. 9) (Borland 와 MS 컴파일러들은

.asm확장자를가지는MASM문법의어셈블리파일을만든다)그림 7.15은DJGPP의

어셈블리코드를 NASM 문법으로변환한것이다. 문장의의미를파악하기쉽게하기

위해 일부 주석을 달았다. 가장 첫 번째 행에서 set_data 메서드에 클래스의 이름과

인자에 대한 정보를 맹글링한 것으로 이루어진 라벨이 붙었음을 알 수 있다. 라벨에

클래스에 이름에 관한 정보도 포함된 이유는 다른 클래스가 set_data 라는 이름의

메서드를가질수있기때문이다.또한인자들에관한정보도평범한 C++함수들처럼

오버로드할수있도록포함되었다.그러나이전과같이다른컴파일러들은이정보를

다른식으로맹글링할것이다.

그리고 2와 3행에서우리에게친숙한부분이나타난다. 5 행에서스택에저장된

인자는 EAX로저장된다.이는 x인자가아니다.대신에이는숨겨진인자 10)로작업하는

객체를가리키는포인터이다. 6행에서, x인자를 EDX에저장하고 7행에서는 EDX를
EAX가가리키는더블워드에저장한다.이는우리가현재작업하는 Simple의객체의

data멤버이다. 이멤버는클래스의유일한데이타로 Simple의오프셋 0 에위치해

있다.

9) gcc 컴파일러 시스템은 자체적인 어셈블러인 gas 를 가지고 있다. gas 어셈블러는 AT&T 문법을 사
용하므로 컴파일러에 의해 생성된 어셈블리 코드는 gas 형식 으로 되어 있다. 인터넷 상에서 IN-
TEL 과 AT&T 문법의 차이점을 서술한 페이지가 몇 개 있다. 또한 a2i 라는 이름의 무료 프로그램
(http://www.multimania.com/placr/a2i.html) 을통해 AT&T 형식을 NASM 형식으로변환할수있다.

10) 어셈블리에서는숨겨지는것이없다!

1 class Big_int {

2 public :
3 /*
4 * 인자 :
5 * size −정수의크기로, unsigned int가
6 * 사용된수로나타낸다

7 * initial_value − Big_int가보통의 unsigned int였을때초기값

8 */
9 explicit Big_int ( size_t size ,

10 unsigned initial_value = 0);
11 /*
12 * 인자 :
13 * size −정수의크기로, unsigned int가
14 * 사용된수로나타낸다

15 * initial_value − Big_int의초기값의16진수의문자열형태

16 */
17 Big_int ( size_t size ,
18 const char * initial_value );
19

20 Big_int ( const Big_int & big_int_to_copy);
21 ~Big_int ();
22

23 // Big_int의크기를반환(unsigned int의수로)

24 size_t size () const;
25

26 const Big_int & operator = ( const Big_int & big_int_to_copy);
27 friend Big_int operator + ( const Big_int & op1,
28 const Big_int & op2 );
29 friend Big_int operator− ( const Big_int & op1,
30 const Big_int & op2);
31 friend bool operator == ( const Big_int & op1,
32 const Big_int & op2 );
33 friend bool operator < ( const Big_int & op1,
34 const Big_int & op2);
35 friend ostream & operator << ( ostream & os,
36 const Big_int & op );
37 private :
38 size_t size_ ; // 부호없는배열의크기

39 unsigned * number_; //값을가지는부호없는배열을가리키는포인터

40 };

그림 7.16: Big_int 클래스의정의

1 //어셈블리루틴의원형

2 extern ”C” {

3 int add_big_ints( Big_int & res ,
4 const Big_int & op1,
5 const Big_int & op2);
6 int sub_big_ints( Big_int & res ,
7 const Big_int & op1,
8 const Big_int & op2);
9 }

10

11 inline Big_int operator + ( const Big_int & op1, const Big_int & op2)
12 {

13 Big_int result (op1.size ());
14 int res = add_big_ints( result , op1, op2);
15 if ( res == 1)

16 throw Big_int :: Overøow();
17 if ( res == 2)

18 throw Big_int :: Size_mismatch();
19 return result ;
20 }

21

22 inline Big_int operator− ( const Big_int & op1, const Big_int & op2)
23 {

24 Big_int result (op1.size ());
25 int res = sub_big_ints( result , op1, op2);
26 if ( res == 1)

27 throw Big_int :: Overøow();
28 if ( res == 2)

29 throw Big_int :: Size_mismatch();
30 return result ;
31 }

그림 7.17: Big_int 클래스산술연산코드

예제

이 단원은 이 장의 내용을 바탕으로 임의의 크기의 부호가 없는 정수를 나타낸는

C++ 클래스를만든다. 정수의크기가임의로정해지기때문에이는부호가없는정수

(더블워드)들의배열에저장될것이다.이배열은동적할당을이용하면아무크기로나

만들수있다. 더블워드는역순으로저장된다. 11) (i.e 최하위더블워드는 0 의위치에

있다) 그림 7.16 는 Big_int클래스의정의를보여준다. 12) Big_int의크기는데이

타를저장하는데사용되는부호없는배열의크기로계산된다.클래스의 size_데이타

멤버는오프셋이 0이고, number_멤버는오프셋이 4이다.

이예제들을단순화하기위해같은크기의배열을가지는객체인스턴스(instance)
만이더해지거나빼질수있다.

이클래스는 3개의생성자를가진다.하나는 ( 9행)보통의부호없는정수를이용

하여클래스인스턴스를초기화하는것이고다른하나는 ( 18행) 16진수의값을보관

11) 왜냐하면덧셈연산이배열의처음부분부터진행되기때문이다.
12) 이예제의완전한코드에대한예제소스코드를보라.본문은오직코드의일부분만이야기할것이다.

하는문자열을이용하여인스턴스를초기화한다. 마지막으로는 ( 21 행) 복사생성자

(copy constructor) 이다.

여기에서는덧셈과뺄셈연산자들이어떻게작동하는지에대해초점을맞출것인데,

왜냐하면여기서어셈블리가사용되기때문이다.그림 7.17은이연산자들과관계있는

헤더파일의일부분을보여주고있다.이는어셈블리루틴을호출하기위해연산자들이

어떻게설정되어있는지보여준다.서로다른컴파일러가연산자함수를위해다른종

류의맹글링규칙을사용하기때문에인라인연산자함수들은 C연계어셈블리루틴을

호출하게설정되어있다.이는보통의직접적인호출과속도가비슷하며,다른컴파일

러에의해컴파일되어도잘작동하게할수있다.이기술은또한어셈블리에서예외를

던지는(throw) 상황을막을수있다.

왜여기에어셈블리가사용될까?다중정밀도연산을사용하기위해,캐리가하나의

더블워드로부터다음더블워드로전달되야한다는사실을상기해라. C++ (그리고 C)
는 프로그래머로 하여금 CPU 의 캐리 플래스에 접근하는 것을 막고 있다. 이 덧셈을

수행하는유일한방법은 C++ 이독립적으로캐리플래그를다시계산한뒤에그다음

더블워드에더하는수밖에없다.하지만어셈블리에선캐리플래그에마음대로접근할

수있으며, ADC명령을통해자동적으로캐리플래그를더할수있으므로효과적이다.

너무글이길어지는것을막기위해오직 add_big_ints어셈블리루틴만여기에서

다룰것이다.아래는이루틴을위한코드이다. (big_math.asm에서가져옴)

big_math.asm
1 segment .text
2 global add_big_ints, sub_big_ints
3 %define size_offset 0

4 %define number_offset 4

5

6 %define EXIT_OK 0

7 %define EXIT_OVERFLOW 1

8 %define EXIT_SIZE_MISMATCH 2

9

10 ; 덧셈과 뺄셈 서브 루틴들을 위한 인자들

11 %define res ebp+8
12 %define op1 ebp+12
13 %define op2 ebp+16
14

15 add_big_ints:
16 push ebp
17 mov ebp, esp
18 push ebx
19 push esi
20 push edi
21 ;
22 ; op1 을 가리키기 위해 esi 를 설정

23 ; op2 를 가리키기 위한 edi
24 ; res 를 가리키기 위한 ebx

25 mov esi, [op1]
26 mov edi, [op2]
27 mov ebx, [res]
28 ;
29 ; 모든 Big_int 들이 같은 크기여야 함을 명심해라

30 ;
31 mov eax, [esi + size_offset]
32 cmp eax, [edi + size_offset]
33 jne sizes_not_equal ; op1.size_ != op2.size_
34 cmp eax, [ebx + size_offset]
35 jne sizes_not_equal ; op1.size_ != res.size_
36

37 mov ecx, eax ; ecx = Big_int 의 크기

38 ;
39 ; 레지스터를 이에 대응하는 배열을 가리키게 맞춘다.

40 ; esi = op1.number_
41 ; edi = op2.number_
42 ; ebx = res.number_
43 ;
44 mov ebx, [ebx + number_offset]
45 mov esi, [esi + number_offset]
46 mov edi, [edi + number_offset]
47

48 clc ; 캐리 플래그 초기화

49 xor edx, edx ; edx = 0

50 ;
51 ; addition loop
52 add_loop:
53 mov eax, [edi+4*edx]
54 adc eax, [esi+4*edx]
55 mov [ebx + 4*edx], eax
56 inc edx ; 캐리 플래그를 변경하지 말라

57 loop add_loop
58

59 jc overflow
60 ok_done:
61 xor eax, eax ; 리턴값 = EXIT_OK
62 jmp done
63 overflow:
64 mov eax, EXIT_OVERFLOW
65 jmp done
66 sizes_not_equal:
67 mov eax, EXIT_SIZE_MISMATCH
68 done:

1 #include ”big_int .hpp”
2 #include <iostream>

3 using namespace std;
4

5 int main()
6 {

7 try {

8 Big_int b(5,”8000000000000a00b”);
9 Big_int a(5, ”80000000000010230”);

10 Big_int c = a + b;
11 cout << a << ” + ” << b << ” = ” << c << endl ;
12 for( int i=0; i < 2; i++ ) {

13 c = c + a;
14 cout << ”c = ” << c << endl ;
15 }

16 cout << ”c−1 = ” << c − Big_int(5,1) << endl ;
17 Big_int d(5, ”12345678”);
18 cout << ”d = ” << d << endl ;
19 cout << ”c == d ” << (c == d) << endl ;
20 cout << ”c > d ” << (c > d) << endl ;
21 }

22 catch( const char * str ) {

23 cerr << ”Caught: ” << str << endl ;
24 }

25 catch( Big_int :: Overøow ) {

26 cerr << ”Overøow” << endl ;
27 }

28 catch( Big_int :: Size_mismatch ) {
29 cerr << ”Size mismatch” << endl;
30 }

31 return 0;
32 }

그림 7.18: Big_int의단순한사용

69 pop edi
70 pop esi
71 pop ebx
72 leave
73 ret big_math.asm

위의대부분의코드는독자여러분이이해하기에쉽다. 25 에서 27행에서함수를

통해인자로전달된 Big_int객체를가리키는포인터가저장된레지스터를저장한다.

레퍼런스는단지포인터에불과하다는것을상기해라. 31에서35행은세객체배열들

의크기가같은지확인한다. (데이타멤버에접근하기위해 size_의오프셋이포인터

에더해졌다.) 44에서46행에선레지스터값들을조정하여,객체자기자신이아닌각

객체의배열을가리키게하였다. (여기서도객체포인터에 number_멤버의오프셋이

더해졌다)

52 에서 57 행의루프는최하위더블워드순으로배열에저장된정수들을더한다.



어셈블리와 C++ 141

이런순서로더하는이유는확장정밀도산술연산 ( 2.1.5 참조)때문이다. 59행은오

버플로우가발생하였는지확인한다.오버플로우시,최상위더블워드의마지막덧셈에

의해캐리플래그가세트된다.배열에저장된더블워드들이리틀엔디안형식으로되어

있기때문에루프는배열의처음부분부터시작되어배열의끝에서끝난다.

그림 7.18는 Big_int를사용하는짧은예제클래스를보여준다. Big_int상수는

16행과같이반드시명확하게정의되어야한다.이는두가지연유에서중요하다.먼저,

부호가없는정수를 Big_int로변환하는변환생성자가없다.두번째로는오직같은

크기의 Big_int들만이더해질수있다. 이는변환을복잡하게하는데왜냐하면어떠

한크기로변환해야할지모르기때문이다.클래스를좀더복잡하게만든다면크기가

서로다른것들도더해질수있게할수있다. 그러나이를여기에서보임으로써이예

제를복잡하게만들고싶은생각은없다. (그러나, 독자여러분이이를한번해보기를

권한다)

7.2.5 상속과다형성

상속(Inheritance) 을통해하나의클래스를다른클래스에데이터와메서드를상속

할수있다.예를들어그림 7.19의코드를보자.이는두개의클래스 A와 B가나오는데,

B클래스가 A로부터상속받는다.프로그램의출력결과는

Size of a: 4 Offset of ad: 0

Size of b: 8 Offset of ad: 0 Offset of bd: 4

A::m()
A::m()

두개의클래스(A와상속받은 B)의 ad데이타멤버가동일한오프셋을가짐을주목해

라.이는 f함수에 A객체혹은 A를물려받은(i.e.상속받은)클래스의객체를가리키는

포인터를전달했을수도있기때문이다. (i.e상속된)그림 7.20은함수에대한어셈블

리코드를보여준다. (gcc 에의해생성되었고, NASM 형식으로변환됨)

출력결과에서 A의 m메서드가 a와 b객체로부터호출되었을때모두호출되었다

는것을주목하자.어셈블리에서 A::m()으로호출하는것은어셈블리에서코딩하기가

어렵다. 완전한 객체 지향 프로그래밍에서 호출되는 메서드는 어떠한 객체가 함수에

전달되느냐에따라달려있다.이를다형성(polymorphism)이라한다.C++은기본적으

로이기능을사용하지않는다.우리는 virtual키워드를통해이기능을사용할수있다.

그림 7.21 는 이 두클래스가어떻게 바뀌는지 보여준다. 다른코드들은바뀔필요가

없다.다형성은여러가지형태로구현될수있다.불행이도현재 gcc 에서의구현은이

전의구현되었던것보다훨씬복잡하게되었다.여기에대해선깊이들어가지는않고

윈도우즈기반의마이크로소프트와볼랜드컴파일러들이사용하는형태에대해서만

이야기 하고자 한다. 이 컴파일러들에서의 구현은 몇 년 동안 바뀌지 않았을 뿐더러

앞으로도바뀌지않을것이다.

virtual 키워드를붙임으로써프로그램의출력은아래와같이바뀐다 :

Size of a: 8 Offset of ad: 4

Size of b: 12 Offset of ad: 4 Offset of bd: 8

A::m()
B::m()

1 #include <cstddef>
2 #include <iostream>

3 using namespace std;
4

5 class A {

6 public :
7 void __cdecl m() { cout << ”A ::m()” << endl ; }

8 int ad;
9 };

10

11 class B : public A {

12 public :
13 void __cdecl m() { cout << ”B :: m()” << endl ; }

14 int bd;
15 };
16

17 void f ( A * p )

18 {

19 p−>ad = 5;
20 p−>m();
21 }

22

23 int main()
24 {

25 A a;
26 B b;
27 cout << ”Size of a : ” << sizeof(a)
28 << ” Offset of ad: ” << offsetof (A,ad) << endl ;
29 cout << ”Size of b: ” << sizeof(b)
30 << ” Offset of ad: ” << offsetof (B,ad)
31 << ” Offset of bd: ” << offsetof (B,bd) << endl ;
32 f(&a );
33 f(&b);
34 return 0;
35 }

그림 7.19: 상속예제

1 _f__FP1A: ; 맹글링 된 함수 이름

2 push ebp
3 mov ebp, esp
4 mov eax, [ebp+8] ; eax 는 객체를 가리킨다.

5 mov dword [eax], 5 ; ad 를 위해 오프셋 0 을 사용

6 mov eax, [ebp+8] ; 객체의 주소를 A::m() 에 전달

7 push eax
8 call _m__1A ; A::m() 의 맹글링 된 메서드 이름

9 add esp, 4

10 leave
11 ret

그림 7.20: 간단한상속예제를위한어셈블리코드

1 class A {

2 public :
3 virtual void __cdecl m() { cout << ”A ::m()” << endl ; }

4 int ad;
5 };
6

7 class B : public A {

8 public :
9 virtual void __cdecl m() { cout << ”B :: m()” << endl ; }

10 int bd;
11 };

그림 7.21: 다형적상속(Polymorphic Inheritance)

1 ?f@@YAXPAVA@@@Z:
2 push ebp
3 mov ebp, esp
4

5 mov eax, [ebp+8]
6 mov dword [eax+4], 5 ; p->ad = 5;
7

8 mov ecx, [ebp + 8] ; ecx = p
9 mov edx, [ecx] ; edx = vtable 을 가리키는 포인터

10 mov eax, [ebp + 8] ; eax = p
11 push eax ; "this" 포인터를 푸시

12 call dword [edx] ; vtable 의 첫 번째 함수 호출

13 add esp, 4 ; 스택을 정리한다

14

15 pop ebp
16 ret

그림 7.22: f()함수를위한어셈블리코드

이제, f의두번째호출에선 B객체를전달했기때문에 B::m()메서드를호출하게

된다. 하지만 달라진 것은 이것이 다가 아니다. A 의 크기는 8 이 된다. (그리고 B 는

12) 또한 ad의오프셋은 0이아닌 4가된다.그렇다면오프셋 0에는무엇이있을까?

이것에대한대답은어떻게다형성이구현되었는지에따라달라진다.

가상메서드를한개라도가지는 C++클래스들은메서드포인터들의배열을가리

키는포인터의추가적인숨겨진영역이있다. 13) 이테이블은종종가상테이블(vtable)
이라불린다. A와 B클래스의가상테이블을가리키는포인터는오프셋 0에저장되어

있다.윈도우즈컴파일러는언제나이포인터들을클래스시작부분인상속트리의최

상단에위치해놓는다.함수 f (그림 7.19)의가상메서드버전의어셈블리코드 (그림

7.22)를보면 m메서드호출은라벨이아님을알수있다. 9행은객체로부터가상테이

블의주소값을찾는다. 11 행에서객체의주소값은스택에푸시된다. 12 행에서가상

테이블의첫번째주소14) 로분기함으로써가상메서드를호출한다.이호출은라벨을

13) 가상메서드가없는클래스들의경우 C++컴파일러들은언제나동일한데이타멤버를가지는보통의
C 구조체와호환가능한클래스를만든다

14) 당연하게도이값은이미 ECX 레지스터에들어있다.그값은 8행에서들어갔다.또한 10행을제거

1 class A {

2 public :
3 virtual void __cdecl m1() { cout << ”A ::m1()” << endl ; }

4 virtual void __cdecl m2() { cout << ”A ::m2()” << endl ; }

5 int ad;
6 };
7

8 class B : public A { // B inherits A ’ s m2()

9 public :
10 virtual void __cdecl m1() { cout << ”B :: m1()” << endl ; }

11 int bd;
12 };
13 /*주어진객체의 vtable을출력 */
14 void print_vtable ( A * pa )

15 {

16 // p는 pa를더블워드의배열로생각한다

17 unsigned * p = reinterpret_cast<unsigned *>(pa);
18 // vt는가상테이블을포인터들의배열로생각한다

19 void ** vt = reinterpret_cast<void **>(p [0]);
20 cout << hex << ”vtable address = ” << vt << endl ;
21 for( int i=0; i < 2; i++ )

22 cout << ”dword ” << i << ” : ” << vt[ i ] << endl ;
23

24 // 가상함수를이식성이매우떨어지는방법으로호출한다!

25 void (*m1func_pointer)(A *); // 함수포인터변수

26 m1func_pointer = reinterpret_cast<void (*)(A*)>(vt [0]);
27 m1func_pointer(pa); //함수포인터를통해메서드m1을호출

28

29 void (*m2func_pointer)(A *); // 함수포인터변수

30 m2func_pointer = reinterpret_cast<void (*)(A*)>(vt [1]);
31 m2func_pointer(pa); // 함수포인터를통해메서드m2을호출

32 }

33

34 int main()
35 {

36 A a; B b1; B b2;
37 cout << ”a : ” << endl ; print_vtable (&a );
38 cout << ”b1: ” << endl ; print_vtable (&b);
39 cout << ”b2: ” << endl ; print_vtable (&b2);
40 return 0;
41 }

그림 7.23: 좀더복잡한예제

&B::m1()

&A::m2()

vtable

vtablep
0

그림 7.24: b1의내부적표현

a:
vtable address = 004120E8
dword 0: 00401320

dword 1: 00401350

A::m1()
A::m2()
b1:
vtable address = 004120F0
dword 0: 004013A0
dword 1: 00401350

B::m1()
A::m2()
b2:
vtable address = 004120F0
dword 0: 004013A0
dword 1: 00401350

B::m1()
A::m2()

그림 7.25: 그림 7.23 의프로그램의출력값

사용하지 않고, EDX 가 가리키는 코드 주소로 분기한다. 이러한 형식의 호출은 나중

바인딩(late binding) 라한다. . 나중바인딩은코드가실행될때까지어떠한메서드를

호출할지결정할수없게한다.이를통해객체의올바른메서드를호출하게한다.보통

의경우 (그림 7.20)는호출되는형태는빠른바인딩(early binding) 이라한다. (컴파일

타임(compile time)에무슨메서드로분기해야할지알수있다.)

세심한독자들의경우왜그림 7.21의클래스메서드들을 __cdecl키워드를사용

함으로써명백하게 C호출규약으로정의하였는지에대해궁금해할것이다.기본적으

로마이크로소프트는표준 C규약과다른호출규약을 C++클래스메서드에사용한다.

이는대신에메서드에서작업하는객체를가리키는포인터를스택을사용하는대신에

ECX레지스터에넣어전달한다.스택은메서드의다른인자들을전달하는데사용된다.

__cdecl변경자는표준 C호출규약을사용하게끔한다.볼랜드 C+++은 C호출규약

을기본으로사용한다.

하는대신에그다음행에 ECX 를푸시하게바꾸면된다.이코드는별로효율적이지않는데왜냐하면
컴파일러최적화옵션이켜지지않은채생성된코드이기대문이다

이번에는 약간 더 복잡한 예제를 살펴보자 (그림 7.23). 여기서, 클래스 A 와 B 는

각각2개의메서드를가지고있다 : m1과 m2. B클래스가자신의 m2메서드를정의하는

것이아니라 A클래스의메서드를상속받는다는것을기억해라.그림 7.24 은 b객체

가 메모리 상에 어떻게 나타나는지 보여준다. 그림 7.25 은 프로그램의 출력 결과를

보여준다. 먼저, 각 객체의 가상테이블 주소를 보자. 두 개의 B 객체들의 가상 테이블

주소들은같으므로이들은같은가상테이블을공유한다.가상테이블은클래스의특징

이지객체가아니다 (정적데이타멤버처럼).다음으로가상테이블에있는주소값들을

보라. 출력 결과를 보면 우리는 m1 메서드 포인터가 오프셋 0 (혹은 더블워드 0) 에

있고 m2는오프셋 4 (더블워드 1) 에있음을알수있다. m2메서드포인터는 A와 B의
가상테이블들에서동일한데,왜냐하면 B는 m2메서드를 A클래스로부터상속받았기

때문이다.

25 에서 32 행은객체에대한객체의가상테이블의주소값을읽어서가상함수를

호출할 수 있는지를 보여주고 있다. 15). 메서드 주소는 C 유형의 함수 포인터와 this
포인터가저장된다. 그림 7.25 에서의출력값에서우리는이것이실제로작동한다는

것을알수있다.그러나,코드를이렇게쓰지는말라. 이는오직가상메서드들이가상

테이블을어떻게사용하는지보여주기위해서쓴것이다.

여기서몇가지교훈을얻을수있다. 하나는우리가이진파일에클래스변수들을

읽고쓸때매우주의해야한다는점이다. 우리는단순히전체객체를읽거나쓸수없

는데, 왜냐하면 이는 파일에 가상테이블 포인터를 읽거나 쓸 것이기 때문이다. 이는

프로그램의메모리상에가상테이블이있는곳을가리키는포인터이고, 아는프로그

램마다다를것이기때문이다.이같은문제가 C의구조체에서도발생할수있지만, C
구조체의경우프로그래머가특별히지정해야만포인터를가지게된다. A나 B클래스

에선명백히정의된포인터들이없다.

재차강조하지만컴파일러들은서로다른방식으로가상메서드를구현했다는점을

잊지말아야한다.윈도우즈에서는COM (Component Object Model)클래스객체들이

COM인터페이스를구현하기위해가상테이블을사용한다.16) 마이크로소프트와같은

방법으로가상메서드를가상테이블을이용하여구현한컴파일러만이 COM클래스를

생성할수있다. 이는볼랜드가마이크로소프트와같은구현을사용하였고, 왜 gcc 에
서는 COM 클래스를만들수없는지알수있다.

가상메서드를위한코드는비-가상메서드것과정확이같다. 오직이를호출하는

코드만이다를뿐이다.만일컴파일러가어떤가상메서드가호출되는지정확히알수

있다면, 이는가상테이블을무시하고직접적으로메서드를호출할수있다. (e.g 빠른

바인딩)

7.2.6 다른 C++기능들

다른 C++ 기능(e.g 런타임 형식의 정보, 예외 처리 및 다중 상속)에 대해 이야기

하는것은이책의범위를훌쩍뛰어넘는다.만일독자여러분이더자세히알고싶다면,

Ellis 와 Stroustrup 이쓴 e Annotated C++ Reference Manual 과 Stroustrup 이쓴 e
Design and Evolution of C++ 을읽어보기를권한다.

15) 이코드는오직 MS 와볼랜드컴파일러에서만작동된다. gcc 에서는작동하지않는다
16) COM 클래스는또한 __stdcall 호출규약을사용한다.표준 C 가아니다

A 비-부동소수점명령들

여기서는 인텔 80x86 계열 CPU 들의 비-부동 소수점 명령들에 대한 정보와 사용

형식들에대해설명하고자한다.

명령들의사용형식을나타내기위해아래와같은약어들을이용했다.

R 범용레지스터

R8 8 비트레지스터

R16 16 비트레지스터

R32 32 비트레지스터

SR 세그먼트레지스터

M 메모리

M8 바이트

M16 워드

M32 더블워드

I 즉시데이터

이약어들은여러개의피연산자를가지는명령들을위해조합되어사용될수있다.예

를 들어, R, R 라 한다면 이 명령이 두 개의 레지스터를 피연산자로 갖는 다는 것이다.

두개의피연산자를가지는많은수의명령들은동일한피연산자를가질수있다. O2
는 다음과 같은 피연산자를 표현하는데 사용한다 : R,R R,M R,I M,R M,I. 만일 8 비트

레지스터혹은메모리가피연산자로사용될수있다면 R/M8 과같이나타낸다.

아래의 표는 또한 각 명령에 따라 플래그 레지스터들의 비트들이 어떻게 영향을

받는지보여주고있다. 만일특정한칸이빈칸이라면, 이에대응하는비트는영향을

받지않는다는것이다.만일비트가특정한값으로만바뀐다면1또는0을나타내었다.

만일명령의피연산자에의해비트의값이결정된다면 C 라고나타낸다. 만일비트가

특정한규칙없이바뀐다면 ?로나타낸다.또한,방향플래그의값을바꾸는명령은 CLD
와 STD밖에없기때문에표에는나타내지않았다.

플래그

이름 설명 형식 O S Z A P C
ADC 캐리와더한다 O2 C C C C C C
ADD 정수를더한다 O2 C C C C C C

플래그

이름 설명 형식 O S Z A P C
AND And 비트연산 O2 0 C C ? C 0

BSWAP 바이트를스왑(swap) R32
CALL 루틴을호출 R M I
CBW 바이트를워드로변환

CDQ 더블워드를 쿼드워드로

변환

CLC 캐리초기화 0

CLD 방향플래그초기화

CMC 캐리에보수 C
CMP 정수를비교 O2 C C C C C C
CMPSB 바이트를비교 C C C C C C
CMPSW 워드를비교 C C C C C C
CMPSD 더블워드를비교 C C C C C C
CWD 워드를 더블워드로 변환

후 DX:AX 에저장

CWDE 워드를 더블워드로 변환

후 EAX 에저장

DEC 정수를감소 R M C C C C C
DIV 부호없는나눗셈 R M ? ? ? ? ? ?
ENTER 스택프레임을만듦 I,0
IDIV 부호있는나눗셈 R M ? ? ? ? ? ?
IMUL 부호있는곱셈 R M

R16,R/M16

R32,R/M32

R16,I R32,I
R16,R/M16,I
R32,R/M32,I

C ? ? ? ? C

INC 정수를증가 R M C C C C C
INT 인터럽트발생 I
JA 초과시분기 I
JAE 이상이면분기 I
JB 미만시분기 I
JBE 이하시분기 I
JC 캐리있으면분기 I
JCXZ CX = 0 이면분기 I
JE 같으면분기 I
JG 초과시분기 I
JGE 이상시분기 I
JL 미만시분기 I
JLE 이하시분기 I
JMP 무조건분기 R M I

플래그

이름 설명 형식 O S Z A P C
JNA 초과아니면분기 I
JNAE 미만이면분기 I
JNB 미만아니면분기 I
JNBE 초과면분기 I
JNC 캐리없으면분기 I
JNE 같지않으면분기 I
JNG 초과아니면분기 I
JNGE 미만이면분기 I
JNL 미만아니면분기 I
JNLE 초과면분기 I
JNO 오버플로우없으면분기 I
JNS 부호없으면분기 I
JNZ 0 이아니면분기 I
JO 오버플로우시분기 I
JPE 짝수패리티시분기 I
JPO 홀수패리티시분기 I
JS 부호있으면분기 I
JZ 0 이면분기 I
LAHF AH 에 플래그 레지스터

저장

LEA 주소값을계산 R32,M
LEAVE 스택프레임을떠남.

LODSB 바이트를불러옴

LODSW 워드를불러옴

LODSD 더블워드를불러옴

LOOP 루프 I
LOOPE/LOOPZ 같다면루프 I
LOOPNE/LOOPNZ 다르면루프 I
MOV 데이타를이동 O2

SR,R/M16

R/M16,SR
MOVSB 바이트를이동

MOVSW 워드를이동

MOVSD 더블워드를이동

MOVSX 부호있는것을이동 R16,R/M8

R32,R/M8

R32,R/M16

MOVZX 부호없는것을이동 R16,R/M8

R32,R/M8

R32,R/M16

MUL 부호없는곱셈 R M C ? ? ? ? C

플래그

이름 설명 형식 O S Z A P C
NEG 음을취함 R M C C C C C C
NOP 명령없음

NOT 1 의보수를취함 R M
OR OR 연산 O2 0 C C ? C 0

POP 스택에서팝 R/M16

R/M32

POPA 전체를팝

POPF 플래그레지스터를팝 C C C C C C
PUSH 스택에푸시 R/M16

R/M32 I
PUSHA 전체를푸시

PUSHF 플래그레지스터를푸시

RCL 캐리와함께왼쪽회전 R/M,I
R/M,CL

C C

RCR 캐리와함께오른쪽회전 R/M,I
R/M,CL

C C

REP 반복

REPE/REPZ 같다면반복

REPNE/REPNZ 다르면반복

RET 리턴

ROL 왼쪽으로회전 R/M,I
R/M,CL

C C

ROR 오른쪽으로회전 R/M,I
R/M,CL

C C

SAHF AH 를 플래그 레지스터

에복사

C C C C C

SAL 왼쪽으로쉬프트 R/M,I R/M,
CL

C

SBB 받아내림과함께뺄셈 O2 C C C C C C
SCASB 바이트를찾기 C C C C C C
SCASW 워드를찾기 C C C C C C
SCASD 더블워드를찾기 C C C C C C
SETA 초과시세트 R/M8

SETAE 이상시세트 R/M8

SETB 미만시세트 R/M8

SETBE 이하시세트 R/M8

SETC 캐리를세트 R/M8

SETE 같으면세트 R/M8

SETG 초과시세트 R/M8

SETGE 이상시세트 R/M8

SETL 미만시세트 R/M8

플래그

이름 설명 형식 O S Z A P C
SETLE 이하시세트 R/M8

SETNA 초과가아니면세트 R/M8

SETNAE 미만이면세트 R/M8

SETNB 미만이아니면세트 R/M8

SETNBE 초과면세트 R/M8

SETNC 캐리가없으면세트 R/M8

SETNE 같지않으면세트 R/M8

SETNG 초과가아니면세트 R/M8

SETNGE 미만이면세트 R/M8

SETNL 미만이아니면세트 R/M8

SETNLE 초과면세트 R/M8

SETNO 오버플로우없으면세트 R/M8

SETNS 부호가없으면세트 R/M8

SETNZ 0 이아니면세트 R/M8

SETO 오버플로우면세트 R/M8

SETPE 짝수패리티면세트 R/M8

SETPO 홀수패리티면세트 R/M8

SETS 부호가있으면세트 R/M8

SETZ 0 이면세트 R/M8

SAR 오른쪽으로산술쉬프트 R/M,I R/M,
CL

C

SHR 오른쪽으로논리쉬프트 R/M,I R/M,
CL

C

SHL 왼쪽으로논리쉬프트 R/M,I R/M,
CL

C

STC 캐리플래그를세트 1

STD 방향플래그를세트

STOSB 바이트를저장

STOSW 워드를저장

STOSD 더블워드를저장

SUB 뺄셈 O2 C C C C C C
TEST 논리비교 R/M,R

R/M,I
0 C C ? C 0

XCHG 서로교환 R/M,R
R,R/M

XOR XOR 비트연산 O2 0 C C ? C 0

B 부동소수점명령들

여기에서는 80x86 수치부프로세서명령들에대해살펴보겠다.아래표에서는명

령에대해간단하게설명한다. 공간을절약하기위해명령이스택을팝하는지에대해

서는설명에포함시키지않았다.

명령에대해무슨피연산자가사용될수있는지알려주기위해아래와같은약어를

사용하였다.

STn 부프로세서레지스터

F 메모리상의단일정밀도수

D 메모리상의 2배정밀도수

E 메모리상의확장정밀도수

I16 메모리상의정수워드

I32 메모리상의정수더블워드

I64 메모리상의정수쿼드워드

별표(*))로표기된명령들은펜티엄프로이상의프로세서들에서만지원된다.

명령 설명 형식

FABS ST0 = |ST0|
FADD src ST0 += src STn F D
FADD dest, ST0 dest += STO STn
FADDP dest[,ST0] dest += ST0 STn
FCHS ST0 = −ST0
FCOM src ST0와 src를비교 STn F D
FCOMP src ST0와 src를비교 STn F D
FCOMPP src ST0와 ST1을비교

FCOMI*) src 비교후결과를플래그레지스터에저장 STn
FCOMIP*) src 비교후결과를플래그레지스터에저장 STn
FDIV src ST0 /= src STn F D
FDIV dest, ST0 dest /= STO STn
FDIVP dest[,ST0] dest /= ST0 STn
FDIVR src ST0 = src/ST0 STn F D
FDIVR dest, ST0 dest = ST0/dest STn
FDIVRP dest[,ST0] dest = ST0/dest STn
FFREE dest 비었다고표시 STn
FIADD src ST0 += src I16 I32
FICOM src ST0와 src를비교 I16 I32
FICOMP src ST0와 src를비교 I16 I32
FIDIV src STO /= src I16 I32
FIDIVR src STO = src/ST0 I16 I32
FILD src src 를스택에푸시 I16 I32 I64
FIMUL src ST0 *= src I16 I32
FINIT 부프로세서를초기화

FIST dest ST0저장 I16 I32

명령 설명 형식

FISTP dest ST0저장 I16 I32 I64
FISUB src ST0 -= src I16 I32
FISUBR src ST0 = src - ST0 I16 I32
FLD src src 를스택에푸시 STn F D E
FLD1 1.0 을스택에푸시

FLDCW src 제어워드레지스터를불러옴 I16
FLDPI π 를스택에푸시

FLDZ 0.0 을스택에푸시

FMUL src ST0 *= src STn F D
FMUL dest, ST0 dest *= STO STn
FMULP dest[,ST0] dest *= ST0 STn
FRNDINT ST0를반올림

FSCALE ST0 = ST0× 2bST1c

FSQRT ST0 =
√
STO

FST dest ST0저장 STn F D
FSTP dest ST0저장 STn F D E
FSTCW dest 제어워드레지스터를저장 I16
FSTSW dest 상태워드레지스터를저장 I16 AX
FSUB src ST0 -= src STn F D
FSUB dest, ST0 dest -= STO STn
FSUBP dest[,ST0] dest -= ST0 STn
FSUBR src ST0 = src-ST0 STn F D
FSUBR dest, ST0 dest = ST0-dest STn
FSUBP dest[,ST0] dest = ST0-dest STn
FTST ST0를 0.0 과비교

FXCH dest ST0와dest의값을교환 STn

찾아보기

【기호】

16진법 . . . . . . . . . . . . . . . . . . . . . . . . . 2–4
2의보수 . . . . . . . . . . . . . . . . . . . . . 25–26

산술연산 . . . . . . . . . . . . . . . . 29–33
80x86

CPU. . . . . . . . . . . . . . . . . . . . . . . . . .5

【 A 】

ADC . . . . . . . . . . . . . . . . . . . . . . . . . 32, 47
ADD . . . . . . . . . . . . . . . . . . . . . . . . . 11, 32
AND . . . . . . . . . . . . . . . . . . . . . . . . . . . . 44
array1.asm . . . . . . . . . . . . . . . . . . . 85–89

【 B 】

BSWAP . . . . . . . . . . . . . . . . . . . . . . . . . . 52
BYTE . . . . . . . . . . . . . . . . . . . . . . . . . . . .14

【 C 】

C driver . . . . . . . . . . . . . . . . . . . . . . . . . .17
C 와소통하기 . . . . . . . . . . . . . . . . . . . . 70
C 와함께작업하기 . . . . . . . . . . . . . . . 78
C++ . . . . . . . . . . . . . . . . . . . . . . . 129–146

Big_int 예제 . . . . . . . . . . . 137–141
name mangling . . . . . . . . . . . . . 129
references . . . . . . . . . . . . . . . . . . 132
virtual . . . . . . . . . . . . . . . . . . . . . .141
vtable . . . . . . . . . . . . . . . . . 143, 146
나중바인딩 . . . . . . . . . . . . . . . . 145
다형성 . . . . . . . . . . . . . . . . 141–146
레퍼런스 . . . . . . . . . . . . . . . . . . . 132
멤버함수 . . . . . . 메서드를참고, 2

복사생성자 . . . . . . . . . . . . . . . . 138
빠른바인딩 . . . . . . . . . . . . . . . . 145
상속 . . . . . . . . . . . . . . . . . . 141–146
이름맹글링 . . . . . . . . . . . . . . . . 132
인라인함수 . . . . . . . . . . . 132–134

클래스 . . . . . . . . . . . . . . . . 134–146
형이안전한링킹 . . . . . . . . . . . .130

CALL . . . . . . . . . . . . . . . . . . . . . . . . 60–61
CBW . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27
CLC . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33
CLD . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 92
clock . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .5
CMP . . . . . . . . . . . . . . . . . . . . . . . . . 33–34
CMPSB . . . . . . . . . . . . . . . . . . . . . . . . . . 94
CMPSD . . . . . . . . . . . . . . . . . . . . . . . . . .94
CMPSW . . . . . . . . . . . . . . . . . . . . . . . . . 94
COM . . . . . . . . . . . . . . . . . . . . . . . . . . . 146
CPU. . . . . . . . . . . . . . . . . . . . . . . . . . . .5–6
CWD . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27
CWDE. . . . . . . . . . . . . . . . . . . . . . . . . . .28

【 D 】

DEC . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 12
DIV . . . . . . . . . . . . . . . . . . . . . . . . . .30, 42
do while 루프 . . . . . . . . . . . . . . . . . . . . .38
DWORD . . . . . . . . . . . . . . . . . . . . . . . . .14

【 F 】

FABS . . . . . . . . . . . . . . . . . . . . . . . . . . . 112
FADD . . . . . . . . . . . . . . . . . . . . . . . . . . 109
FADDP . . . . . . . . . . . . . . . . . . . . . . . . . 109
FCHS . . . . . . . . . . . . . . . . . . . . . . . . . . 112
FCOM. . . . . . . . . . . . . . . . . . . . . . . . . .111
FCOMI . . . . . . . . . . . . . . . . . . . . . . . . . 112
FCOMIP . . . . . . . . . . . . . . . . . . . 112, 113
FCOMP . . . . . . . . . . . . . . . . . . . . . . . . 111
FCOMPP . . . . . . . . . . . . . . . . . . . . . . . 111
FDIV . . . . . . . . . . . . . . . . . . . . . . . . . . . 111
FDIVP . . . . . . . . . . . . . . . . . . . . . . . . . 111
FDIVR . . . . . . . . . . . . . . . . . . . . . . . . . 111
FDIVRP . . . . . . . . . . . . . . . . . . . . . . . . 111

FFREE . . . . . . . . . . . . . . . . . . . . . . . . . .109
FIADD . . . . . . . . . . . . . . . . . . . . . . . . . 109
FICOM . . . . . . . . . . . . . . . . . . . . . . . . . 111
FICOMP. . . . . . . . . . . . . . . . . . . . . . . .111
FIDIV . . . . . . . . . . . . . . . . . . . . . . . . . . 111
FIDIVR. . . . . . . . . . . . . . . . . . . . . . . . .111
FILD . . . . . . . . . . . . . . . . . . . . . . . . . . . 109
FIST . . . . . . . . . . . . . . . . . . . . . . . . . . . .109
FISUB . . . . . . . . . . . . . . . . . . . . . . . . . . 110
FISUBR. . . . . . . . . . . . . . . . . . . . . . . . .110
FLD . . . . . . . . . . . . . . . . . . . . . . . . . . . . 109
FLD1 . . . . . . . . . . . . . . . . . . . . . . . . . . .109
FLDCW . . . . . . . . . . . . . . . . . . . . . . . . 109
FLDZ. . . . . . . . . . . . . . . . . . . . . . . . . . .109
FMUL . . . . . . . . . . . . . . . . . . . . . . . . . . 111
FMULP . . . . . . . . . . . . . . . . . . . . . . . . .111
FSCALE . . . . . . . . . . . . . . . . . . . 112, 113
FSQRT . . . . . . . . . . . . . . . . . . . . . . . . . 112
FST . . . . . . . . . . . . . . . . . . . . . . . . . . . . 109
FSTCW. . . . . . . . . . . . . . . . . . . . . . . . .109
FSTP . . . . . . . . . . . . . . . . . . . . . . . . . . . 109
FSTSW . . . . . . . . . . . . . . . . . . . . . . . . . 112
FSUB . . . . . . . . . . . . . . . . . . . . . . . . . . .110
FSUBP . . . . . . . . . . . . . . . . . . . . . . . . . 110
FSUBR . . . . . . . . . . . . . . . . . . . . . . . . . 110
FSUBRP . . . . . . . . . . . . . . . . . . . . . . . . 110
FTST . . . . . . . . . . . . . . . . . . . . . . . . . . . 111
FXCH . . . . . . . . . . . . . . . . . . . . . . . . . . 109

【 G 】

gas . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 135

【 I 】
I/O . . . . . . . . . . . . . . . . . . . . . . . . . . 14–16

asm_io library . . . . . . . . . . . . . . . .16
dump_math . . . . . . . . . . . . . . . 16
dump_mem. . . . . . . . . . . . . . . .16
dump_regs . . . . . . . . . . . . . . . . .15
dump_stack . . . . . . . . . . . . . . . .16
print_char . . . . . . . . . . . . . . . . . 15
print_int . . . . . . . . . . . . . . . . . . .15
print_nl . . . . . . . . . . . . . . . . . . . 15
print_string . . . . . . . . . . . . . . . . 15
read_char . . . . . . . . . . . . . . . . . .15

read_int . . . . . . . . . . . . . . . . . . . 15
어셈블리 I/O 라이브러리 . . . . . 14

IDIV . . . . . . . . . . . . . . . . . . . . . . . . . . . . 30
if 문 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 37
IMUL . . . . . . . . . . . . . . . . . . . . . . . . 29–30
INC . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 12

【 J 】
JC . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JE . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JG . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JGE . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .36
JL . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JLE . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JMP . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34
JNC . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JNE . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JNG . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JNGE . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JNL . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .36
JNLE . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
JNO . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JNP . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JNS . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .35
JNZ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JO . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JP . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JS . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
JZ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35

【 L 】

LAHF . . . . . . . . . . . . . . . . . . . . . . . . . . 112
LEA . . . . . . . . . . . . . . . . . . . . . . . . . .73, 89
listing ĕle . . . . . . . . . . . . . . . . . . . . . . . . 21

